
前言

Excel是微软公司出品的Office系列办公软件中的一个组件。确切地说，它是一个电子表格软件，可以用来制作电子表格，完成许多复杂的数据运算。

本书主要面向希望深入学习Excel公式、函数和基本数据分析方法的用户，目的是帮助读者全面掌握Excel的函数使用方法。本书一方面由浅入深、循序渐进地阐述使用Excel的函数和数据分析的方法，另一方面则通过实例来展现函数功能和数据分析方法，使读者通过实践加深理解。

本书在内容的编排和目录组织上都十分讲究，争取让读者能够快速掌握软件的使用方法。章名统一按照函数功能来命名，这样读者就可以明确知道每一章将要学到什么知识。讲解具体知识的时候，尽量避免冗长的知识讲解，直接切入主题，告诉读者如何实现特定功能，让读者在实际操作中熟悉软件的使用。严格介绍每一个应用的实例，强调实例的讲解，这样可以保证读者在实践的过程中提高制作水平。

本书有如下特点：

·选取实例进行讲解：虽然Excel是比较流行的办公软件之一，许多用户都有使用Excel的经验，但是一般的用户会发觉，Excel入门并不困难，但是希望通过Excel的函数来处理复杂数据，则还需要深入地了解函数和公式的内容。本书不像其他教程一样，只单纯地讲解如何使用各个命令，而是通过精心设计的多个经典实例，将Excel函数的基本功能和操作技巧融入到范例当中进行讲解。

·选取热门领域：作者在讲解每一个知识点之前，充分考虑了Excel的知识和实践工作的结合，精心挑选工程应用、商务管理、人力资源管理等各个领域的应用，使读者不仅仅单纯地学到Excel函数的操作技巧，而且提高Excel的实际应用能力。

·讲解细致：每个实例的制作步骤都以通俗易懂的语言阐述，并穿插讲解使用技巧，在阅读时就像听课一样详细而贴切。读者只需要按照步骤操作，就可以学习到Excel操作的方法和技巧。

·内容深入：本书的实例都有一定的代表性和通用性，因此有些实例的公式很长，综合了多个知识点，读者在学习的过程中要仔细体会。

本书具有知识全面、实例精彩、指导性强的特点，力求以全面的知识性及丰富的实例来指导读者深入学习Excel函数的使用方法。本书不仅可以当作学习Excel函数的范例教材，更可以当作随查随用的工具宝典，最适合自学及进修应用，是追求Excel操作更上层楼的读者的绝佳选择。读者对象是Excel软件使用者、办公室计算机用户、中高级计算机爱好者。

作者

2014年8月


第1章 函数应用基础

什么是函数、函数怎么用、用来干什么，这是大部分对函数还认识不够的读者一看到函数就会想起的问题，本章就将就这些问题展开讲解和讨论，试图让没有函数概念的人们，对函数有一个基本的认识，去除对函数的陌生感，熟悉函数的定义表达方式，以便后面章节中去理解和应用函数。

本章的主要内容是讲解函数应用的基础知识，以及涉及到的概念和元素。建议对函数了解不多的读者对本章及下章有一定认识后再学习后面的章节。

1.1　函数和公式

函数和公式是Excel中重要的概念，Excel的数据处理中离不开函数和公式，在本书中，函数与公式更是随处可见。

1.1.1　什么是公式

公式，是Excel中由用户自定义的计算过程，其形式如“=A2+C2”。在公式中可以进行具体数值的各种数学运算，也可以使用引用将工作表中的数据引入各种计算当中。

1.1.2　什么是函数

函数是Excel中预定义的一些公式，其将一些特定的计算过程通过程序固定下来，使用一些称为参数的特定数值按特定的顺序或结构进行计算，将其命名后供用户调用。用户可以直接用它们对某个区域内的数值进行一系列运算，如分析和处理日期值和时间值、确定贷款的支付额、确定单元格中的数据类型、计算平均值、排序显示和运算文本数据，等等。

1.1.3　公式与函数的关系

函数与公式既有区别又互相联系。函数是公式的一种，是已预先定义计算过程的公式，函数的计算方式和内容已完全固定，用户只能通过改变函数参数的取值来更改函数的计算结果；而公式就拥有更多的自由，可以由用户来自定义计算过程和计算方向，用户可以通过更改公式的所有元素来改变计算的结果。

但是函数的优点在于，其能够准确地完成复杂的计算过程，且能够保证其计算的准确度，而很多时候，用户在自定义公式时，会因种种原因导致计算准确度不足，误差较大。而且对于某种大量的重复计算，使用了函数是更好的选择。

函数与公式各有优点也都有不足，在实际工作当中，二者常常并肩作战，取长补短。在长而复杂的公式中，函数包揽着大量的可以使用预定义计算过程的计算，用户只需要将不同的函数使用适当的运算符连接起来就可完成较具用自由度的运算。

如果函数要以公式的形式出现，它必须有两个组成部分，一个是函数名称前面的等号，另一个则是函数本身。

1.2　函数的结构

函数是预定义计算过程的公式，其具有特定的格式和结构。图1.1是最常用的函数SUM的结构示意图。

图1.1　SUM函数的结构示意图

函数由以下几个元素构成：

（1）函数名称。一个函数只有唯一的一个名称，它决定了函数的功能和用途。

函数名称是函数之间进行区分的标志，同时也可以从英文字意中了解函数的主要功能，如“SUM”在英文中就是加和的意思，SUM函数的主要功能是将指定区域的数据求和。

（2）函数参数。参数是函数中最复杂的组成部分，它规定了函数的运算对象、顺序或结构等。

在函数的使用当中，参数的设置至关重要，只有设置正确参数，函数才给返回正确的计算结果。在本章的下一小节，将专门讲解函数的参数。

（3）连接符。包换“=”、“,”、“（”和“）”。

任何函数和公式都以“=”开头，输入“=”后，Excel会自动将其后的内容作为公式处理。注意只有“=”在公式首位时才有此作用。在公式首位之外的地方输入“=”都将导致公式错误。

函数以函数名称开始，其参数则以“（”开始，至“）”结束。每个函数必定会对应一对括号，在多层函数嵌套使用时，尤其要注意一个函数一定要对应一对括号。“，”在函数中将各个参数区分开来。

注意：以上的连接符都必须是英文符号，中文连接符将导致公式错误。

1.3　函数的使用方法

对函数的概念有一定了解后，再来学习如何使用函数。本节将带领读者一步步学习从选用函数到插入函数，再到函数的嵌套等一系列内容。

1.3.1　选择适用的函数

Excel中一共有336个函数可供用户直接调用。其中又分10大类，如表1.1所示。

表1.1　函数的分类

如此众多的函数，如何从中选择合适的函数完成工作需求，这是一个重要问题。

用户可以有几种选择：

第一种，使用“插入函数”对话框选择。

【操作步骤】

（1）选择“公式”标签，然后单击“插入函数”按钮，打开“插入函数”对话框，选择大类如图1.2所示。

（2）选择相应的大类后，点选各个函数，可以对话框的下部看到对所选函数的功能注释，可以据此进行选择，如图1.3所示。

图1.2　选择函数的大类

图1.3　查看每个函数的功能

第二种，利用Excel帮助进行查找。

【操作步骤】

（1）根据所需要进行的计算，按“F1”键调用Excel帮助进行查找，如查找“求和”，所有与求和相关的Excel帮助内容都会被查找出来，如图1.4所示。

（2）从图中可以看到，与求和相关的Excel帮助信息共39条，其中用于求和的函数SUM就在第三项，单击该项，可以将SUM函数的帮助显示出来，用户可以根据内容判断是否使用该函数。，如图1.5所示。

图1.4　帮助搜索结果

图1.5　SUM函数帮助内容

第三种：使用相应的工具书。

使用相应的工具书，对函数的功能有一个笼统概念后，在使用函数时就能更清楚知道，哪些运算可以使用函数解决。

1.3.2　插入函数

使用函数有两种方法，一种是使用Excel命令借助“插入函数”对话框将函数插入指定位置，另一种是手动输入函数。第一种使用方法在使用时有相应的操作过程，如下：

【操作步骤】

（1）选中单元格，单击编辑栏上的插入函数快捷按钮 [image: ]
 ，弹出“插入函数”对话框，如图1.6所示。

（2）单击“选择类别”列表框右边的下三角按钮，在打开的列表中选择相应的类别，本例中选了“数学和三角函数”，如图1.7所示。

图1.6　“插入函数”对话框


[image: ]



图1.7　选择函数类别

（3）在“选择函数”列表框中选择相应的函数，本例中选择了“SUM”函数，如图1.8所示。

（4）单击“确定”铵钮后，弹出“函数参数”对话框，对函数的参数进行设置，如图1.9所示。

图1.8　选择相应的函数

图1.9　设置函数的参数

（5）再次单击“确定”按钮后，之前选定的单元格中出现了函数运行后的结果，此时公式为“=SUM(A3:A7,B2:B6)”，如图1.10所示。

图1.10　函数运行后的结果

1.3.3　手动输入函数

手动输入函数是使用函数的第二种方法。当需要嵌套使用函数，或者函数只是某个复杂公式中的一环时，需要对函数进行手动输入。在输入公式或函数之前，先将输入法状态调到英文状态，这样比较不容易出错，因为函数中的连接符必须使用英文状态下的符号。

手动输入函数的步骤如下：

【操作步骤】

（1）选中单元格，单击编辑栏输入框，将其激活，如图1.11所示。

注意：编辑栏输入框中有光标闪烁即为激活。

（2）输入“=”表示之后的内容为计算公式，输入函数名称“SUM”，可以看到输入时Excel会自动出现函数名称输入提示列表，以帮助用户输入较复杂的英文函数名称，如图1.12所示。

图1.11　激活编辑栏输入框

图1.12　输入函数名称

（3）在输入的函数名称之后输入一个完整的括号“（）”，如图1.13所示。

注意，这里之所以要输入完整的括号，是因为一方面如果括号不完整，不方便调出“函数参数”对话框；另一方面，在函数进行多层嵌套使用时，任何一层括号的不成对，都会使公式返回错误值，甚至使公式无法运行。

（4）单击编辑栏上的插入函数快捷按钮 [image: ]
 ，可以调出“函数参数”对话框，对函数的参数进行设置，也可以直接输入函数参数，本例中使用前一种方法，如图1.14所示。

注意，即使是在复杂公式的输入中，只要函数的名称及括号完整，都可以使用单击函数快捷按钮 [image: ]
 的方式对函数的参数进行设置。这种方法的优点在于，每个函数的用法和要求，在单击相应的栏位时都有帮助信息出现，且不用担心由于分隔符号“，”不是英文符号而导致的公式错误。

图1.13　输入括号

图1.14　设置函数参数

（5）单击“确定”按钮后，之前选定的单元格中出现了函数运行后的结果，此时公式为“=SUM(B1:B6,C2:C6,D2:E8)”，如图1.15所示。

图1.15　函数运行结果


第2章 逻辑函数

逻辑函数虽然并不太多，但是在Excel中应用十分广泛。其中AND、OR和NOT函数分别代表的是常见的逻辑运算：求并集、求交集和取反。灵活应用这些逻辑函数可以处理许多条件问题。而IF函数则是Excel中使用最广泛的函数之一，在后面的章节中，用户会发现IF函数经常和其他函数一起结合起来使用。本章将详细介绍逻辑函数的内容。

2.1　函数应用实战

2.1.1　FALSE和TRUE函数：返回逻辑值

这两个函数的功能是返回逻辑值FALSE和TRUE，都没有参数。

FALSE函数的表达式是FALSE()，

TRUE函数的表达式是TRUE()。

【实际应用】

用户可以使用这两个函数来直接返回逻辑值，也可以使用公式来返回逻辑值，如图2.1所示。

图2.1　使用TRUE和FALSE 函数

上面的公式“=IF(FALSE(),1,2)”中，首先FALSE函数返回逻辑值FALSE，然后用IF函数取得数值2。

2.1.2　AND函数：进行交集运算

AND函数的功能是对多个逻辑值进行交集运算，函数的返回值是逻辑值。当所有参数的逻辑值为真时，返回TRUE；只要有一个参数的逻辑值为假，即返回FALSE。

AND函数的表达式是AND(logical1,logical2, ...)。

参数logical1、logical2……表示待检测的1~30个条件值，各条件值可为TRUE或FALSE。

【实际应用】

下面的例子显示的是一次考试成绩单，教师想找出三门成绩都在平均分以上的学生，结果显示为TRUE是三科成绩都超过平均分的学生，原始数据如图2.2所示。

在单元格E2中输入公式“=AND(B2>=AVERAGE($B$2:$B$11),C2>=AVERAGE

($C$2:$C$11),D2>=AVERAGE($D$2:$D$11))”，首先通过三个公式来获得三个逻辑值，然后使用AND函数得到最后的结果。最后利用自动填充功能，得到其他单元格的数据，如图2.3所示。      

图2.2　原始数据

图2.3　最后的结果

2.1.3　IF函数：判断真假

IF函数的功能是执行真假值判断，根据逻辑计算的真假值，返回不同结果。

IF函数的表达式是IF(logical_test,value_if_true,value_if_false)。

参数logical_test表示计算结果为TRUE或FALSE的任意值或表达式；value_if_true表示logical_test为TRUE时返回的值；value_if_false表示logical_test为FALSE时返回的值。

【实际应用】

IF函数有广泛的应用，下面的实例使用IF函数返回中间的数值，原始数据如图2.4所示。

图2.4　原始数据

在单元格F2中输入公式“=IF(B2>C2,IF(C2>D2,C2,IF(B2>D2,D2,B2)),IF(B2>D2,B2,IF(C2>D2,D2,C2)))”，然后利用填充功能得到的结果如图2.5所示。

图2.5　得到结果

上面的公式中一共进行了3次IF函数的嵌套，通过逻辑判断得到相应的数值。IF函数在实际应用中最多可以使用7次嵌套。

2.1.4　NOT函数：取反

NOT函数的功能是对参数值求反。

NOT函数的表达式是NOT(logical)。

参数logical表示一个可以计算出TRUE或FALSE的逻辑值或逻辑表达式。

【实际应用】

在某个数据库输入的过程中，用户输入长度值，但是为了保证长度的值必须大于0，用户可以使用NOT函数来控制这个条件，如图2.6所示。

图2.6　接受结果

在单元格B2中输入公式“=IF(NOT(A2>0),"请输入正值",A2)”。如果A2中的数据不大于0，将会提示“请输入正值”；如果大于0，那么接受用户的输入值长度。

2.1.5　OR函数：进行并集运算

OR函数的功能是对多个逻辑值进行并集运算。在其参数组中，任何一个参数逻辑值为TRUE，即返回TRUE。

OR函数的表达式是OR(logical1,logical2,...)。

参数logical1、logical2……表示需要进行检验的1~30个条件，逻辑结果为TRUE或FALSE。

【实际应用】

下面选择和AND函数相同的例子，教师想找出有一门成绩在平均分以上的学生，结果显示为TRUE，如图2.7所示。

图2.7　最后的结果

在单元格E2中输入公式“=OR(B2>=AVERAGE($B$2:$B$11),C2>=AVERAGE($C$2:$C$11),D2>=AVERAGE($D$2:$D$11))”，选择某个学生的成绩当中是否存在大于平均值的学科。然后利用自动填充功能可以得到最后的结果。

2.2　综合实例

逻辑函数在各个领域中都有广泛的应用。本节中将运用逻辑函数处理一些数值问题。学习这些案例可以使用户更加清楚地了解逻辑函数的用法。

2.2.1　函数求值

本例主要介绍函数求值的实际案例的问题。在实际运用中，函数的应用可能会是多种形式。有时函数的形式比较简单，可以直接用前面讲解的数学函数来求解。但是有时比较复杂，是间断函数或者其他形式。这个例子讲解的就是这种情况，函数的求解规则如图2.8所示。

用户在本例中需要根据上面的条件，求解各种不同的参数条件下的函数结果。其中不同的参数条件如图2.9所示。

图2.8　求解函数值

图2.9　参数条件

步骤1：在单元格C10中输入公式“=IF(OR(A10=0,A10=B10),0,IF(A10<=1/2*B10,A10*2,IF(A10<B10,2*(B10-A10),"")))”，计算函数结果，如图2.10所示。

分析上面的公式：上面的公式就是根据前面的函数求解条件列出，首先利用OR(A10=0,A10=B10)分析了两种情况：“P=0”和“P=T”，这两种情况的函数结果都是0。因此，满足上面两种情况中的任何一种情况，函数的结果都是0。除了这两种情况，接着考虑其他的情况。IF(A10<=1/2*B10,A10*2,IF(A10<B10,2*(B10-A10),""))，表示当A10<=1/2*B10时，结果是A10*2；当1/2*B10<A10<B10，结果是2*(B10-A10)。这个公式一共嵌套了3个IF函数，将函数的各种条件都包括了。

步骤2：完成其他单元格的内容。利用自动填充功能来计算其他单元格的内容，结果如图2.11所示。

图2.10　输入函数计算公式

图2.11　计算结果

2.2.2　判断闰年

本例用来判断某一个年份是否是闰年。根据闰年的定义，年数能被4整除而不能被100整除，或者能被400整除的年份是闰年。本例将根据这个条件，来判断这个年份是否是闰年，如图2.12所示。

步骤1：在单元格B1中输入公式“=IF(OR(AND(MOD(A2,4)=0,MOD(A2,100)<>0),MOD

图2.12　判断是否是闰年

图2.13　分析年份是否是闰年

步骤2：完成其他单元格的内容。利用Excel的自动填充功能来计算其他单元格的内容，结果如图2.14所示。

图2.14　计算结果


第3章 数学、三角和数组函数

用户在使用Excel的时候，经常需要进行各种数学运算。为了提高数学运算的效率，Excel提供了丰富的数学、三角和数组函数。本章将介绍数学、三角运算和数组的常用函数，将对它们的功能、表达式及参数进行详细的讲述，然后对每个函数都举出了对应的案例。在本章的最后将举出几个综合的案例，使读者进一步理解常见函数在实际工作中的应用。

3.1　数学函数应用实战

数学函数是主要运用于数学计算的各种公式。在实际生活中，用户如果能够灵活地应用这些数学函数将会解决许多复杂的问题。一般情况下，用户在解决某个具体问题时需要使用多个数学函数。

3.1.1　ABS函数：计算绝对值

ABS函数的功能是计算一个数值的绝对值。只有一个参数，表示需要计算绝对值的实数。

ABS函数的表达式为ABS(number)。

【实际应用】

在如图3.1所示的表格中比较了某月的1~10号中的武汉和北京两个城市的日平均温度，现在用户需要判断在具体的日期中哪个城市的温度较高，而且需要计算两个温度的差值，具体数据如图3.1所示。


[image: ]



图3.1　温度的数据

步骤1：判断温度较高的城市。在单元格D2中输入公式“=IF(B2>C2,"武汉","北京")”，这个公式是判断哪个城市的温度较高，将会在D列显示温度较高的城市，如图3.2所示。

图3.2　判断温度较高的城市

步骤2：计算温度相差的数值。在单元格E2中输入公式“=ABS(B2-C2)”，计算在１日中两个城市的温度差值，如图3.3所示。


[image: ]



图3.3　计算温度的差值

步骤3：计算其他日期的温度差值。利用Excel的自动填充功能来判断其他日期的温度较高的城市和温度相差值，如图3.4所示。

图3.4　最后的计算结果

3.1.2　CEILING函数：按条件向上舍入

CEILING函数的功能是对某个数值按照条件进行向上舍入计算。

CEILING函数的表达式是CEILING(number,significance)。

上面的两个参数中，number表示需要计算的数值，significance表示需要进行舍入的乘数，也就是舍入基准。

【实际应用】

某电信运营商按照下面的标准对其国际长途进行收费：收费标准以每5秒为一个计时单位，不足5秒仍以一个计时单位计算，每个计算单位价格为0.2元。具体通话的数据如图3.5所示。

步骤1：计算通话的秒数。在单元格C2中输入公式“=MINUTE(B2)*60+SECOND(B2)”，得到第一次通话的总秒数，然后利用Excel中的自动填充功能来计算其他通话的总秒数，如图3.6所示。

图3.5　通话的数据

图3.6　计算通话的总秒数

步骤2：计算通话的费用。在这个步骤中需要使用CEILING函数对不满一个计时单元的时间向上舍入，在单元格E2中输入公式“=CEILING(C2/5,1)*D2”，得到第一次通话费用，然后利用Excel中的自动填充功能来计算其他通话的总秒数，如图3.7所示。

【注意事项】

初次使用CEILING函数的用户容易将CEILING函数和数学里面的四舍五入联系起来。但是，实际上CEILING函数并不同于四舍五入，它是按照远离0的方向向上舍入。下面举了一些具体的例子，如图3.8所示。

图3.7　计算通话的费用

图3.8　CEILING函数的应用

通过上面的表格可以看到函数CEILING的计算方法，除了了解上面的计算规则以外，用户还十分有必要注意一些常见的错误，如图3.9所示。

图3.9　CEILING函数常见的错误

如果有的参数为非数值型，那么函数会返回错误值#VALUE。如果两个参数的符号不一致，那么函数会返回错误值#NUM!。

3.1.3　COMBIN函数：计算组合数

COMBIN函数的功能是计算从给定数目的对象集合中提取若干对象的组合数。

COMBIN函数的表达式是COMBIN（number，number_chosen）。

它有两个参数，其中number表示对象总数量，number_chosen表示每一组合对象的数目。

【实际应用】

某个海岛上信号站的值班员总用红、黄，白三色各三面旗向附近海域出示旗语，在旗杆上纵排挂，可以是一面，二面，三面。现在需要判断的是一共有多少种旗语。

这是一个典型的数字排列组合的问题，可以这样分析，按照使用不同的旗数来判断不同的旗语，使用一面旗的旗语有C31种，使用两面旗的旗语有C32P22+C31种，使用三面旗的旗语有P33+C32P31P22+C31种，下面使用Excel进行组合数的计算，如图3.10所示。

步骤1：计算使用一面旗的旗语个数。在单元格D2中输入“=COMBIN(3,1)”，得到使用一面旗的旗语个数。

步骤2：计算使用两面旗的旗语个数。在单元格D3中输入“=COMBIN(3,1)+COMBIN(3,2)*COMBIN(2,2)*COMBIN(2,1)”，得到使用两面旗的旗语个数。

步骤3：计算使用三面旗的旗语个数。在单元格D4中输入“=FACT(3)+COMBIN(3,2)*COMBIN(3,1)*FACT(2)+COMBIN(3,1)”，得到使用三面旗的旗语个数，其中的FACT是计算阶乘的函数，将在后面详细介绍。

步骤4：计算总共的旗语个数。在单元格D5中输入公式“=SUM(D2:D4)”，计算总共可能有的旗语数。

【注意事项】

用户在使用COMBIN函数的时候，同样需要注意参数类型的选择，如图3.11所示。

图3.10　计算组合数

图3.11　COMBIN函数的注意事项

根据排列组合的数学含义，因此两个参数中不能有小于0的数值。同样的，在该函数中不能出现非数值类的参数。

3.1.4　FACT函数：计算阶乘

FACT函数的功能是计算给定正数值的阶乘。

FACT函数的表达式是FACT(number)。

其中的参数number表示需要计算阶乘的数字。

【实际应用】

下面简要介绍一下FACT函数的使用方法，如图3.12所示。

图3.12　使用FACT函数

【注意事项】

FACT函数的参数number必须是非负数，如果参数number是负数，函数将会返回#NUM！。由于计算FACT函数的时候，number越大，计算结果就会越大。目前Excel只能支持170以下的数的阶乘计算，超过170后，FACT函数将会返回错误值#NUM！。

3.1.5　FLOOR函数：按条件向下舍入

FLOOR函数的功能是对某个数值按条件进行向下舍入计算。

FLOOR函数的表达式是FLOOR(number，significance)。

这个函数有两个参数，其中number表示的是进行计算的数值，significance表示需要进行舍入的乘数。

【实际应用】

某公司的财会部门有如下的计算和统计规则：对所有的计算结果取两位有效数字，两位有效数字是指从第一个不是零的数字起，取两位数字，销售基本数据如图3.13所示。

图3.13　销售基本数据

上面的财会规则中并不是将有关的数据进行四舍五入，而是选取有效的位数固定为2，因此在单元格D2中输入公式“=FLOOR(C2,SIGN(C2)*10^(INT(LOG(ABS(C2)))-1))”（INT函数在下一小节介绍），然后根据Excel的自动填充功能，得出其他单元格的会计数据，如图3.14所示。

图3.14　计算相关的会计数据

在公式“=FLOOR(C2,SIGN(C2)*10^(INT(LOG(ABS(C2)))-1))”中，SIGN返回数字的符号。当数字为正数时返回1，为零时返回0，为负数时返回-1。INT(LOG(ABS(C2)))-1)计算的是数字不为零的位数，通过FLOOR可以取得相应的最接近的倍数。

【注意事项】

FLOOR函数的参数都是数值类型的。如果出现非数值型的参数，Excel会返回相应的错误。如果两个参数的符号不同，同样会返回错误提示，如同3.15所示。

图3.15　FLOOR函数的注意事项

3.1.6　INT函数：向下取整

INT函数的功能是将数字向下舍入到最接近的整数。

INT函数的表达式是INT(number)。

这个函数只有一个参数，表示需要取整的数。

【实际应用】

某顾客购物，预备了5000元，并且根据商家提供的信息记录了一些物品的价格，现在分析分别购买各种物品的最大数量，基本数据如图3.16所示。

在单元格C2中输入公式“=INT(5000/B2)”，得到商品A最多购买的数量，然后根据Excel的自动填充功能，得出其他商品的最大购买数量，最后得到的结果如图3.17所示。

图3.16　基本数据资料

图3.17　计算结果

3.1.7　LN、LOG和LOG10函数：计算对数

三个函数的功能都是计算对数。它们均有参数number，表示将要计算对数的数。LOG函数还有一个参数base，表示对数的底数。

LN函数的表达式是LN(number)，返回以常数项e为底的自然对数。

LOG函数的表达式是LOG(number,base)，该函数按所指定的底数，返回一个数的对数。

LOG10函数的表达式是LOG10(number)，返回以10为底的对数。

【实际应用】

已知三家上市公司2001年12月到2002年6月的股票月末收盘价格，现在希望计算每只股票的月收益率，以连续复利计。原始的股票价格如图3.18所示。

由于月收益率是以连续复利计，因此计算收益率必须使用LN函数，在单元格B13中输入公式“=LN(B4/B3)”，然后使用Excel的自动填充功能计算其他月份其他股票的月收益率，如图3.19所示。

图3.18　原始股票价格


[image: ]



图3.19　计算每只股票的月收益率

【注意事项】

对数计算只能对正数计算，如果在运算过程中出现了对负数进行对数运算，Excel数会返回错误#NUM!，如图3.20所示。

图3.20　参数为负数时会出现错误

3.1.8　MOD函数：返回相除的余数

MOD函数的功能是计算两个数相除后的余数。

MOD函数的表达式是MOD(number,divisor)。

它只有两个参数，number表示被除数，divisor表示除数。

【实际应用】

用户可以使用MOD函数来验证一个数是否为素数。根据素数的定义，除了1和它本身，与任何数相除都不能整除的正数是素数。现在通过计算4和7能否被比它们小的正数整除，来判断它们是不是素数。具体的操作步骤如图3.21所示。

图3.21　使用MOD函数

步骤1：判断单元格B6中的数据能否被单元格B7中的数据整除。在单元格B8中输入公式“=IF(MOD($B$6,B7)=0,"能","不能")”，就可以判断单元格B6能够被单元格B7中的数值整除。

步骤2：判断单元格B6中的数据能否被其他单元格的数据整除。利用Excel的自动填充功能完成其他单元格的公式填充，分别计算能否被整除。

步骤3：判断是否是素数。在单元格B9中输入公式“=IF(AND(C8=D8,D8=E8,E8=F8,F8=G8),"是","不是")”，来判断其中除数的结果是否相同。如果都是“不能”，那么就是素数，如果有“能”，那么被除数就不是素数。

【注意事项】

MOD函数中的返回余数的数值符号和除数的符号一致，在使用MOD函数过程中用户需要注意到这点，如图3.22所示。

图3.22　MOD函数的注意事项

3.1.9　PRODUCT函数：计算乘积

PRODUCT函数的功能是将所有以参数形式给出的数字相乘，并返回乘积值。

PRODUCT函数的表达式是PRODUCT(number1,number2,...)。

【实际应用】

下面的例子简要地介绍了PRODUCT函数的使用方法，如图3.23所示。

图3.23　使用PRODUCT函数

3.1.10　RAND函数：返回随机数

RAND函数的功能是返回大于等于0及小于1的均匀分布随机数。每次计算工作表时都将返回一个新的数值。

RAND函数的表达式是RAND()。

【实际应用】

在某次投筛子游戏中，需要随机返回1~6之间的数字，用户可以在Excel中模拟这个游戏，如图3.24所示。

图3.24　模拟投掷筛子

步骤1：产生第一个随机数。在单元格B4中输入公式“=RAND()*($D$1-$B$1)+$B$1”，在单元格B4中产生一个1~6之间的随机数。

步骤2：利用自动填充功能，完成单元格B5到单元格B13的公式，得到连续10次随机投掷筛子的结果。

上面的结果中有较多的重复值，例如数字4就重复了4次，如果用户需要返回的是不重复的随机值，需要使用其他的方法，这些内容将在后面的综合案例中介绍。

3.1.11　ROMAN函数：返回罗马数字

ROMAN函数的功能是将阿拉伯数字转换为文本形式的罗马数字。

ROMAN 函数的表示式是ROMAN(number,form)。

其中number为需要转换的阿拉伯数字；form为一数字，指定所需的罗马数字类型。罗马数字的样式范围可以从经典到简化，随着form值的增加趋于简单。

【注意事项】

下面简单介绍一下ROMAN函数的应用，如图3.25所示。

图3.25　ROMAN 函数的应用

在函数ROMAN中，如果数字为负，则返回错误值#VALUE!；如果数字大于3999，则返回错误值#VALUE!。如果其中参数form输入的是逻辑值TRUE时，返回的罗马数字是经典类型；当form输入的是逻辑值FALSE时，返回的罗马数字是简化类型。

3.1.12　ROUND、ROUNDDOWN和ROUNDUP函数：按位数舍入

三个函数的功能是返回某个数字按指定位数取整后的数字。

ROUND函数的表达式是ROUND(number,num_digits)。返回某个数字按指定位数舍入后的数字。

ROUNDDOWN函数的表达式是ROUNDDOWN(number,num_digits)。靠近零值，向下（绝对值减小的方向）舍入数字。

ROUNDUP函数的表达式是ROUNDUP(number,num_digits)。远离零值，向上（绝对值增大的方向）舍入数字。

它们都有两个参数，其中number表示需要进行四舍五入的数字；num_digits表示指定的位数，按此位数进行四舍五入。

【实际应用】

某出租车的计费标准为当历程数没有达到下一个计费单位时，按照上一个计费单位收费。假设出租车的计费标准为：从8元开始打表，前10公里每一公里跳一次表，以后每半公里跳一次表，每跳一次表要加收1.5元钱，现在需要针对不同的公里数计算费用，原始数据如图3.26所示。

图3.26　原始数据

步骤1：计算第一个公里数的费用。在单元格B3中输入公式“=IF(B2<=10,8+ROUNDDOWN(B2,0)*1.5,23+ROUNDDOWN((B2-10)*2,0)*1.5)”，这个公式首先判断公里数是否大于10，如果不大于10，那么按照跳表次数和1.5元的乘积加上打表基数8元就是费用；如果大于10，那么由于每半个小时跳表，所以需要将结果乘以2，然后加上基数23元就是费用。

步骤2：计算其他公里数的费用。利用自动填充功能将B3中公式填充到其他单元格，得到的结果如图3.27所示。

图3.27　计算费用

3.1.13　SUM函数：计算数字之和

SUM函数的功能是计算某一单元格区域中所有数字之和。

SUM函数的表达式是SUM (number1,number2,...)。

【实际应用】

如图3.28所示的表格中的数据是2004年某市部分毕业生在中招考试中的成绩，要求统计达到本市一中录取分数线的学生人数。

在单元格C10中输入公式“=SUM(IF(C2:C9>=C11,1,0))”，然后同时按下Shift+Ctrl+Enter组合键，输入数组公式，就可以计算出过线人数。

注意这里的数组公式，如果输入的是一般公式，Excel会返回错误。

3.1.14　SUMIF函数：条件求和

SUMIF函数的功能是根据指定条件对若干单元格求和。

SUMIF的表达式是SUMIF(range,criteria,sum_range)。

其中参数range表示用于条件判断的单元格区域；参数criteria是表示确定哪些单元格将被相加求和的条件，其形式可以为数字、表达式或文本；参数sum_range是需要求和的实际单元格。

【实际应用】

某公司需要统计销售额在10000元以上的销售点的销售人员的数量，各个销售点的销售额和销售人员的具体数据如图3.29所示。

图3.28　计算过线人数

图3.29　统计销售人员的总数

在单元格D9中输入公式“=SUMIF(B2:B7,">10000",C2:C7)”，这个公式首先根据条件区域来判断销售额“>10000”，然后将对应的区域C2：C7中的销售人员累加起来，得到题目中需要的销售人员数量。

3.1.15　TRUNC函数：按位数舍入

TRUNC函数的功能是将数字的小数部分截去，返回整数。

TRUNC函数的表达式是TRUNC(number,num_digits)。

其中参数number为需要截尾取整的数字；参数num_digits表示用于指定取整精度位数的数字。num_digits的默认值为0。

【实际应用】

下面简单列出了函数TRUNC的用法，如图3.30所示。

图3.30　使用TRUNC函数

3.2　三角函数应用实战

三角函数主要适用于数学几何的运算。本节将主要介绍8个常用的三角函数。

3.2.1　正三角函数SIN、COS和TAN

这三个函数的功能是计算给定角度的正弦、余弦和正切值。

SIN函数的表达式为SIN(number)。

COS函数的表达式为COS(number)。

TAN函数的表达式为TAN(number)。

【实际应用】

下面内容简要介绍了SIN、COS和TAN三角函数的应用，如图3.31所示。

图3.31　三角函数的应用

在单元格B2中输入公式“=A2*PI()/180”，然后利用自动填充功能，将A列的角度值转换为弧度值，然后在后面的三列中分别计算角度的正弦、余弦和正切值。

3.2.2　反三角函数ASIN、ACOS和ATAN

这三个函数的功能是计算给定正弦、余弦和正切值的角度。

ASIN函数的表达式为ASIN(number)。

ACOS函数的表达式为ACOS(number)。

ATAN函数的表达式为ATAN(number)。

【实际应用】

下面将简要介绍ASIN、ACOS和ATAN函数的应用，如图3.32所示。

图3.32　反三角函数的应用

在上面的单元格中首先使用公式“=RAND()*2-1”计算出某角度的正弦和余弦值，由于SIN和COS 的值介于-1~1之间，因此公式“=RAND()*2-1”返回的也是-1~1之间的值。

3.2.3　DEGREES和RADIANS函数：转换度和弧度

函数DEGREES的功能是将弧度转换为角度，函数RADIANS的功能是将角度转换为弧度。

DEGREES函数的表达式为DEGREES(angle)。

RADIANS函数的表达式为RADIANS(angle)。

【实际应用】

上面的三角函数和反三角函数中都出现过角度和弧度的互换，这些内容在使用三角函数过程中经常会使用到，下面简要总结互换的公式，如图3.33所示。

图3.33　角度和弧度互换

3.3　数组函数应用实战

Excel中还包含有很多的数组函数。和其他函数不同，这些函数的参数都是数组类型的，本小节将介绍几个常见的数组函数。

3.3.1　MDETERM函数：计算行列式

MDETERM函数的功能是返回一个数组的矩阵行列式的值。

MDETERM函数的表达式是MDETERM(array)。

其中参数array是行数和列数相等的数值数组。

【实际应用】

MDETERM函数可以进行矩阵运算，判断线性方程是否有解，以及判断系数矩阵是否可逆，如图3.34所示。

图3.34　判断是否可逆

在单元格C5中输入公式“=IF(MDETERM(A1:C3)<>0,"可逆","不可逆")”。该公式中首先使用MDETERM函数计算矩阵行列式的值，如果行列式的值不等于0，那么该矩阵可逆，如果行列式的值等于0，那么该矩阵不可逆。

【注意事项】

array可以是单元格区域，例如A1:C3；或是一个数组常量，如{1,2,3;4,5,6;7,8,9}；或是区域或数组常量的名称。如果array中单元格是空白的或包含文字，则函数MDETERM返回错误值#VALUE!；如果array的行和列的数目不相等，则函数MDETERM也返回错误值#VALUE!。

3.3.2　MINVERSE函数：返回逆矩阵

MINVERSE函数的功能是返回数组矩阵的逆距阵。

MINVERSE函数的表达式是MINVERSE(array)。

【实际应用】

沿用上面的例子，上面的矩阵经过判断，是可逆矩阵。因此用户可以求出其相应的逆矩阵，如图3.35所示。

图3.35　求矩阵的逆矩阵

在单元格A5中输入公式“=MINVERSE(A1:C3)”，然后选择A5:C7单元格区域，再同时按下Shift+Ctrl+Enter组合键，得到计算出来的逆矩阵。

如果用户只是在单元格A5中输入公式，那么得到的结果就是0.0545455，而不是整个逆矩阵。

3.3.3　MMULT函数：计算矩阵乘积

函数MMULT的功能是返回两数组的矩阵乘积。

MMULT函数的表达式是MMULT(array1,array2)。

其中array1, array2是要进行矩阵乘法运算的两个数组，函数的结果矩阵的行数与array1的行数相同，结果矩阵的列数与array2的列数相同。

【实际应用】

在上面的小节中用户已经求出原来矩阵的逆矩阵，现在用户可以将这两个函数相乘，看看得到的结果是否是单位矩阵，如图3.36所示。

图3.36　两个矩阵相乘

【注意事项】

array1的行数必须与array2的列数相同，而且两个数组中都只能包含数值；array1和array2可以是单元格区域、数组常量或引用；如果单元格是空白单元格或含有文本字符串，或是array1的行数与array2的列数不相等时，则函数MMULT返回错误值#VALUE!。

3.3.4　SUMPRODUCT函数：计算数组求和

函数SUMPRODUCT的功能是在给定的几组数组中，将数组间对应的元素相乘，并返回乘积之和。

SUMPRODUCT的表达式是SUMPRODUCT(array1,array2,array3, ...)。

【实际应用】

假设到月底某公司需要计算单位销售人员的销售成绩，销售人员的销售情况及各个组件的单价如图3.37所示。

图3.37　计算销售成绩

步骤1：在单元格G3中输入公式“=SUMPRODUCT(B3:F3,$B$9:$F$9)”，将两个数组的元素对应相乘，然后将乘积相加，得到第一个销售人员的销售成绩。

步骤2：利用自动填充功能得到其他销售人员的销售成绩。

3.4　综合实例

在这个部分利用两个综合实例来介绍如何使用数学和三角函数来处理日常生活中的一些常见问题。

3.4.1　计算原材料的用量

某公司需要统计生产产品AB所使用的各个原材料用量，这样公司可以方便地了解生产产品所需要耗费的原材料的量。生产产品AB所使用的物料清单BOM表如图3.38所示。其中A列中用彩色底纹标出的数据就是生产产品的原材料。

图3.38　物料清单

根据上面的已知条件，下面分步骤介绍如何计算各个原材料的用量。

步骤1：分析计算方法。首先以A0001构件为例来分析如何计算原材料的用量。A0001的上级构件是A1，而A1的上级构件是A。生产1个单位产品AB需要5个单位的构件A，而每一个单位的构件A包含了2个单位的构件A1，最后每一个单位构件A1包含7个单位的A0001构件。因此，生产1个单位的产品AB需要的A0001构件的个数为5*2*7=70。其他原材料的个数也是使用相同的统计方法来计算。

步骤2：在单元格D7中输入公式“=IF(COUNTIF(B$5:B$21,A7),,PRODUCT(IF(ISNUMBER(MATCH($A$5:$A7&"*",B7,)),$C$5:$C7))*C7)”，然后同时按下Shift+Ctrl+Enter组合键，求出结果，如图3.39所示。

图3.39　输入公式

分析上面的公式。在本公式中使用了一些查找函数，这些函数在后面章节将会详细介绍。首先，通过IF函数中的COUNTIF(B$5:B$21,A7)公式来判断单元格A7中的构件是否是其他构件的上级构件。如果是某个构件的上级构件，那么该单元格中的构件就不是原材料，直接返回0。如果是原材料，那么使用后面的PRODUCT(IF(ISNUMBER(MATCH($A$5:$A7&"*",B7,)),$C$5:$C7))*C7计算原材料的个数。其中MATCH($A$5:$A7&"*",B7,)用来统计在A列中包含有B7单元格中的构件的单元格数字。在这个例子中，单元格B7中的数值是A1，则A5：A7单元格中，符合条件的就是单元格A5和A6中的A、A1构件，而符合条件的数值就是该构件的上级构件，。然后使用IF(ISNUMBER(MATCH($A$5:$A7&"*",B7,)),$C$5:$C7)函数返回对应的C列的构件用量，在本例中返回的是5和2，然后使用PRODUCT函数将两者相乘，最后和C7 单元格中的数值相乘，也就是5*2*7=70。

步骤3：填充其他单元格。通过上面的公式分析，可以看到，整个思路就是步骤1中分析得到的计算方法。利用自动填充功能，可以计算其他单元格中的构件用量，如图3.40所示。

图3.40　填充其他单元格

3.4.2　随机分配

某个工地中有31人，现在有工资额总共为350000元，需要分配给这31人。但是工地上的班长将31人原始出工的花名单遗失了，为了能够较为公平地分配这个工资额，工地决定使用下面的分配方法：随机分配工资但是必须满足各个民工所得到的工资在整体工资的平均值±30%内变化，并且350000元的工资总额不能有剩余。

根据已知条件，本例需要解决的问题其实类似于设定某种方法，来保证下面几个条件：随机分配，总和固定，同时每个数值不能超过某个限制。只要满足上面条件的分配方法都可以选取，因此，最后的分配方法可能不唯一。下面给出其中一种。

步骤1：计算平均工资额。在单元格H5中输入公式“==$B$35/$H$4”，其中单元格B35中保存的是工资总额350000元，单元格H4中保存的是人数31。

步骤2：分配第一个人的工资额。在单元格B4中输入公式“=ROUND($H$5*(0.7+0.6*RAND()),2)”，得到第一个人分配的工资额，其中0.7=1-0.3，；0.6=2*0.3。(0.7+0.6*RAND())的取值范围是（0.7,1.3）。因此，该人分配的工资额在平均工资额的0.7~1.3之间，没有超过±30%。同时使用RAND函数，保证了随机性，因此满足条件，如图3.41所示。

步骤3：计算第一个人工资的差额。为了能够保证每个人的工资额都在工资的平均值的±30%内变化，用户可以设置“验算查值”一列，计算每个人工资的查值比例，在单元格C4中输入公式“=1-B4/$H$5”，如图3.42所示。


[image: ]



图 3.41　分配第一个人的工资额

图3.42　计算工资的差额 

步骤4：计算第二个人的工资额。在单元格B5中输入公式“=ROUND(MAX($H$5*0.7,$B$35-SUM($B$4:$B4)-(31-ROW(2:2))*$H$5*1.3)+(MIN($H$5*1.3,$B$35-SUM($B$4:$B4)-(31-ROW(2:2))*$H$5*0.7)-MAX($H$5*0.7,$B$35-SUM($B$4:$B4)-(31-ROW(2:2))*$H$5*1.3))*RAND(),2)”，如图3.43所示。

图3.43　计算第二个人的工资额

分析上面的公式：上面的公式比较复杂，现在简要分析一下上面公式的含义。公式MAX($H$5*0.7,$B$35-SUM($B$4:$B4)-(31-ROW(2:2))*$H$5*1.3)，单元格H5中的数值是工资的平均值，单元格B35中的数值是工资总额，$B$35-SUM($B$4:$B4)-(31-ROW(2:2))*$H$5*1.3，数值小于$H$5*1.3，同时MAX函数保证了数值大于$H$5*0.7。因此，MAX函数的数值范围是$H$5*(0.7,1.3)。同样的可以分析另外一个公式，函数MIN($H$5*1.3,$B$35-SUM($B$4:$B4)-(31-ROW(2:2))*$H$5*0.7)，数值的范围也是$H$5*(0.7,1.3)。而整个公式“MAX($H$5*0.7,$B$35-SUM($B$4:$B4)-(31-ROW(2:2))*$H$5*1.3)+(MIN($H$5*1.3,$B$35-SUM($B$4:$B4)-(31-ROW(2:2))*$H$5*0.7)-MAX($H$5*0.7,$B$35-SUM($B$4:$B4)-(31-ROW(2:2))*$H$5*1.3))*RAND()”保证了数值范围，同时使用了RAND函数保证了随机性，因此满足分配方法。

步骤5：计算第二个人的工资差额。在单元格C5中输入公式“=1-B5/$H$5”，计算出第二人的工资差额，如图3.44所示

步骤6：计算其他人的工资。利用自动填充功能，将单元格B5中的公式拖动到B列所有的单元格中，计算其他人的工资额，如图3.45所示。

图3.44　计算第二个人的工资差额

图3.45　计算其他人的工资 

步骤7：计算其他人的工资差额。利用Excel的自动填充功能，将单元格C5中的公式拖动到C列所有的单元格中，计算其他人的工资额，如图3.46所示。

步骤8：检测上面随机分配的结果。用户需要检测分配工资总额是否为350000元，同时还要检测差值是否在平均值±30%变动，在单元格H7中输入公式“=SUM(B4：B34)”，如图3.47所示。

图3.46　计算其他人的工资差额

图3.47　检测上面分配结果 

步骤9：检测随机性。用户可以按下F9键，更新随机分配的工资额，如图3.48所示。

图3.48　检测随机性


第4章 统计函数

统计函数是用于对数据区域进行统计分析的函数。本章介绍的统计函数分为一般统计函数和数理统计函数。一般的统计函数用来统计给定的某个区域的数据的平均值、最大值或者最小值等。数理统计函数则是对数据进行相关的概率分布的统计，进行回归分析等。

本章将着重介绍统计函数，对它们的功能、表达式及参数进行详尽的讲述，然后对每个函数的应用举例说明。

4.1　函数应用实战

4.1.1　AVERAGE和AVERAGEA函数：计算平均值

AVERAGE函数的功能是计算所有包含数值单元格的平均值，而AVERAGEA则是计算所有非空单元格的平均值，两者的主要区别在于如何对待非数值类的单元格。AVERAGE函数和AVERAGEA函数的表达式分别为：

AVERAGE(value1,value2,...)。

AVERAGEA(number1,number2,...)。

两者的参数都是可选的，表示需要计算平均值的参数，而且个数在1~30之间。

【实际应用】

某公司在2005年的上半年有一个简单的进货清单，如图4.1所示。其中有两个月的进货量为零，需要分别用到上面的函数来计算两个不同的平均值。

在单元格E3中输入函数表达式=AVERAGEA(B2:B7)，可以计算上半年的进货量，函数AVERAGEA自动认为没有进货的两个月份进货量是0。

在单元格E5中输入函数表达式=AVERAGE(B2:B7)，可以计算上半年进货月份的平均进货量，没有进货的月份，AVERAGE将自动忽略。

图4.1　计算结果

【注意事项】

通过上面的例子，可以看出两个计算平均值函数的区别。对有可能不是数值的单元格AVERAGE函数将自动忽略，不参与计算；AVERAGEA函数将其视为0参与计算。

4.1.2　COUNT和COUNTA函数：统计单元格个数

COUNT函数的功能是统计参数列表中，含有数值数据的单元格个数，而COUNTA则是统计参数列表中所有非空值的单元格个数，两者的主要区别同样在于如何对待非数值类的单元格。COUNT函数表达式和COUNTA函数表达式分别为：

COUNT(value1,value2,....)。

COUNTA(number1,number2,...)。

两者的参数都是可选的，参数个数在1~30之间，表示需要统计的值。

【实际应用】

某公司在某季度，对销售员工进行销售量的统计，基本数据如图4.2所示。需要统计不同的指标对员工进行评估，因此需要同时使用上面的两个函数，同时可以将其进行对比。

在单元格D7中输入=COUNT(B2:B7)，在单元格D8中输入=COUNTA(B2:B7)。然后分别通过拖动单元格来完成单元格的自动填充，得到的统计结果见图4.2。

图4.2　计数结果

上面单元格中画出的阴影部分表示的是有效统计量。在E列中，用户需要知道的是已经定价的销售货物的个数，因此用COUNT统计有数值的单元格个数，而由于F列中“收款情况”不是按照数值来表达的，因此使用COUNTA统计的数值是有效的。

【注意事项】

上面的例子可以看到，不同的统计量用户可以选用不同的统计函数，这样可以达到相应的统计结果。对比的结果也可以提示用户，如果使用不当的话，那么统计的结果就没有效用。

4.1.3　COUNTBLANK函数：统计空白单元格

COUNTBLANK函数的功能是统计指定单元格区域中空白单元格的个数。COUNTBLANK函数的表达式为：

COUNTBLANK(range)。

参数range表示需要计算其中空白单元格个数的区域。

【实际应用】

某公司对各个部门员工的抽烟情况进行统计，作为人力资源搭配的一个数据依据。对于调查对象，只能回答Y或者N，分别表示抽烟或者不抽烟。由于某种情况，有些部门没能及时反馈信息，现在统计没有登记的部门个数，统计结果如图4.3所示。

图4.3　统计没有登记的部门个数

图4.3中单元格B2：E11中的数据就是需要统计的数值，在这个单元格区域之中，没有输入任何字母的单元格就是没有统计结束的个数，因此在单元格E13中输入公式“=COUNTBLANK(B2:E11)”，统计空白单元格的个数。

【注意事项】

使用COUNTBLANK函数需要注意的是统计区域的确定。在上面的例子中如果用户选择的公式为“=COUNTBLANK(A1:E11)”，那么得出的结果就是17。这个结果显然是不对的，因为单元格A1也是空白单元格，所以在使用COUNTBLANK函数的时候要注意选择合适的统计区域。

4.1.4　COUNTIF函数：按条件统计

COUNTIF函数的功能是统计指定单元格区域中满足条件的单元格的个数。COUNTIF函数的表达式为：

COUNTIF(range,criteria)。

参数range表示需要计算单元格个数的区域，参数criteria表示确定的条件。

【实际应用】

某车行统计了1998年一定时期的销售情况，用户需要对各个销售和服务类型进行统计。具体的统计分析情况如图4.4所示。

图4.4　统计结果

图中利用COUNTIF函数进行了各种不同条件的统计。首先根据条目的不同，分别统计了出售刹车片、轮胎的个数，下面分步骤详细介绍：

步骤1：在单元格F12中输入公式“=COUNTIF(B2:B10,"刹车片")”，统计到出售煞车片刹车片的个数为2。

步骤2：在单元格F13中输入公式“=COUNTIF(B2:B10,"轮胎")”，可以统计到出售轮胎的个数为3。

步骤3：统计开支大于100的项目个数。在单元格F14中输入公式“=COUNTIF(D2:D10,">=100")”，此时条件变成了“>=100”。得到支出高于100的项目个数为5。

步骤4：最后用户需要统计的是保修的个数，在单元格F16中输入公式“=COUNTIF(B2:B10,D16)”，得到保修的个数为2。从这个公式中可以看出，该函数的条件不仅可以是用户自己输入的条件，还可以是单元格中的内容。

4.1.5　FREQUENCY函数：计算频率分布

FREQUENCY函数的功能是统计指定单元格区域中数据的频率分布。FREQUENCY函数的表达式为：

FREQUENCY(data_array,bins_array)。

参数data_array表示用来计算频率的数组，参数bins_array表示分组的节点。

【实际应用】

一家公司统计某月的销售统计数据，由于这个销售数据是根据相关的原始数据进行计算得到的，因此数据形式是矩阵形式。在本例中命名为data，对其进行频率统计，如图4.5所示。

图4.5　频率统计

本例中的data代表A1:E20的数据矩阵，在计算频率分布的时候用到的公式为“=FREQUENCY(data,G7:G16)”，其中的G7:G16就是Bins列中的数据，表示的是频率分布的节点，因此可以得到相应的频率统计数值。

【注意事项】

使用FREQUENCY函数需要注意的是，节点之间间隔是确定的，如果用户需要对多个间隔的节点进行对比统计，那么就需要使用比较复杂的公式。

4.1.6　LARGE和SMALL函数：按条件返回最值

LARGE函数的功能是返回数据集中的第k个最大值，SMALL函数的功能是返回数据集中的第k个最小值。

LARGE函数的表达式是LARGE(array,k)。

SMALL函数的表达式是SMALL(array,k)。

参数array表示需要计算的数组和区域，参数k表示返回值在数组或数据单元格区域中的位置。

【实际应用】

某公司统计在各个不同的区域的销售统计量，需要统计最大销售额和第二大销售额，最小销售额和第二小销售额，基本的数据如图4.6所示。

分别在单元格中输入相应的统计公式，可以得到相应的统计结果，如图4.7所示。

图4.6　基本数据

图4.7　统计结果

【注意事项】

使用LARGE和SMALL函数时需要注意的是数组不能为空，而且参数k不能超过数组的范围。如果出现上面的错误，函数会返回错误值，如图4.8所示。

图4.8　常见的错误

4.1.7　MAX和MAXA函数：返回最大值

MAX函数的功能是统计所有数值数据的最大值，而MAXA的功能是统计所有非空单元格的最大值。如果在统计的数组中含有非数值的单元格，那么MAX根据相应的逻辑值来赋值，也就是说含TRUE的参数就当作是1，含FALSE的参数就是0。

MAX函数的表达式是MAX(value1,value2,....)。

MAXA函数的表达式是MAXA(number1,number2,...)。

同样两者的参数都是可选的，而且个数在1~30之间，表示需要统计的值。

【实际应用】

在某个实验的过程中，我们统计了不同实验试件的温度。在得出结果后需要知道温度最大的试件，下面分别使用MAX和MAXA函数来进行统计，看看结果会有什么不同，如图4.9所示。

图4.9　统计结果

显然，在本次实验中需要使用MAX函数统计的结果。因为其中有个试件没有测出温度，我们不应该将其进行统计。在MAXA中，将其当作是逻辑FALSE处理，在数值中就认为是0，因此统计的最高温度的结果是0。

4.1.8　MEDIAN函数：返回中值

MEDIAN函数的功能是返回给定数值集合的中值。

MEDIAN函数的表达式是MEDIAN(value1,value2,...)。

参数是可选的，而且个数在1~30之间，表示需要计算中值的数值。

【实际应用】

某超市统计了其中某个柜台的商品的价格。下面用户需要统计出这些价格的中值。为了使用户了解中值的真正含义，在后面统计了部分价格的中值，如图4.10所示。

图4.10　统计结果

图中的统计过程中，首先在单元格B11中输入公式“=MEDIAN(B2:B9)”，统计出全部商品的价格中值。在单元格B12中输入公式“=MEDIAN(B2:B8)”，统计前面7个商品的价格中值。第一个中值统计的数目是偶数，因此该函数返回的是中间两个数的平均值；第二个中值统计的数目是奇数，因此返回的就是中间的数值。

4.1.9 MIN和MINA函数：返回最小值

MIN函数的功能是统计所有数据的最小值，而MINA的功能是统计所有非空单元格的最小值。如果在统计的数组中含有非数值的单元格，那么MINA根据相应的逻辑值来赋值，也就是说含TRUE的参数就当作是1，含FALSE的参数就是0。

MIN函数的表达式是MIN(value1,value2,...)。

MINA函数的表达式是MINA(number1,number2,...)。

同样两者的参数都是可选的，而且个数在1~30之间，表示需要统计的值。

【实际应用】

根据前面提到的MAX和MAXA函数的例子，现在用户需要知道温度最小的试件。下面分别使用MIN和MINA函数来进行统计，如图4.11所示。

图4.11　统计结果

使用两种函数得到的结果是相同的，但不是说在其他地方两个函数都会得到相同的结果。本例中对于非数值单元格，函数MIN直接忽略，函数MINA则返回0，但是最后的统计结果是一样的。

4.1.10　MODE函数：返回频率最多的数值

MODE函数的功能是统计在某一个数组或是数据区域中出现频率最多的数值。使用这个函数的时候需要注意，如果统计的数组中频率是均等的，也就是所有的数据都是出现一次，那么这个函数返回的是空值。

MODE函数的表达式是MODE(value1,value2,...)。

参数是可以选择的，而且个数在1~30之间，表示需要统计的值。

【实际应用】

在某个班上期中考试的分数如图4.12所示。我们需要知道分数的众数，也就是出现次数最多的分数值是多少，这样我们直接使用MODE函数就可以了。在单元格中输入函数=MODE(B2:B11)，就可以得出次数最多的分数值了。

图4.12　统计结果

本例数值比较少，我们可以看到其中88这个分数出现的次数最多，因此其统计的结果实正确的。

4.1.11　RANK函数：返回排位

RANK函数的功能是返回某个数字在数字列表中的排位。

RANK函数的表达式是RANK(number,ref,order)。

其中number表示需要找到排位的数字，ref表示数字列表数组或对数字列表的引用，order表示排位的方式。

【实际应用】

某个公司统计在各个不同的区域的销售统计量。下面需要将各个区域的销售额进行排列，如图4.13所示。

在单元格C2中输入公式“=RANK(B2,$B$2:$B$10)”，计算出“中部001”的销售排名。用户需要注意的是，公式中的绝对引用方式，不能使用相对引用方式代替。通过使用Excel中的自动填充功能，完成其他数据列的排列。

【注意事项】

使用RANK函数进行排名的时候，用户需要注意的是，这个排名是美式的排名，对于并列的情况直接进行增位排名。如果用户需要的是直接大小排名，需要自行进行排名，使用RANK排名会出现错误。下面举例来说明如何确定有重复数值的排序问题，简单地选取几个有重复数值的数据，如图4.14所示。

图4.13　各个销售地点的排名情况

图4.14　基本数据

下面需要选取上面基本数据中第二大的数据，显然应该是2，但是如果直接用rank就得不到结果。用户需要输入公式“=LARGE(IF(FREQUENCY(A1:A5,A1:A5),A1:A5),2)”。下面分析这个公式的含义。

FREQUENCY(A1:A5,A1:A5)返回的是一个频率数组，根据FREQUENCY函数的定义，其返回的是数组(2,0,2,0,1,0）；IF(FREQUENCY(A1:A5,A1:A5),A1:A5)返回的数组是(1,FLASE,2, FLASE,5, FLASE)；LARGE(IF(FREQUENCY(A1:A5,A1:A5),A1:A5),2)公式返回上面的数组中第二大的数值，对于FLASE系统会将视为0，因此，结果会返回为2，如图4.15所示。

图4.15　分析的过程和步骤演示

4.1.12　CONFIDENCE函数：计算置信区间

CONFIDENCE函数的功能是返回一个值，可以使用该值构建总体平均值的置信区间。

CONFIDENCE函数的表达式是CONFIDENCE（alpha,standard_dev,size）。

其中alpha表示用于计算置信度的显著水平参数，对应的置信度是100*(1-alpha)%；standard_dev表示数据区域的总体标准偏差；size表示样本容量。

【实际应用】

某个样本取自70名乘车上班的旅客，他们花在路上的平均时间45分钟，总体的偏差为4.5分钟。选择alpha=0.05，如图4.16所示计算置信区间的返回值。

图4.16　计算置信区间的值

在单元格A1中输入公式“=CONFIDENCE(0.05,4.5,70)”，根据返回值1.054172，可以得到这个时间的置信区间为[45-1.054172,45+1.054172]。也就是说，在这70个旅客中，有95%的旅客在路上花费的时间在[43.95,46.05]之间。

4.1.13　CORREL函数：计算相关系数

CORREL函数的功能是单元格区域中的两个数列之间的相关系数。

CORREL函数的表达式是CORREL(array1,array2)。

其中array1和array2表示的都是数据列，CORREL函数可以用来分辨两种属性之间的关系。

【实际应用】

某个公司正在考虑是否继续投入公司的广告费用，因此需要利用上半年的统计数据来计算广告费与销售额之间的相关系数，具体的数据如图4.17所示。

在单元格B2中输入公式“=CORREL(B2:B13,C2:C13)”，用来计算B列数据和C列数据之间的相关系数，得到的数据是0.922518177，十分接近1，正相关度很高，因此可以认为广告费用和销售额基本正相关。

相关系数根据数学定义，其绝对值小于等于1。越接近于1，就可以认为相关度越高。下面再举一个负相关的例子。某厂家统计了同一车型中旧车的车龄及其售价数据,现在希望了解价格与车龄的相关系数，具体的数据表格如图4.18所示。

图4.17　计算广告费用和销售额的相关系数

图4.18　计算车龄和价格的相关系数

图中的计算结果可以看出他们的相关系数为-0.993240519，可见他们的负相关程度很大。

4.1.14　BINOMDIST函数：返回二项分布概率

BINOMDIST函数可以返回一元二项式分布的概率值，适用于固定次数的独立试验，当试验的结果只包含成功或失败两种情况，且当成功的概率在实验期间固定不变的情况。

BINOMDIST函数的表达式为BINOMDIST (number_s,trials,probability_s,cumulative)。

其中参数number_s表示试验成功的次数；trials表示独立试验的次数；probability_s表示每次试验中成功的概率。cumulative是一个逻辑值，用于确定函数的形式。如果cumulative为TRUE，函数BINOMDIST返回累积分布函数，即至多number_s次成功的概率；如果为FALSE，返回概率密度函数，即number_s次成功的概率。

【实际应用】

某个推销员进行业务推销，每次推销成功的概率是0.3，该推销员今天一共打了6个电话进行推销，需要进行成功概率的统计计算，如图4.19所示。

图4.19　计算推销成功的概率

在单元格B7中输入公式“=BINOMDIST(A7,$B$2,$B$3,0)”，计算成功的次数为0的概率；在单元格C7中输入公式“=BINOMDIST(A7,$B$2,$B$3,1)”，计算成功的次数不大于0的概率；在后面的单元格中分别通过两列的数据的差值求得响应的概率。然后使用自动填充功能来计算其他成功次数的概率，结果如图4.20所示。

图4.20　计算成功次数的概率统计表

图中的计算结果可以看出它们的相关系数为-0.993240519，可见它们的负相关程度很大。

【注意事项】

在函数BINOMDIST的参数中，number_s和trials将被截尾取整。如果number_s、trials或probability_s为非数值型，函数BINOMDIST返回错误值#VALVE!。如果number_s<0或number_s>trials，函数BINOMDIST返回错误值#NUM!。如果probability_s<0或probability_s>1，函数BINOMDIST返回错误值#NUM!。

4.1.15 CRITBINOM函数：返回累计二项分布数值

CRITBINOM函数返回使累积二项式分布大于等于临界值的最小值。此函数可以用于质量检验。CRITBINOM可以认为是BINOMDIST函数的反函数。

CRITBINOM 的表达式为CRITBINOM(trials,probability_s,alpha)。

其中参数trials表示伯努利试验次数；probability_s表示每次试验中成功的概率；alpha表示的是临界值。

【实际应用】

根据前面的例子，可以求这个问题的反问题。也就是，需要知道该推销员成功的概率小于等于某个alpha临界值时，伯努利试验中必须成功的次数，如图4.21所示。

图4.21　反求伯努利试验成功的次数

上面的表格是函数BINOMDIST中计算的概率矩阵，现在反求必须成功的次数。选取成功两次的累积概率0.74431，并在单元格D12中输入公式“=CRITBINOM(6,0.3,0.74431)”反求成功的次数，得到的结果正是2次。

4.1.16 NORMDIST和NORMINV函数：返回正态分布

NORMDIST函数返回指定平均值和标准偏差的正态分布函数。此函数在统计方面应用范围广泛。

NORMINV函数返回指定平均值和标准偏差的正态累积分布函数的反函数。

NORMDIST函数的表达式是NORMDIST(x,mean,standard_dev,cumulative)。

其中x是需要计算其分布的数值；mean是分布的算术平均值；standard_dev表示分布的标准偏差；cumulative是一逻辑值，指明函数的形式。如果cumulative为TRUE，函数NORMDIST返回正态分布函数；如果为FALSE，返回概率密度函数。

NORMINV函数的表达式为NORMINV(probability,mean,standard_dev)

其中probability是正态分布的概率值；mean是分布的算术平均值；standard_dev表示分布的标准偏差。

【实际应用】

假设某品牌的汽车电池的寿命是常态分布，平均数为800天，标准差为100天。现随机抽取一个汽车电池，其寿命小于500天的概率有多大？大于1000天的概率有多大? 介于700天至900天的概率有多大? 如果该公司想定一个保固期，在保固期限内可以免费更换电池，公司最多可以承担1%的免费更换，保固期应该定多久?

这些现实的问题，可以使用NORMDIST函数和NORMINV函数进行解决，如图4.22所示。

图4.22　计算概率

【注意事项】

在NORMDIST函数中，参数会有明显的限制，列举如下：

如果mean或standard_dev为非数值型，函数NORMDIST返回错误值#VALUE!。

如果standard_dev≤0，函数NORMDIST返回错误值#NUM!。

如果mean=0，standard_dev=1，且cumulative=TRUE，则函数NORMDIST返回标准正态分布。

如果已给定概率值，函数NORMINV使用NORMDIST(x,mean,standard_dev,TRUE)=probability求解数值x。因此，函数NORMINV的精度取决于函数NORMDIST的精度。NORMINV使用迭代搜索技术。如果搜索在100次迭代之后没有收敛，则函数返回错误值#N/A。

4.1.17　POISSON函数：返回泊松分布

POISSON函数返回泊松分布。泊松分布通常用于预测一段时间内事件发生的次数。

POISSON函数的表达式为POISSON(x,mean,cumulative)。

其中x表示事件数；mean表示期望值；cumulative是逻辑值，确定所返回的概率分布形式。如果cumulative为TRUE，函数POISSON返回泊松累积分布概率，即随机事件发生的次数在0~x之间（包含0和1）；如果为FALSE，则返回泊松概率密度函数，即随机事件发生的次数恰好为x。

【实际应用】

假设某公司订购了100件产品，每一件产品是次品的概率是0.01，该批产品验收合格的条件是次品数不能超过3件，要求这批产品的合格的概率。

可以认为次品数为随机变量X，其服从参数为100×0.01=1的泊松分布，计算如图4.23所示。

图4.23　计算概率

4.1.18　STDEV和STDEVA函数：返回标准偏差

STDEV函数的表达式为STDEV(number1,number2,...)。

STDEVA函数的表达式为STDEVA(number1,number2,...)。

两个函数的功能是计算样本的标准偏差。参数可选，个数在1~30之间。

STDEV函数是计算所有数值的标准偏差，STDEVA函数是求所有非空白单元格的标准偏差。

【实际应用】

某销售公司有8个销售人员，在某一个月内，各个销售人员完成的销售任务如图4.24所示，现在分别利用STDEV函数和STDEVA函数来计算各个销售量的标准偏差，计算结果见图。

图4.24　计算结果

【注意事项】

在函数STDEV中，假设其参数是总体中的样本。如果数据代表全部样本总体，则应该使用函数STDEVA来计算标准偏差。此处标准偏差的计算使用“无偏差”或“n-1”方法。

4.1.19 VAR和VARA函数：返回样本的方差

VAR函数的表达式为VAR(number1,number2,...)。

VARA函数的表达式为VARA(number1,number2,...)。

两个函数的功能是计算样本的方差。参数可选，个数在1~30之间。

VAR函数是计算所有数值的样本方差，VARA函数是求所有非空白单元格的样本方差。

【实际应用】

某个销售公司有8个销售人员，在某一个月内，各个销售人员完成的销售任务如图4.25所示如，现在分别利用VAR函数和VARA函数来计算各个销售量的样本方差，计算结果见图。

图4.25　计算结果

4.2　综合实例

统计函数在实际中有着广泛的应用，在各个行业中Excel的统计函数都可以有灵活的应用，本节将介绍几个使用Excel中的统计函数的实际例子。

4.2.1　成绩单统计

成绩单是使用Excel进行数据分析和统计的典型案例。本小节将通过使用多种统计函数对成绩单进行各种指标的统计，为了简化说明过程，这里选取的是单科成绩单的统计分析，对于实际情况中的多科成绩统计的问题，原理是相同的。具体的统计数据如图4.26所示。

步骤1：统计人数。用户首先需要进行人数的统计，根据情况分为总人数、男生人数、女生人数、参考人数及缺考人数等几个类别，统计结果如图4.27所示。

在单元格F1中输入公式“=COUNT(A2:A13)”，统计学号的登记个数，也就是总人数；在单元格F2中输入公式 “=COUNTIF(B2:B13,"男")”，表示在性别一列中统计男生的人数；在单元格F3中输入公式“=COUNTIF(B2:B13,"女")”，统计在性别一列中的女生人数；在单元格F4中输入公式“=COUNTA(C2:C13)”，统计“成绩”一列中非空单元格的人数，也就是存在分数的人数，也就是参考人数；缺考人数在单元格F5中可以输入公式“=COUNTBLANK(C2:C13)”，统计出在“成绩”一栏中为空格的单元格个数。或者也可以通过总人数和参考人数相减得出缺考的人数，也就是“=F1-F4”。

步骤2：统计考生的分数。下面用户需要进行分数的统计，主要统计的有平均成绩、最高分和最低分等，统计的最后结果如图4.28所示。

图4.26 成绩单数据

图4.27 人数的统计结果

图4.28 成绩的统计结果

在单元格F8中输入公式“=SUMIF(B2:B13,"男",C2:C13)/F2”，统计男生的平均成绩。其中“SUMIF(B2:B13,"男",C2:C13)”表示的是男生的总成绩之和，而F2代表的是男生的人数，因此两者相除得到的就是男生的平均成绩。

在单元格F9中输入公式 “=SUMIF(B2:B13,"女",C2:C13)/(F3-1)”，原理和上面计算男生的平均成绩大致相同，首先计算女生成绩的总和然后除以参考的女生人数，得到的就是女生的平均成绩。

在单元格F10中输入公式 “=AVERAGE(C2:C13)”，就可以直接统计出整个班的平均成绩，前面已经讲过AVERAGE函数会直接忽略没有分数的单元格，得到参考人的平均分数。

在单元格F12中输入公式“=MAX(C2:C13)”，得到成绩的最高分，为96。

在单元格F13中输入公式“=MIN(C2:C13)”，得到成绩的最低分，为51。

步骤3：统计成绩频率。在本步骤中用户需要进行成绩频率的统计。首先统计4个分数段的人数，然后统计分数的众数。最后统计的结果如图4.29所示。

根据成绩的分界点的数字，首先选择在单元格I2中输入下面的公式“=FREQUENCY(C2:C13,I7:I9)”。然后同时按下Shift+Ctrl+Enter组合键，选择单元格I2:I5区域得到分数频率统计的结果，可以方便地得到相应的分数的整体分布。从结果中可以看到，在75~89分的人数最多，有6人。说明这个分数段比较有效。

步骤4：统计分数的众数。最后用户需要统计出现次数最多的分数，在单元格I11中输入公式“=MODE(C2:C13)”，得到的答案是88。

通过上面各个方面的统计数据，用户可以对某个成绩单的分数及相应的分布有一个比较直观的了解。最后的综合结果如图4.30所示。

图4.29　成绩的频率统计结果

图4.30　成绩的最终统计结果

4.2.2　成绩单的标准化

上一节中的成绩单是比较简单的统计分析结果。在实际的成绩分析过程中，一个十分重要的任务就是进行成绩的标准化。对于各次不同的考试结果，一般会有不同的平均值和标准差。在标准化的过程中，需要控制考试的平均值和标准差，经过调整后使得各个学生的成绩都得到某种程度的修改，最后得到一个比较整齐的成绩单。

本节将演示成绩的标准化过程，首先统计原始成绩的平均值和标准差，然后进行分数的标准化，最后将检查标准化后的成绩的分布。

本节例子中的原始数据包含了两个班级的三次考试的成绩，不同的班级的同学的成绩是混在一起的，在后面的数据处理中必须将这两个班级的同学的成绩情况区分开来，如图4.31所示。

步骤1：统计1班学生的人数。在单元格D15中输入公式“=COUNTIF($B$2:$B$11,1)”，该公式是统计在B数据列中班级号为1的同学人数，得到的结果为5，如图4.32所示。

图4.31　原始成绩数据

图4.32　统计1班的学生人数

然后利用自动填充功能统计其他三次考试的学生人数。

步骤2：统计1班学生第一次考试的平均分。在单元格B15中输入公式“=SUMIF(B2:B11,1,C2:C11)/D15”，首先通过SUMIF函数统计班级号是1的同学的成绩总和，然后除以人数，得到平均分，如图4.33所示。

然后利用自动填充功能统计其他三次考试的平均分，得到的结果如图4.34所示。

图4.33　统计1班第一次考试的平均分

图4.34　统计1班学生三次考试的平均分

步骤3：统计1班学生第一次考试成绩的标准差。在单元格C15中输入公式“=STDEV(IF($B$2:$B$11=1,$C$2:$C$11,"FLASE"))”，然后同时按下Shift+Ctrl+Enter组合键，得到的就是1班同学第一次考试成绩的标准差，如图4.35所示。

分析上面的公式：上面的公式比较复杂，在本步骤中将详细分析。首先利用函数IF($B$2:$B$11=1,$C$2:$C$11,"FLASE")返回一个数组：当B列的班级号为1的时候，选取C列的成绩；当B列的班级号为2的时候，则返回字符串“FLASE”，得到的数组中只有1班同学的考试成绩，其他的成绩都被处理为FLASE。最后通过使用STDEV求得这些数值的标准差，因为STDEV函数忽略非数值类的数据，因此可以得到正确答案。用户需要特别注意的是，当不是1班的同学时，应该赋值为“FLASE”，如果忽略不赋值，那么Excel会当作是0来处理，得到的标准差就会出现错误。

步骤4：统计1班学生第二次和第三次考试成绩的标准差。在单元格C16中输入公式“=STDEV(IF($B$2:$B$11=1, $D$2:$D$11,"FLASE"))”，然后同时按下Shift+Ctrl+Enter组合键，得到第二次考试成绩的标准差；在单元格C17中输入公式“=STDEV(IF($B$2:$B$11=1,$E$2:$E$11,"FLASE"))”，然后同时按下Shift+Ctrl+Enter组合键，得到第三次考试成绩的标准差，得到的结果如图4.36所示。

图4.35　统计成绩的标准差

图4.36　一1班原始原始成绩统计结果

步骤5：统计2班学生考试成绩的数字特征。上面的几个步骤用户已经统计了1班学生考试成绩的数字特征，主要包括人数、平均分和标准差等。下面用户可以使用类似的方法来统计2班学生的考试成绩的数字特征，得到的结果如图4.37所示

图4.37　2班原始成绩统计结果

分析成绩标准化的方法：前面的步骤已经完成了原始考试成绩的统计，下面的步骤可以对成绩进行标准化。标准化的原理是分数分布符合t分布，根据标准化的条件，标准化后的成绩必须以80为均值，以5为标准差。而现在原始的分数，1班第一次考试的成绩以87.6为平均值，以8.73为标准差，需要进行转换。

转换的公式为：调整后的分数=（原始分数-原始平均分数）/原始标准差*5+80。

上面的线性公式形式，是T分布的性质。

步骤6：进行成绩转换。根据上面的转换公式及前面的步骤中统计出来的标准差和均值的参数，用户可以很方便地求出调整后的标准化成绩，如图4.38所示。

在单元格C20中输入公式：“=IF(B20=1,(C2-$B$15)/$C$15*5+80,(C2-$G$15)/$H$15*5+80)”。由于两个班的统计中，标准差和均值都是互不相同的，因此，需要进行条件判断，如果是1班的学生，使用1班学生成绩的标准差和均值进行转化，2班的学生则使用2班的学生成绩的标准差和均值进行转化。

步骤7：转换其他考试成绩。用户可以利用与步骤6中类似的公式将其他考试成绩的进行转换，得到的结果如图4.39所示。

图4.38　标准化后的成绩

图4.39　转换其他的考试成绩

步骤8：检验标准化后的成绩。在最后用户需要对转化后的标准化成绩进行检验。检验的方法就是对转换后的标准化成绩进行均值和标准差的分析统计，是否符合均值为80，标准差为5的调整目标，检查后的情况如图4.40所示。


[image: ]



图4.40　检查标准化后的成绩

在图中单元格C32中输入公式：“=STDEV(IF($B$20:$B$29=1,$C$20:$C$29,"FLASE"))”，这个公式的原理和前面1班的考试成绩的标准差是相同的，这里就不详细描述了。根据上面的结果，可以看到两个班的同学在三次考试中的考试成绩都是均值为80，标准差为5，达到了调整目标，本例的完整的演示过程如图4.41所示。

图4.41　成绩的标准化


第5章 查询和引用函数

查询和引用函数是在数据清单或工作表中查找特定数值，或者查找某一个单元格引用的函数。

引用函数中的参数主要是引用的各种类型，用户可以用不同的方法进行引用。当用户需要进行单元格引用时，参数一般是单元格的行或列，例如在ADDRESS和INEDX中，参数row_num表示数组中某行的行序号，函数从该行返回数值；参数column_num表示数组中某列的列序号，函数从该列返回数值。

如果需要引用的是单元格区域或者多个区域，例如在函数AREAS、COLUMN和ROW中，一般会使用参数reference；如果需要得到其列数或行数的数组或数组公式，例如在COLUMNS和ROWS函数中，一般会使用参数array。

这些参数在后面小节的内容中指代的意义都是相同的，在后面就不再重复了。

查询函数一般会有参数lookup_value，表示需要在数据表进行查找的数值，可以为数值、引用或文本字符串；参数table_array或者lookup_array，表示需要在其中查找数据的数据表。在一些查询函数中还会有逻辑参数，表示的是查询的类型或者匹配类型。

查询和引用函数在Excel中使用比较广泛，下面举了简单的例子来说明如何使用这些函数进行简单的引用和查询，如图5.1所示。

图中使用了函数INDIRECT引用了相应单元格的数据，在公式“=INDIRECT($A$2)”中，表示引用的是单元格B4的数据，返回的是数值“62”；在公式“=INDIRECT($A$1)中”，表示引用的是单元格B3的数据，返回的是数值“10”。

同样用户可以从下面比较简单的查询函中了解查询函数的使用方式，如图5.2所示。

图5.1　使用引用函数

图5.2　使用查询函数

本章将着重介绍查询和引用函数功能，对它们的功能、表达式及参数进行详尽的讲述，对每个函数的应用都举例说明。在本章的最后，将讲述2个综合实例，让用户进一步理解常见的查询和引用函数的使用方法，以及如何灵活应用在实际工作中。

5.1　函数应用实战

5.1.1　ADDRESS函数：返回引用地址

ADDRESS函数的功能是按照给定的行号和列标，建立文本类型的单元格地址。

ADDRESS函数的表达式是ADDRESS(row_num,column_num,abs_num,a1,sheet_text)。

其中abs_num表示指定返回的引用类型。a1用以指定A1或R1C1引用样式的逻辑值（A1 和 R1C1 是Excel中的两种引用样式。A1的引用样式是Excel的默认引用类型。这种类型引用字母标志列（从 A 到 IV ，共 256 列）和数字标志行（从 1 到 65536）。R1C1 引用样式中，Excel 使用R加行数字和C加列数字来指示单元格的位置。）。如果a1为TRUE或省略，函数ADDRESS返回A1样式的引用；如果a1为FALSE，函数ADDRESS返回R1C1样式的引用。sheet_text是文本，指定作为外部引用的工作表的名称，如果省略sheet_text，则不使用任何工作表名。

【实际应用】

某公司需要检查单位内10名员工的销售情况，为了保证公平，实行随机选取员工编号的方式来检查。检查的规则是在A列随机产生员工编号的行数，然后返回相应的员工编号，如图5.3所示。

图5.3　随机选取员工编号

在上面的单元格A2中输入公式“=RAND()×9+2”，这样会返回出2~11之间的随机数；在单元格B12中输入公式“=INDIRECT(ADDRESS(A2,2))”，其中的INDIRECT函数将在后面介绍，表示显示选取地址的内容；在单元格B13中输入公式“=ADDRESS(A2,2)”，返回需要选择员工编号的地址，在上面的情况中返回的是$B$7，因此得到了员工编号为130。

【注意事项】

在ADDRESS函数中的参数abs_num 返回的引用类型归纳如下：

1或省略，表示绝对引用

2表示绝对行号，相对列标

3表示相对行号，绝对列标

4表示相对引用

5.1.2　AREAS函数：返回区域个数

AREAS函数的功能是返回引用中包含的区域个数。区域表示连续的单元格区域或某个单元格。

AREAS函数的表达式是AREAS(reference)。

【实际应用】

AREAS函数的使用比较简单，下面举个常见的例子，如图5.4所示。

图5.4　使用AREAS函数

5.1.3　CHOOSE函数：从列值中选择数值

CHOOSE函数的功能是返回数值参数列表中的数值。使用函数CHOOSE可以基于索引号返回多达29个基于待选数值中的任一数值。

CHOOSE函数的表达式是CHOOSE(index_num,value1,value2,...)。

其中参数index_num表示用以指明待选参数序号的参数值。index_num必须为1~29之间的数字，或者是包含数字1~29的公式或单元格引用。

【实际应用】

在一次比赛中，统计了某个小组中的各运动员的成绩，现在需要给该小组的前三名的运动员颁发奖牌，其他的运动员则没有奖牌。各运动员在单位时间内的跑步距离的数据如图5.5所示。

图5.5　跑步距离数据

在单元格C2中输入公式“=RANK(B2,$B$2:$B$6)”，可以得到１号运动员的排名，然后利用自动填充功能来计算其他运动员的排名。下面要计算各位运动员的奖牌情况，如图5.6所示。

图5.6　奖牌分配情况

在单元格D2中输入公式：=IF(C2<=3,CHOOSE(C2,"Gold","Silver","Bronze"),"unplaced")。

上面的公式中首先通过条件C2<=3来判断是否会分配奖牌，如果C2>3那么就是unplaced，如果C2<=3，则通过CHOOSE(C2,"Gold","Silver","Bronze")选择相应的奖牌，如果是第一名就是Gold，第二名就是Silver，第三名就是Bronze。

5.1.4　COLUMN和ROW函数：返回列标和行号

COLUMN函数的功能是返回给定引用的列标；ROW函数的功能是返回引用的行号。

COLUMN函数的表达式是COLUMN(reference)。

ROW函数的表达式是ROW(reference)。

【实际应用】

某公司每天统计当天的销售情况。根据公司的规则，在下面的单元格中每个行数为偶数的销售量表示的下午的销售量，下面根据情况统计最近三天下午的销售总量，如图5.7所示。

带有阴影的单元格表示的就是下午的销售量，在单元格B8中输入公式：

=SUM(IF(MOD(ROW(A1:A6),2)=0,A1:A6))

然后同时按下Shift+Ctrl+Enter组合键，就可以得到销售的总量。在这个公式中IF(MOD(ROW(A1:A6),2)=0是用来判断单元格的行数是否为偶数，如果是偶数，那么返回相应的单元格数据；如果不是偶数，就返回0，然后再求和得到总销量。

【注意事项】

函数COLUMN和ROW的参数可以省略，这个时候两个函数返回的分别是当前函数所在的单元格的列数和行数，如图5.8演示了两个函数的常用参数及相应的说明。

图5.7　销售总量

图5.8　函数的参数使用

一般来讲，两个函数返回的是列标或行号，但是有时可以使用两个函数返回数组的数值，这个时候必须同时按下“Shift+Ctrl+Enter”键，如图5.9所示。

图5.9　返回数组的数值

5.1.5　COLUMNS和ROWS函数：返回引用的数目

COLUMNS函数的功能是返回数组或引用的列数；ROWS函数的功能是返回引用或数组的行数。

COLUMNS函数的表达式是COLUMNS(array)。

ROWS函数的表达式是ROWS(array)。

【实际应用】

下面演示两个函数在Excel中的应用方式，如图5.10所示。

图5.10　使用COLUMNS和ROWS函数

5.1.6　HLOOKUP函数：实现水平查找

HLOOKUP函数的功能是在表格或数值数组的首列查找指定的数值，并返回表格或数组当前行中指定列处的数值。HLOOKUP函数的表达式是：

HLOOKUP(lookup_value,table_array,row_index_num,range_lookup)。

参数row_index_num表示table_array中待返回的匹配值的行序号；range_lookup是逻辑值，指明函数HLOOKUP查找时是精确匹配，还是近似匹配。如果为TRUE或省略，则返回近似匹配值。也就是说，如果找不到精确匹配值，则返回小于lookup_value的最大数值。如果range_lookup为FALSE，函数HLOOKUP将查找精确匹配值，如果找不到，则返回错误值#N/A!。

【实际应用】

某个建材经销商提供价格表中显示了砖、木材和玻璃的单价，在单价表中标识出来。如图5.11所示的采购表是某公司的采购预算，现在公司需要及时地更新价格，保持和单价表相同的单价。

图5.11　采购预算

在图中的单元格C12中输入公式“=HLOOKUP(A12,$A$2:$C$3,2,FALSE)”，表示查找在A2:C3数组中，“玻璃”列中的第二行的数值，就是相应的单价。然后利用自动填充功能得到其他采购项目的单价。

【注意事项】

HLOOKUP函数中的参数table_array的第一行的数值可以为文本、数字或逻辑值；如果range_lookup为TRUE，则table_array的第一行的数值必须按升序排列：…-2、-1、0、1、2…、A~Z、FALSE、TRUE；否则，函数HLOOKUP将不能给出正确的数值。如果range_lookup为FALSE，则table_array不必进行排序；如果是文本，文本不区分大小写。

5.1.7　HYPERLINK函数：创建跳转

HYPERLINK函数的功能是创建一个快捷方式（跳转），用以打开存储在网络服务器中的文件。

HYPERLINK函数的表达式是HYPERLINK(link_location,friendly_name)。

其中参数link_location是文档的路径和文件名，当单击函数HYPERLINK所在的单元格时，MicrosoftExecl将打开存储在link_location中的文件，此文档可以作为文本打开；friendly_name是单元格中显示的跳转文本值或数字值。单元格的内容为蓝色并带有下划线。如果省略friendly_name，单元格将link_location显示为跳转文本。

【实际应用】

某公司的人事部门在进行人事资料的整理，需要批量将数据的D列设置超链接：要显示的文字为:A列的内容（员工的姓名），地址为D列的内容（员工的照片），如图5.12所示。

图5.12　设置超链接

在上面的单元格D2中输入公式“=HYPERLINK("d:\pic\02.jpg",A2)”，表示链接的文件是d:\pic\02.jpg，在链接的地方显示的名称为A2中的内容。其他的几个单元格的设置方法也是类似的，但是不要使用自动填充功能，因为连接的图片文件是互不相同的。

【注意事项】

HYPERLINK中的参数link_location还可以指向文档中的某个更为具体的位置，如Execl工作表或工作簿中特定的单元格或命名区域，或是指向Microsoft Word文档中的书签。路径可以是存储在硬盘驱动器上的文件，或是在Internet上的“统一资源定位符”（URL）路径。link_location可以为括在引号中的文本字符串，或是包含字符串链接的单元格。如果在link_location中指定的跳转不存在或不能访问，则当单击单元格时将出现错误信息。

5.1.8　INDEX函数：返回指定内容

INDEX函数可以返回列表或数组中的元素值，此元素由行序号和列序号的索引值给定。

INDEX函数有两种语法形式：数组和引用。数组形式通常返回数值或数值数组，引用形式通常返回某个引用地址。当函数INDEX的第一个参数为数组常数时，使用数组形式。

数组形式的表达式是INDEX(array,row_num,column_num)。

引用形式的表达式是INDEX(reference,row_num,column_num,area_num)。

其中area_num表示选择引用中的一个区域，并返回该区域中row_num和column_num的交叉区域。

【实际应用】

某超市整理了20种商品的价格，现在需要统计的是这20种商品中价格超过某个标准的个数，这个标准在单元格B1中输入，可以随时更改，如图5.13所示。

图5.13　检测结果

在图中的单元格B22中输入公式“=COUNTIF(A1:A20,">="&INDEX(B1,1))”，其中INDEX(B1,1)返回的就是单元格B1中的数值，因此得到的就是在前面20个数据中大于等于B1单元格的标准的数值。

5.1.9　INDIRECT函数：返回指定的引用

INDIRECT函数的功能是返回由文本字符串指定的引用。此函数立即对引用进行计算，并显示其内容。

INDIRECT函数的表达式是INDIRECT(ref_text,a1)。

其中参数ref_text是对单元格的引用，此单元格可以包含a1样式的引用、R1C1样式的引用、定义为引用的名称或对文本字符串单元格的引用。如果ref_text不是合法的单元格的引用，函数INDIRECT返回错误值#REF!；a1是逻辑值，指明包含在单元格ref_text中的引用的类型。



【实际应用】



某销售公司在7个部门的相对于上个月的销售情况按照地理情况分布，记录在A1:G1单元格中，其中负数表示相对于上个月销售情况减少，正数表示增加。现在需要知道第一个个销量增加的部门的增加数量，如图5.14所示。

图5.14　查看销售情况

在图中的单元格A2中输入公式“=IF(COUNTIF(A1:G1,">0")=0,0,INDIRECT(ADDRESS(1,SMALL(IF(A1:G1>0,COLUMN(A1:G1)),1))))”，返回在单元格A1:G1中第一个大于0的数值。其中SMALL(IF(A1:G1>0,COLUMN(A1:G1)),1)返回其中第一个大于0的单元格的列数，ADDRESS(1,SMALL(IF(A1:G1>0,COLUMN(A1:G1)),1))返回的就是这个单元格的地址。最后再用INDIRECT函数返回单元格的数值。IF函数保证了必须是大于0的数值才会返回单元格的数值，如果所有的数值都小于0，那么就会返回0。

【注意事项】

如果ref_text是对另一个工作簿的引用（外部引用），则那个工作簿必须被打开。如果源工作簿没有打开，函数INDIRECT返回错误值#REF!。如果参数a1为TRUE或省略，ref_text被解释为A1样式的引用；如果参数a1为FALSE，ref_text被解释为R1C1样式的引用。

5.1.10　LOOKUP函数：查找数据

LOOKUP函数有向量形式和数组形式两种。向量形式是在单行区域或单列区域（向量）中查找数值，然后返回第二个单行区域或单列区域中相同位置的数值。LOOKUP函数的数组形式是在数组的第一行或第一列中查找指定数值，然后返回最后一行或最后一列中相同位置处的数值。

向量形式的表达式为LOOKUP(lookup_value,lookup_vector,result_vector)。

lookup_vector是只包含一行或一列的区域；lookup_vector的数值可以为文本、数字或逻辑值；result_vector表示只包含一行或一列的区域，其大小必须与lookup_vector相同。

数组形式的表达式为LOOKUP(lookup_value,array)。

参数array表示包含文本、数字或逻辑值的单元格区域，它的值用于与lookup_value进行比较。

【实际应用】

某学校统计了某次考试中部分班级的学生考试成绩，学校需要统计本次考试中的最高分和最高分学生的姓名和班级，如图5.15所示。

图5.15　分数的统计情况

在图中的单元格B9中输入公式“=MAX(F2:F4)”，得到了本次考试成绩的最高分；然后在单元格B7中输入公式“=LOOKUP($B$9,$F$2:$F$4,A2:A4)”，得到最高分同学所在的班级；最后在单元格B8中输入公式“=LOOKUP($B$9,$F$2:$F$4,B2:B4)”，得到最高分同学的姓名。

【注意事项】

在函数LOOKUP中参数lookup_vector的数值必须按升序排序：…-2、-1、0、1、2…、A-Z、FALSE、TRUE；否则，函数LOOKUP不能返回正确的结果。如果函数LOOKUP找不到lookup_value，则查找lookup_vector中小于或等于lookup_value的最大数值。如果lookup_value小于lookup_vector中的最小值，函数LOOKUP返回错误值#N/A。

5.1.11　MATCH函数：在数组中查找

MATCH函数的功能是返回在指定方式下与指定数值匹配的数组中元素的相应位置。

MATCH函数的表达式为MATCH(lookup_value,lookup_array,match_type)。

其中match_type是数字-1、0或1，它指明Excel如何在lookup_array中查找lookup_value。

【实际应用】

各个小车生产厂家和配件的金额情况已经列表给出，某个小车生产厂商需要根据自己的条件重新列出一个新表，如图5.16所示。

在图中的表格中，用户需要填写的是各个厂商的各个不同配件的报价情况。在单元格D9中输入公式“=HLOOKUP(B9,$B$1:$D$6,MATCH(C9,$A$2:$A$6,0)+1,FALSE)”，在公式中首先用函数MATCH(C9,$A$2:$A$6,0)求得配件所在的行数，由于相应的同行配件是C9单元格中的文本，然后用HLOOKUP函数利用厂商名称找到相应的报价。然后利用自动填充功能得到其他配件的报价，如图5.17所示。

图5.16　原始的列表情况

图5.17　得到相应的配件表

5.1.12　OFFSET函数：调整新的引用

OFFSET函数的功能是以指定的引用为参照系，通过给定偏移量得到新的引用。返回的引用可以为一个单元格或单元格区域。并可以指定返回的行数或列数。

OFFSET函数的表达式为OFFSET(reference,rows,cols,height,width)。

其中参数reference作为偏移量参照系的引用区域；rows表示相对于偏移量参照系的左上角单元格，上（下）偏移的行数；cols表示相对于偏移量参照系的左上角单元格，左（右）偏移的列数；height表示高度，即所要返回的引用区域的行数。height必须为正数；width表示宽度，即所要返回的引用区域的列数。

【实际应用】

某销售公司统计了六个月的销售额，现在需要求得销售总额。在Excel中可以简单使用SUM公式，但是在本例中，需要OFFSET函数来求解这个问题，方便用户了解OFFSET函数的用法，如图5.18所示。

图5.18　使用OFFSET函数求和

在在单元格B9中输入公式“=SUM(OFFSET(B2,,,ROW()-ROW(B2)))”，在公式中OFFSET函数返回的是B2:B7的数列，因此再使用SUM公式求解相应的总和。

【注意事项】

如果行数和列数偏移量超出工作表边缘，函数OFFSET返回错误值#REF!；如果省略height或width，则假设其高度或宽度与reference相同；函数OFFSET实际上并不移动任何单元格或更改选定区域，它只是返回一个引用。函数OFFSET可用于任何需要将引用作为参数的函数。

5.1.13　VLOOKUP函数：实行竖直查找

VLOOKUP函数的功能是在表格或数值数组的首列查找指定的数值，并由此返回表格或数组指定列处的数值。

VLOOKUP函数的表达式为VLOOKUP(lookup_value,table_array,col_index_num,range_lookup)

col_index_num表示table_array中待返回的匹配值的列序号；range_lookup表示逻辑值，指明函数VLOOKUP返回时是精确匹配还是近似匹配。

【实际应用】

本例子沿用前面在HLOOKUP函数中的例子。某个建材经销商提供价格表中显示了砖、木材和玻璃的单价，在单价表中标识出来。采购表是某公司的采购预算，现在公司需要及时地更新价格，保持和单价表相同的单价，如图5.19所示。

图5.19　查找对应的单价

在单元格C10中输入公式“=VLOOKUP(A10,$F$2:$G$4,2,FALSE)”，A10单元格中的数值表示的就是需要查找的项目名称，$F$2:$G$4表示需要查找的数据表，2表示返回的是该数据表中的第二列数据，FALSE表示是精确匹配。然后利用自动填充功能来完成其他项目的单价搜索功能。

下面开始查找相应的折扣，如图5.20所示。

图5.20　查找对应的折扣

在单元格D10中输入公式“=VLOOKUP(B10,$A$3:$D$5,MATCH(A10,$B$2:$D$2,0)+1,TRUE)”，B10单元格中的数值表示的就是需要查找的数量，$A$3:$D$5表示需要查找的数据表，MATCH(A10,$B$2:$D$2,0)+1表示返回的是项目在$A$3:$D$5数据表中的列数，TRUE表示是不精确匹配。然后利用自动填充功能来完成其他项目的单价搜索功能，得到各个项目开发数量的折扣。最后在单元格E10中输入公式“=(B10×C10)-(B10×C10×D10)”，得到最后的项目用量。

5.2　综合实例

查询和引用函数在Excel中有广泛的应用，通过使用查询函数可以在原来的数据表中选择适合条件的数据，然后利用其他章节的函数对数据进行处理。引用函数是对原来数据表的数据进行选择的另外一种函数，同样会有广泛的使用。在本节中，将选择两个综合的案例来演示综合使用查询和引用函数的方法。

5.2.1　元件配比

下面的原始数据表中记录了某个工厂中3月1日到3月4日的零件主件使用情况，同时根据产品的创建情况，在另外的一个数据表中提供了主件和元件的配比情况。下面用户需要计算每天元件的使用情况，具体情况如图5.21所示。

图5.21　原始的配比数据

下面计算3月1日中元件CN674005-1使用的件数，首先需要分析如何计算具体的件数，通过单元格E8中的元件CN674005-1号在元件B列数据进行匹配，然后通过在日期中的主件序号在主件A列数据进行匹配，最后根据相应的匹配数目，得到3月1日中使用元件CN674005-1的具体数目。

步骤1：统计3月1日中元件CN674005-1使用的件数。根据上面的分析思路，在单元格F8中输入公式：

=SUM(LOOKUP(LOOKUP(ROW($A$8:$A$15),IF(($A$8:$A$15<>""),ROW($A$8:$A$15)),$A$8:$A$15),$C$3:$C$4,D$3:D$4)×($B$8:$B$15=$E8)×$C$8:$C$15)。

分析上面的公式：这个公式比较复杂，从这个步骤开始对整个公式的各个部分进行分析。

IF(($A$8:$A$15<>""),ROW($A$8:$A$15))：首先根据配比情况数据表的A8:A15单元格情况，可以看到，本工厂只有两种主件分别为CN744058和CN744059，通过这个逻辑函数返回的数组1：(8，FALSE，FALSE，FALSE，12，FALSE，FALSE，FALSE)。在有主件标识的行中返回行数，在没有主件标识的行中，返回“FALSE”。

LOOKUP(ROW($A$8:$A$15),IF(($A$8:$A$15<>""),ROW($A$8:$A$15)),$A$8:$A$15)：这个公式的目的其实是填充A列的主件元素，根据LOOKUP函数的性质，如果没有找到匹配的数据，那么返回的数组是最接近的小于原来列数的数值，因此，这个函数返回的是如下的数组2：(CN744058，CN744058，CN744058，CN744058，CN744059，CN744059，CN744059，CN744059)。

LOOKUP(LOOKUP(ROW($A$8:$A$15),IF(($A$8:$A$15<>""),ROW($A$8:$A$15)),$A$8:$A$15),$C$3:$C$4,D$3:D$4)：返回的是3月1日中各个主件的使用数目。在C3:C4列中选择上面的数组2中的主件标号，返回相应的D3:D4中的使用数目，得到的结果是下面的数组3：(30，30，30，30，20，20，20，20)。

$B$8:$B$15=$E8：返回的是逻辑数组，当B8:B15中存在等于E8的数值时，返回TURE，否则返回的是FALSE。得到的下面的数组4：(FALSE，TRUE，FALSE，FALSE，FALSE，TRUE，FALSE，FALSE)。

最后返回$C$8:$C$15的配比数据，进行数组的乘法计算，(30，30，30，30，20，20，20，20)×(0，1，0，0，0，1，0，0)×(0，4，3，2，0，2，1，1)=4×30+20×2=160。

在数组计算中FALSE当作0处理，TURE当作1处理。上面的三个数组的乘法正好就是3月1号的元件使用数据，上面的计算过程如图5.22所示。

图5.22　计算过程

根据上面的分析和解答过程，可以看到得到的结果如图5.23所示。

图5.23　计算3月1日的元件使用情况

步骤2：利用自动填充功能，计算其他日期中的各个元件的使用情况，得到的最终统计结果如图5.24所示。

图5.24　最终的统计结果

在上面的实例分析的过程中，需要提醒用户注意的有以下几个问题：

在使用ROW函数返回各行的数据的时候，用户需要使用的是Shift+Ctrl+Enter组合键进行输入，得到的是数组数据而不是返回相应的行数。

在利用自动填充功能时，需要十分注意的是引用模式的问题。在上面的公式中使用了相对引用、绝对引用和相对行或列引用等各个引用模式。例如在ROW($A$8:$A$15)中单元格的引用模式是绝对引用，因为这些数据是固定不变的，如果使用的是相对引用，那么这些数据会在某个单元格中相对变化；而在公式LOOKUP(ROW($A$8:$A$15),IF(($A$8:$A$15<>""),ROW($A$8:$A$15)),$A$8:$A$15),$C$3:$C$4,D$3:D$4)中，D$3:D$4选择的是绝对行引用，相对列引用。因为在日期表中3，4行分别表示两个主件的使用个数是保持不变的，因此需要绝对引用，但是每天所在的列是不同的，因此需要相对引用。

数组的计算方法是十分重要的一个方法。而且数组的乘法显然和数据的乘法是不相同的，特别是逻辑数组在数组乘法中可以方便地转化为数据数组的计算。

5.2.2　欠款人员统计

某单位的财务部门统计了公司的债权情况。由于每个借款人员的还款时间都是互不相同的，因此，需要设置一个这样的一个公式，根据借款清单，将未结清的名单自动顺序列出，要求不留空格。结清后又自动删除，重新排序，而且不需要进行手工处理。目前的还款和欠款情况如图5.25所示。

图5.25　原始的欠款清单

图中，现在依然欠款的姓名名单用黄色底纹标出，在欠款人员名单中显示了相应的人员名称和欠款的数额。由于这些欠款情况会随时改变的，在不允许手工改变的情况下，如何随时将实际的欠款情况反映到欠款人员名单就是关键的地方。

要实现上面提到的目标，需要用到查询和引用函数。用户可以分成两步来解答这个问题。首先通过查询原始的“借款清单”中的还款情况，得到欠款人员的姓名，然后利用查询出来的姓名在原来的“借款清单”中的借款金额中查到相应的欠款数额，最后利用求和公式求得目前的欠款总数。

步骤1：在单元格K5中输入公式：=IF(ROW(1:1)>COUNTIF(D$5:D$19,""),"",INDEX(B:B,SMALL(IF(D$5:D$19=0,ROW($5:$19)),ROW(1:1))))，然后同时按下Shift+Ctrl+Enter组合键输入数组公式得到了第一个欠款人员的姓名。下面详细介绍这个公式的含义。

INDEX(B:B,SMALL(IF(D$5:D$19=0,ROW($5:$19)),ROW(1:1)))公式返回的就是在B列中没有还款人员的名单。IF(D$5:D$19=0,ROW($5:$19))返回的是没有还款的数据的行数，INDEX函数返回在B列中该行的数据，也就是欠款人员的姓名。SMALL函数保证了在欠款人员名单中的顺序和原来在借款清单中的顺序一致。

为了保证可以及时更新欠款的情况，“欠款人员名单”的行数和原始的“借款名单”的行数保持一致。因为欠款人最多的情况下，目前的欠款人员和原来借款人员一样。

步骤2：在单元格K5输入公式以后，利用Excel的自动填充功能，将该公式填充到K6:K19单元格区域中，得到的结果如图5.26所示。

图5.26　搜索欠款人员的姓名

步骤3：搜索欠款人员的欠款金额。在单元格L5中输入公式：

=IF(K5="","",IF(K5="合计",SUM(L$4:L4),VLOOKUP(K5,B:C,2,0)))

然后同时按下Shift+Ctrl+Enter组合键输入数组公式得到了第一个欠款人员的欠款金额。下面详细介绍这个公式的含义。

IF(K5="合计",SUM(L$4:L4),VLOOKUP(K5,B:C,2,0))，表示首先判断单元格K5是否是“合计”，如果是欠款人员的名单那么返回的是VLOOKUP(K5,B:C,2,0)的数值，利用K5在Ｂ:C列中查询第二列的数据，也就是相应的欠款金额；如果是“合计”，那么就用SUM(L$4:L4)计算前面所有欠款人员的欠款金额总和。最外层的IF函数判断是否存在数据，如果没有数据，返回空格。

步骤4：在单元格L5输入公式以后，利用自动填充功能，将该公式填充到L6:L19单元格区域中，得到的结果如图5.27所示。

图5.27　搜索欠款人员的金额

步骤5：检查函数功能。在结束本例子之前，用户可以根据情况，自己检查函数是否能够达到原来的要求。用户可以在原来的数据中，将前面三个客户的还款的时间删除，相当于这三个客户没有还款，可以查看欠款人员清单如何变化，如图5.28所示。

图5.28　更新的欠款人员清单

从上面的例子中可以看出，当前面三个用户的还款时间被删除以后，相应的欠款人员的名单也随着更新了，所以，前面设置的公式达到要求。


第6章 数据库函数

在Excel中包含了一些工作表函数。它们用于对存储在数据清单或数据库中的数据进行分析，判断是否符合特定条件，这些函数统称为数据库函数。这一类函数具有下面一些共同特点：

·每个函数均有三个参数：database、field和criteria。这些参数指向函数所使用的工作表区域。

·每个函数都以字母D开头。

·如果将字母D去掉，其中大多数数据库函数已经在Excel的其他类型函数中出现过了。比如，DAVERAGE将D去掉的话，就是求平均值的函数AVERAGE。

每一个数据库函数都具有相同的三个参数，因此在本章的开始将首先介绍一下该类函数的几个参数。数据库函数的语法形式为函数名称(database,field,criteria)。

参数database为构成数据清单或数据库的单元格区域。数据库是包含一组相关数据的数据清单，其中包含相关信息的行为记录，而包含数据的列为字段。数据清单的第一行包含着每一列的标志项。

参数field为指定函数所使用的数据列。数据清单中的数据列必须在第一行具有标志项。field可以是文本，即两端带引号的标志项，如“使用年数”或“产量”；此外，field也可以是代表数据清单中数据列位置的数字：1表示第一列，2表示第二列，等等。

参数criteria为一组包含给定条件的单元格区域。可以为参数criteria指定任意区域，只要它至少包含一个列标志和列标志下方用于设定条件的单元格。

每一个数据库函数都有条件区域，其中条件是指查询或筛选的结果集中包含哪些记录的条件。建立条件区域的基本要求包括有：

·在可用作条件区域的数据清单上插入至少三个空白行。

·条件区域必须具有列标志。

·确保在条件值与数据清单之间至少留了一个空白行。

在创建数据筛选条件的时候，用户需要在列标志下面的一行中，键入所要匹配的条件文本。所有以该文本开始的项都将被筛选。如果用户键入文本“Dav”作为条件， Excel将查找“Davolio”、“David”和“Davis”。如果只匹配指定的文本，可键入公式''=text''，其中“text”是需要查找的文本。如果要查找某些字符相同但其他字符不一定相同的文本值，则可使用通配符。

对于每一个数据库函数，都需要一个基础数据库，为了方便讲解各个数据库函数，本章中将使用统一的数据清单，如图6.1所示。

图6.1　数据清单

下面利用图中的数据清单，演示在前面提到的条件区域的设置方法。以产品和商标的区域查询条件为例，如图6.2所示。

图6.2　条件区域

在图中的表格中，第一行的“产品”和“商标”是条件区域的列名部分，第二行是数据库查询的条件部分。对于上面的列名部分，需要提醒用户的是，最好不要使用直接输入的方法，因为自己手工输入经常会产生误差，使得用户无法查询到符合条件的数据记录。最好的方法应该是使用Excel的“复制”和“粘贴”命令，或者使用公式引用列名所在的单元格。对于条件部分，可以根据相应的列名输入条件文本或公式，条件除了单一的数据查询条件之外，还可以是多种条件，例如图中就是两个查询条件。

6.1　函数应用实战

6.1.1　DAVERAGE函数：返回条目的平均值

DAVERAGE函数的功能是返回清单或数据库中满足指定条件的列中数值的平均值。

DAVERAGE函数的表达式是DAVERAGE(database,field,criteria)。

【实际应用】

根据上面的数据清单，下面举例按照各种不同的条件计算不同的平均值，如图6.3所示。

步骤1：输入条件区域。用户需要根据需要计算的情况输入条件区域，上面的3种情况包括了商标、产品、瓦数等列数，因此，需要输入的条件区域如图6.4所示。

图6.3　需要计算的情况

图6.4　输入条件区域

步骤2：计算第一种情况。在单元格E20中输入公式“=DAVERAGE(A1:H17,E1,A24:A25)”，该公式表示计算商标为上海的平均单价，计算结果如图6.5所示。

步骤3：计算第二种情况。在单元格E21中输入公式“=DAVERAGE(A1:H17,E1,A24:B25)”。

步骤4：计算第三种情况。在单元格E22中输入公式“=DAVERAGE(A1:H17,E1,B24:C25)”，得到的最后结果如图6.6所示。

图6.5　输入公式

图6.6　计算结果

【注意事项】

上面的例子中，多个查询条件是相交的条件，表示必须全部满足的条件。在Excel中，还可以使用相并的条件，表示满足其中的某一个条件。

6.1.2　DCOUNT函数：返回单元格个数

DCOUNT函数的功能是返回数据库或数据清单的列中满足指定条件并且包含数字的单元格个数。

DCOUNT函数的表达式是DCOUNT(database,field,criteria)。

【实际应用】

现在根据情况需要使用DCOUNT函数计算如图6.7所示的3种情况。

图6.7　需要计算的情况

步骤1：输入条件区域。根据上面需要计算的情况输入条件区域，上面的3种情况包括了采购盒数、产品、瓦数等列数，因此，需要输入的条件区域如图6.8所示。

步骤2：计算第一种情况。在单元格E20中输入公式“=DCOUNT(A1:H17,G1,A24:B25)”，该公式表示计算采购盒数为5的白炽灯的数量，计算结果如图6.9所示。

图6.8　输入条件区域

图6.9　输入公式

步骤3：计算第二种情况。在单元格E21中输入公式“=DCOUNT(A1:H17,B1,B24:C25)”，该公式表示计算瓦数大于等于80的白炽灯数量。

步骤4：计算第三种情况。在单元格E22中输入公式“=DCOUNT(A1:H17,B1,B24:D25)”，该公式表示瓦数大于等于80小于等于100的白炽灯数量，得到的最后结果如图6.10所示。

图6.10　计算结果

6.1.3　DGET函数：返回符合条件的数值

DGET函数的功能是从数据清单或数据库的列中提取符合指定条件的单个值。

DGET函数的表达式是DGET(database,field,criteria)。

【实际应用】

现在根据情况需要使用DGET函数计算如图6.11所示的4种情况。

步骤1：输入条件区域。用户需要根据需要计算的情况输入条件区域，上面的4种情况包括了每盒数量，产品，商标，瓦数，寿命，因此，需要输入的条件区域如图6.12所示。

图6.11　需要计算的情况

图6.12　输入条件区域

步骤2：计算第一种情况。在单元格G20中输入公式“=DGET(A1:H17,G1,B24:D25)”，该公式表示计算商标为上海的白炽灯瓦数为100的采购盒数，计算结果如图6.13所示。

图6.13　输入公式

步骤3：计算后面三种情况。在单元格G21中输入公式“=DGET(A1:H17,C1,A26:C27)”，在单元格G22中输入公式“=DGET(A1:H17,B1,B26:D27)”，在单元格G23中输入公式“=DGET(A1:H17,E1,A24:C25)”得到的最后结果如图6.14所示。

图6.14　计算结果

6.1.4　DMAX函数：返回符合条件的最大值

DMAX函数的功能是返回数据清单或数据库的列中满足指定条件的最大数值。

DMAX函数的表达式是DMAX(database,field,criteria)。

【实际应用】

现在根据情况需要使用DMAX函数计算如图6.15所示的3种情况。

步骤1：输入条件区域。用户需要根据上面需要计算的情况输入条件区域，上面的3种情况包括了产品、商标、瓦数，因此需要输入的条件区域如图6.16所示。

图6.15　需要计算的情况

图6.16　输入条件区域

步骤2：计算第一种情况。在单元格F20中输入公式“=DMAX(A1:H17,H1,A24:A25)”，该公式表示计算上海品牌的最大价值，计算结果如图6.17所示。

步骤3：计算后面两种情况。在单元格F21中输入公式“=DMAX(A1:H17,H1,B24:C25)”，在单元格F22中输入公式“=DMAX(A1:H17,E1,A24:B25)”，得到的最后结果如图6.18所示。

图6.17　输入公式

图6.18　计算结果

6.1.5　DMIN函数：返回符合条件的最小值

DMIN函数的功能是返回数据库的列中满足指定条件的最小数值。

DMIN函数的表达式是DMIN(database,fiel,criteria)。

【实际应用】

现在根据情况需要使用DMIN函数计算如图6.19所示的3种情况。

步骤1：输入条件区域。根据上面需要计算的情况输入条件区域，上面的3种情况包括了商标、产品、瓦数，因此，需要输入的条件区域如图6.20所示。

图6.19　需要计算的情况

图6.20　输入条件区域

步骤2：计算第一种情况。在单元格F20中输入公式“=DMIN(A1:H17,H1,A24:A25)”，该公式表示计算上海品牌的最小价值，计算结果如图6.21所示。

步骤3：计算后面的两种情况。在单元格F21中输入公式“=DMIN(A1:H17,H1,B24:C25)”，在单元格F22中输入公式“=DMIN(A1:H17,E1,A24:B25)”，得到的最后结果如图6.22所示。

图6.21　输入公式

图6.22　计算结果

6.1.6　DPRODUCT函数：返回指定数值的乘积

DPRODUCT函数的功能是返回数据库的列中满足指定条件的数值的乘积。

DPRODUCT函数的表达式是DPRODUCT(database,field,criteria)。

【实际应用】

由于在DPRODUCT函数中的参数一般是指同类数据的相乘，在实际运用中这些情况不是太多。下面举个简单的例子来说明DPRODUCT函数的使用情况，如图6.23所示。

图6.23　使用DPRODUCT函数

图中的单元格中用户计算了白炽灯产品的采购盒数的乘积，在单元格F20中输入公式“=DPRODUCT(A1:H17,G1,B22:B23)”，条件区域也相对比较简单，在B22:B23中输入相应的选择条件即可。

6.1.7　 DSTDEV函数：返回样本的标准偏差

DSTDEV函数的功能是将列表或数据库的列中满足指定条件的数字作为一个样本，估算样本总体的标准偏差。

DSTDEV函数的表达式是DSTDEV(database,field,criteria)。

【实际应用】

现在根据情况需要使用DSTDEV函数计算如图6.24所示的3种情况。

步骤1：输入条件区域。需要根据上面需要计算的情况输入条件区域，上面的3种情况包括了产品、瓦数，因此，需要输入的条件区域如图6.25所示。

图6.24　需要计算的情况

图6.25　输入条件区域

步骤2：计算第一种情况。在单元格F20中输入公式“=DSTDEV(A1:H17,E1,A24:A25)”，该公式表示计算白炽灯的单价的标准偏差，计算结果如图6.26所示。

步骤3：计算后面两种情况。在单元格F21中输入公式“=DSTDEV(A1:H17,G1,A24:

A25)”，在单元格F22中输入公式“=DSTDEV(A1:H17,B1,A24:B25)”，得到最后结果如图6.27所示。

图6.26　输入公式

图6.27　计算结果

6.1.8　DSTDEVP函数：返回总体的标准偏差

DSTDEVP函数的功能是将列表或数据库的列中满足指定条件的数字作为样本总体，计算总体的标准偏差。

DSTDEVP函数的表达式是DSTDEVP(database,field,criteria)。

【实际应用】

现在根据情况需要使用DSTDEVP函数计算如图6.28所示的3种情况。

步骤1：输入条件区域。用户需要根据上面需要计算的情况输入条件区域，上面的3种情况包括了产品、瓦数，因此，需要输入的条件区域如图6.29所示。

图6.28　需要计算的情况

图6.29　输入条件区域

步骤2：计算第一种情况。在单元格F20中输入公式“=DSTDEVP(A1:H17,E1,A24:A25)”，该公式表示计算白炽灯的单价的总体标准偏差，计算结果如图6.30所示。

步骤3：计算后面两种情况。在单元格F21中输入公式“=DSTDEVP(A1:H17,G1,A24:A25)”，在单元格F22中输入公式“=DSTDEVP(A1:H17,B1,A24:B25)”，得到的最后结果如图6.31所示。

图6.30　输入公式

图6.31　计算结果

6.1.9　DSUM函数：返回指定条件的数值之和

DSUM函数的功能是返回列表或数据库的列中满足指定条件的数值之和。

DSUM函数的表达式是DSUM(database,field,criteria)。

【实际应用】

现在根据情况需要使用DSUM函数计算如图6.32所示的3种情况。

步骤1：输入条件区域。用户需要根据上面需要计算的情况输入条件区域，上面的3种情况包括了瓦数、产品、商标。因此，需要输入的条件区域如图6.33所示。

图6.32　需要计算的情况

图6.33　输入条件区域

步骤2：计算第一种情况。在单元格F20中输入公式“=DSUM(A1:H17,H1,C24:C25)”，该公式表示计算推算商标为上海的总价值，计算结果如图6.34所示。

步骤3：计算后面两种情况。在单元格F21中输入公式“=DSUM(A1:H17,H1,B24:C25)”，在单元格F22中输入公式“=DSUM(A1:H17,H1,A24:B25)”，得到的最后结果如图6.35所示。

图6.34　输入公式

图6.35　计算结果

6.1.10　DVARP函数：计算总体的方差

DVARP函数的功能是将列表或数据库的列中满足指定条件的数字作为样本总体，计算总体的方差。

DVARP函数的表达式是DVARP(database,field,criteria)。

【实际应用】

现在根据情况需要使用DVARP函数计算如图6.36所示的3种情况，。

步骤1：输入条件区域。用户需要根据上面需要计算的情况输入条件区域，上面的3种情况包括了产品、瓦数，因此，需要输入的条件区域如图6.37所示。

图6.36　需要计算的情况


[image: ]



图6.37　输入条件区域

步骤2：计算第一种情况，在单元格F20中输入公式“=DVARP(A1:H17,E1,A24:A25)”，该公式表示计算白炽灯的单价的总体方差，计算结果如图6.38所示。

图6.38　输入公式

步骤3：计算后面两种情况。在单元格F21中输入公式“=DVARP(A1:H17,G1,A24:A25)”，在单元格F22中输入公式“=DVARP(A1:H17,B1,A24:B25)”，得到的最后结果如图6.39所示。

图6.39　计算结果

6.2　综合实例

前面一节已经介绍了关于数据库函数的基本内容，数据库函数的使用范围相对比较小，在本节中再介绍一个综合案例，复习上面讲过的函数。本节中选择的数据清单是关于住房的例子，如图6.40所示。

图6.40　数据清单

对于图中的数据清单，用户需要统计的内容有：

计算一二环路之间房屋每平米的平均房价。

计算住房面积大于100平米小于200平米的房屋数量。

计算没有游泳池的房屋数量。

计算住房面积大于100平米且有游泳池的房屋的最早交房时间。

计算二环路以外最低的房价。

针对上面需要统计的内容，下面按照步骤来进行统计。

步骤1：在单元格F14中输入公式“=DAVERAGE(B1:J11,H1,B20:B21)”，统计一二环路之间房屋每平米的平均房价，如图6.41所示。

图6.41　计算平均房价

步骤2：在单元格F15中输入公式“=DCOUNT(A1:I11,D1,B20:C21)”，统计住房面积大于100平米小于200平米的房屋数量，如图6.42所示。

图6.42　统计房屋数量

步骤3：在单元格F16中输入公式“=DCOUNT(A1:I11,D1,B20:B21)”，统计没有游泳池的房屋数量，如图6.43所示。

图6.43　统计没有游泳池的房屋数量

步骤4：在单元格F17中输入公式“=DMIN(A1:I11,C1,B20:C21)”，统计住房面积大于100平米且有游泳池的房屋的最早交房时间，如图6.44所示。

图6.44　统计最早交房时间

步骤5：在单元格F18中输入公式“=DMIN(A1:I12,I1,B20:B21)”，统计二环路以外最低的房价，如图6.45所示。

图6.45　统计最低房价


第7章 文本函数

文本函数，就是可以在公式中处理字符串的函数。例如，改变大小写或确定文本的长度、替换某些字符或者去除某些字符等。

根据文本函数的功能，处理文本的内容可以是截取文本中的某个特殊的字符，查询或者搜索文本中的某个特殊的字符，连接几个相关的文本，也可以是关于文本的其他信息，例如文本的代码编号、长度等。

根据上面的分类，这些函数有着不同的参数类型。处理文本基本信息的函数一般只有参数text，表示需要处理的文本对象。根据函数的名称，用户大致可以想到这些函数的功能，如图7.1所示。

除了上面处理文本基本信息的函数，还有一类文本函数，需要对文本进行截取、复制、替代、连接等。这些函数的参数除了text之外，一般还有参数表示进行截取、复制、替代的具体位置，如图7.2所示。有些函数会更加复杂，例如MID函数的表达式是MID(text,start_num,num_chars)，包含了起始数字和数字个数等number类型的数据。其他函数的特殊参数将会在后面的小节中详细介绍。

图7.1　使用UPPER函数


[image: ]



图7.2　使用LEFT函数

下面将分节介绍文本函数的各个主要函数的功能、表达式及参数，对每个函数都有举例说明。最后通过讲述两个实例，帮助读者进一步理解常用文本函数的使用方法。

7.1　函数应用实战

7.1.1　CHAR函数：返回代码的字符

CHAR函数的功能是返回对应于数字代码的字符。

CHAR函数的表达式是CHAR（number）。

只有一个参数number，是用于转换的字符代码，介于1~255之间。

【实际应用】

某公司在某个资料统计中，不同的数据列代表不同的部门，现在需要随时检查数据所在的列，也就是选择数据所在的部门，这个问题可以使用CHAR函数解决，通过返回各个单元格列的字符代码得到相应的列，如图7.3所示。

在单元格C14中输入公式“=CHAR(64+COLUMN())”。这个公式首先返回当前列的数值，然后加上64就可以得到当前数据列的列字符代码，然后使用CHAR函数返回相应的字符，得到相应的数据列字符。（在第5章中讲到COLUMN函数，表示返回列标号。）

【注意事项】

在CHAR函数中如果参数number大于255或者小于0，函数会返回错误值；如果参数是小数，将会被截尾取整，如图7.4所示。

图7.3　返回当前数据列

图7.4　函数的注意事项

7.1.2　CODE函数：返回字符对应的代码

CODE函数的功能是返回字符串中第一个对应字符的数字代码。

CODE函数的表达式是CODE(text)。

【实际应用】

公司的某个部门有一些员工姓名，现在需要判断员工是否姓“张”。这个问题需要用到CODE函数，通过CODE函数得到姓名的第一个字符的代码，判断是否和“张”的代码相同，如图7.5所示。

在单元格C2中输入公式“=IF(CODE(B2)=CODE("张"),"是","否")”，通过CODE函数比较了是否和“张”的代码相同，通过IF函数得到结果。

【注意事项】

在CODE函数中如果参数不是字符类型，那么函数会返回错误值，如图7.6所示。

图7.5　判断员工是否姓“张”

图7.6　函数的注意事项

7.1.3　CLEAN函数：返回不能打印的字符

CLEAN函数的功能是删除文本中不能打印的字符。

CLEAN函数的表达式是CLEAN(text)。

【实际应用】

对从其他应用程序中输入的文本，使用CLEAN函数可以删除其中无法打印的字符。例如，可以删除通常出现在数据文件头部或尾部、无法打印的计算机代码，如图7.7所示。

图7.7　使用CLEAN函数

7.1.4　CONCATENATE函数：合并字符串

CONCATENATE函数的功能是将几个字符串合并成一个字符串。

CONCATENATE函数的表达式是CONCATENATE（text1,text2,...）。

参数是可以选择的，而且个数需要在1~30之间，表示需要合并成单个文本项的文本项。

【实际应用】

某公司记录了某个月的销售情况，需要在某一列中将每个销售人员的记录输出，需要用到CONCATENATE函数，如图7.8所示。

图7.8　销售结果统计

步骤1：在单元格D2中输入公式“=CONCATENATE($A$1,A2,$B$1,B2,$C$1,C2)”，将相关单元格中的相关文字合并在一起，得到的是用户需要的结果。

步骤2：利用自动填充功能完成其他单元格的结果。

【注意事项】

CONCATENATE函数和运算符&的功能是相同的，用户可以使用运算符&完成文本合并，如图7.9所示。

图7.9　使用运算符

7.1.5　EXACT函数：比较字符串是否相同

EXACT函数的功能是测试两个字符串是否完全相同。如果它们完全相同，则返回TRUE；否则，返回FALSE。

EXACT函数的表达式是EXACT（text1,text2）。

函数EXACT能区分大小写，但忽略格式上的差异。利用函数EXACT可以测试输入文档内的文本。

【实际应用】

根据EXACT函数的功能，用户可以使用EXCAT函数来判断用户输入的内容是否正确，如图7.10所示。

图7.10　判断输入是否符合条件

步骤1：在单元格C2 中输入公式“=IF(EXACT(A2,B2),"是","否")”，可以判断单元格B2中的内容是否完全相等，然后用IF来返回逻辑判断。

步骤2：利用自动填充功能判断其他单元格的输入结果。

7.1.6　FIND和FINDB函数：查找文本字符串

FIND函数的功能是查找字符串内的子字符串，并将该子字符串首字符起始位置编号返回。FINDB函数的功能是查找字符串内的子字符串，并基于每个字符所使用的字节数将该子字符串首字符起始位置编号返回。

FIND函数的表达式是FIND(find_text,within_text,start_num)。

FINDB函数的表达式是FINDB(find_text,within_text,start_num)。

两个函数的区别在于FIND函数以字符为单位，FINDB函数以字节为单位。其中参数find_tex表示的是要查找的文本，within_text表示的是包含要查找文本的文本，start_num表示指定开始进行查找的字符。

【实际应用】

在一次数据游戏中，存在下面的规则：如果C列的数据为空，且B列的数据中含有1、2或者3，则D列中的数值为“B列数据×4-5”。如果不满足上面的要求，则在D列中返回空格，用户需要用Excel实现上面的游戏规则，如图7.11所示。

图7.11　Excel实现游戏规则

步骤1：在单元格D2中输入公式“=IF(C2="",IF(ISERROR(FIND(1,B2,1)>0),IF(ISERROR(FIND(2,B2,1)>0),IF(ISERROR(FIND(3,B2,1)),"",B2×4-5),B2×4-5),B2×4-5),"")”（ISERROR函数用于判断参数是否为错误值），通过FIND函数来判断输入数字中是否含有1、2或者3，然后通过IF函数得到第一个输入数字的游戏结果。

步骤2：利用自动填充功能完成其他输入数字的游戏结果。

7.1.7　LEFT和LEFTB函数：返回左边字符

LEFT函数的功能是基于所指定的字符数返回字符串中的第一个或前几个字符。LEFTB函数的功能是基于所指定的字节数返回字符串中的第一个或前几个字符。两个函数的区别在于LEFT函数以字符为单位，LEFTB函数以字节为单位。

LEFT函数的表达式是LEFT(text,num_chars)。

LEFTB函数的表达式是LEFTB(text,num_bytes)。

其中text表示包含要提取字符的字符串；num_chars表示要提取的字符数或字节数。

【实际应用】

在某个单位中，用户需要根据已知的信息得到相应的购买者的称呼，如图7.12所示。

图7.12　输入称呼

步骤1：在单元格D2 中输入公式“=CONCATENATE(C2,LEFT(A2,1),IF(B2="男","先生","女士"))”，通过LEFT提取购买者的姓，然后根据城市和性别来完成称呼。

步骤2：利用自动填充功能完成其他购买者的称呼。

7.1.8　LEN和LENB函数：返回字符串长度

LEN函数的功能是返回字符串中的字符数。LENB函数的功能是返回字符串中用于代表字符的字节数。两个函数的区别在于LEN函数以字符为单位，LENB函数以字节为单位。

LEN函数的表达式是LEN(text)。

LENB函数的表达式是LENB(text)。

【实际应用】

某个销售单位需要将每个员工的销售成绩按照百十元区分开来，由于每个员工的销售成绩都在三位数以下，完成这样的功能需要使用函数LEN和LEFT，如图7.13所示。

步骤1：返回百位数。在单元格C3 中输入公式“=IF(LEN(B3)=3,LEFT(B3,1),0)”，得到在单元格C3中的数据。这个公式首先判断在B3单元格中文本的位数，如果位数是3，那么选取其中的第一个字符，否则就返回0，如图7.14所示。

图7.13　原始数据

图7.14　得到每个销售成绩的百位数

步骤2：返回十位数。在单元格D3 中输入公式“=IF(C3=0,IF(LEN(B3)=2,LEFT(B3,1),0),LEFT(B3-C3×100,1))”，得到在单元格D3中的数据。这个公式首先判断在C3单元格中文本是否是0，如果不是0，表明B3单元格中是三位数，那么公式LEFT(B3-C3×100,1)返回十位数。如果是0，表示B3单元格中是两位数，那么公式IF(LEN(B3)=2,LEFT(B3,1),0)返回十位数，如图7.15所示。

步骤3：返回个位数。在单元格E3 中输入公式“=IF(C3=0,IF(LEN(C3)=1,LEFT(B3,1),LEFT(B3-D3×10,1)),LEFT(B3-C3×100-D3×10,1))”，得到在单元格E3中的数据。这个公式首先判断在C3单元格中文本是否是0，如果不是0，表明B3单元格中是三位数，那么公式LEFT(B3-C3×100-D3×10,1)返回个位数。如果是0，表示B3单元格中是两位数，那么公式IF(LEN(C3)=1,LEFT(B3,1),LEFT(B3-D3×10,1))返回个位数，如图7.16所示。

图7.15　得到每个销售成绩的十位数

图7.16　得到每个销售成绩的个位数

步骤4：利用Excel的自动填充功能完成其他单元格的结果，如图7.17所示。

图7.17　最后的结果

7.1.9　LOWER函数：将文本转换为小写

LOWER函数的功能是将一个字符串中的所有大写字母转换为小写字母。

LOWER函数的表达式是LOWER(text)。

【实际应用】

某公司的登录系统需要检查用户的密码，但是为了简便，将不区分用户密码中的大小写内容，因此可以使用LOWER函数将输入的密码转化为小写后再进行比较，如图7.18所示。

图7.18　验证过程

步骤1：在单元格C2中输入公式“=IF(EXACT(LOWER(B2),A2),"是","否")”，由于在数据库密码列中字母都是小写，而密码又不区别大小写，因此需要首先将所有输入格式的字母通过LOWER函数转化为小写，判断单元格B2中输入的数据是否通过验证。

步骤2：利用Excel的自动填充功能来检验其他输入数据是否通过验证。

7.1.10　MID和MIDB函数：返回特定字符

MID函数和MIDB函数的功能是返回字符串中从指定位置开始的特定数目的字符，该数目由用户指定。

两个函数的区别在于MID函数以字符为单位，MIDB函数以字节为单位。

MID函数的表达式是MID(text,start_num,num_chars)。

MIDB函数的表达式是MIDB(text,start_num,num_bytes)。

参数text 表示的是要提取字符的字符串。start_num 表示的是文本中要提取的字符串的开始提取位置。num_chars指定希望 MID 从文本中返回字符的个数。num_bytes指定返回字节的个数。

【实际应用】

这个函数的例子可以沿用上面在函数LEN中提到的例子，某个销售单位需要将每个员工的销售成绩按照百十元区分开来，如图7.19所示。

步骤1：返回百位数。在单元格C3 中输入公式“==IF(LEN(B3)=3,MID(B3,1,1),0)”，得到在单元格C3中的数据。这个公式首先判断在B3单元格中文本的位数，如果位数是3，那么选取其中的第一个字符，否则就返回0，如图7.20所示。

图7.19　原始数据

图7.20　得到每个销售成绩的百位数

步骤2：返回十位数。在单元格D3 中输入公式“=IF(C3=0,IF(LEN(B3)=2,MID(B3,1,1),0),MID(B3,2,1))”，得到在单元格D3中的数据。这个公式首先判断在C3单元格中文本是否是0，如果不是0，表明B3单元格中是三位数，那么公式MID(B3,2,1)返回十位数。如果是0，表示B3单元格中是两位数，那么公式IF(LEN(B3)=2,MID(B3,1,1),0)返回十位数，如图7.21所示。

步骤3：返回个位数。在单元格E3 中输入公式“=IF(C3=0,IF(LEN(C3)=1,MID(B3,1,1),MID(B3,2,1)),MID(B3,3,1))”，得到在单元格E3中的数据。这个公式首先判断在C3单元格中文本是否是0，如果不是0，表明B3单元格中是三位数，那么公式MID(B3,3,1)返回个位数。如果是0，表示B3单元格中是两位数，那么公式IF(LEN(C3)=1,MID(B3,1,1),MID(B3,2,1))返回个位数，如图7.22所示。

图7.21　得到每个销售成绩的十位数


[image: ]



图7.22　得到每个销售成绩的个位数

步骤4：利用 Excel的自动填充功能完成其他销售人员的金额，如图7.23所示。

图7.23　最后的结果

7.1.11　REPLACE和REPLACEB函数：替代文本

REPLACE函数和REPLACEB函数的功能是使用其他字符串并根据所指定的字符数替换某字符串中的部分文本。

两个函数的区别在于REPLACE函数以字符为单位，REPLACEB函数以字节为单位。

REPLACE函数的表达式是REPLACE(old_text,start_num,num_chars,new_text)。

REPLACEB函数的表达式是REPLACEB(old_text,start_num,num_bytes,new_text)。

参数old_text表示要替换部分字符的文本；start_num表示要用new_text替换的old_text中字符的位置；num_chars表示希望REPLACE使用new_text替换old_text中字符的个数；num_bytes表示希望REPLACE使用new_text替换old_text中字节的个数；new_text表示要用于替换old_text中字符的文本。

【实际应用】

某个区县的电话号码升级，升级的方法是在区号后加上“3”，原始数据如图7.24所示。

步骤１：在单元格C2中输入公式“=REPLACE(B2,1,3,"0273")”，得到在单元格C2中的数据。这个公式实现将B2单元格中的数据的第一位开始到第三位数据，替代为“0273”，如图7.25所示。

图7.24　原始数据

图7.25　替代数据

步骤2：利用自动填充功能完成其他单元格的数据，如图7.26所示。

图7.26 最后的结果

7.1.12　REPT函数：复制文本

REPT函数的功能是按照给定的次数重复显示文本。可以通过函数REPT来不断地重复显示某一字符串，对单元格进行填充。

REPT函数的表达式是REPT(text,number_times)。

参数text表示需要重复显示的文本；number_times指定文本重复次数的正数。

【实际应用】

某班统计了同学的考试成绩，现在需要根据成绩的高低来分配星号图标，规则如下：如果成绩小于60，没有图标；其余的情况下图标将会按照成绩增加30分则多一个图标。基本的成绩数据如图7.27所示。

步骤1：在单元格C2中输入公式“=IF(B2>60,REPT("☆",INT(B2/30)),"")”，得到在单元格C2中的图标，如图7.28所示。

步骤2：利用自动填充功能完成其他单元格的图标，如图7.29所示。

图7.27　原始数据

图7.28　计算图标

图7.29　最后的结果

7.1.13　RIGHT和RIGHTB函数：返回右边字符

RIGHT和RIGHTB函数的功能是根据指定的字符数或字节数返回字符串中最后一个或多个字符或字节。两个函数的区别在于RIGHT函数以字符为单位，RIGHTB函数以字节为单位。

RIGHT函数的表达式是RIGHT(text,num_chars)。

RIGHTB函数的表达式是RIGHTB(text,num_bytes)。

参数text 表示包含要提取字符的字符串；num_chars指定希望 RIGHT 提取的字符数。num_bytes指定提取的字节数。

【实际应用】

某抽奖公司随机抽奖，获得如图7.30所示的5个获奖消费者的身份证号码，需要判断获奖者的性别，其判断的标准是如果最后一位是奇数，则是男性；如果是偶数，则是女性。

步骤1：在单元格C2中输入公式“=IF(MOD(RIGHT(B2,1),2)=0,"女","男")”，得到在单元格C2中的性别，如图7.31所示。

图7.30　原始数据

图7.31　判断性别

步骤2：利用Excel中的自动填充功能完成其他消费者的性别，如图7.32所示。

图7.32　最后的结果

7.1.14　SEARCH和SEARCHB函数：查找文本字符串

SEARCH和SEARCHB函数的功能是返回从start_num开始首次找到特定字符或字符串的位置上特定字符的编号。两个函数的区别在于SEARCH函数以字符为单位，SEARCHB函数以字节为单位。

SEARCH函数的表达式是SEARCH(find_text,within_text,start_num)。

SEARCHB函数的表达式是SEARCHB(find_text,within_text,start_num)。

参数含义和函数FIND的含义相同。

【实际应用】

在某个银行中有下面的统计规则：如果“承兑行”一栏含有“黄冈”字样，则“实际天数”等于“天数”；如果“承兑行”一栏不含有“黄冈”字样，则“实际天数”等于“天数”加上3，基本数据如图7.33所示。

步骤1：在单元格C2中输入公式“=IF(ISERROR(SEARCH("黄冈",A2)),B2+3,B2)”（ISERROR函数用于判断参数是否为错误值），得到在单元格C2中的实际天数，如图7.34所示。

图7.33　原始数据

图7.34　计算实际天数

步骤2：利用自动填充功能完成其他承兑行的天数，如图7.35所示。

图7.35　最后的结果

7.1.15　SUBSTITUTE函数：替换文本

SEARCH和SEARCHB函数的功能是返回从start_num开始首次找到特定字符或字符串的位置上特定字符的编号。

SEARCH函数的表达式是SEARCH(find_text,within_text,start_num)。

SEARCHB函数的表达式是SEARCHB(find_text,within_text,start_num)。

参数和函数FIND的含义相同。

【实际应用】

用户需要统计某一段文字中包含有几个“的”字，用户可以使用SUBSTITUTE函数来完成这个工作，如图7.36所示。

图7.36　统计“的”字出现的次数

在单元格B3中输入公式“=LEN(A1)-LEN(SUBSTITUTE(A1,"的",""))”，在公式中首先使用SUBSTITUTE函数将所有的“的”字删除，然后将删除后的文本和原来的文本相减，得到“的”字出现的次数。

7.1.16　TEXT函数：将数值转换为文本

TEXT函数的功能是将数值转换为按指定数字格式表示的文本。

TEXT函数的表达式是TEXT(value,format_text)。

参数value 是数字，或对包含数字值的单元格的引用；Format_text是“单元格格式”对话框中“数字”选项卡上“分类”框中的文本形式的数字格式。

【实际应用】

用户需要将原始数据中的销售成绩转换为收入，用户可以使用TEXT函数，如图7.37所示。在单元格C2中输入公式“=TEXT(B2,"$0.00")”。这个公式将单元格中相应的数据文本转换为美元格式，然后利用自动填充功能完成其他单元格的格式转换，如图7.38所示。

图7.37　将销售成绩转换为工作收入

图7.38　转换格式

7.1.17　TRIM函数：清除文本中的空格

TRIM函数的功能是除了单词之间的单个空格外，清除文本中所有的空格。

TRIM函数的表达式是TRIM(text)。

【实际应用】

用户可以使用TRIM函数来清除文本中多余的空格。例如在图7.39所示的文本中在输入的过程中输入了多余的空格，可以使用TRIM函数来保证格式完整。

图7.39　删除多余的空格

TRIM函数只能删除多余的空格，但是不能删除所有的空格。在默认的情况下，TRIM函数中提供的函数存在一个空格。

7.1.18　UPPER函数：将文本转换为大写

UPPER函数的功能是将文本转换成大写形式。

UPPER函数的表达式是UPPER(text)。

【实际应用】

某个公司的登陆系统需要检查用户的密码，但是为了简便，将不区分用户密码中的大小写内容，因此可以使用UPPER函数将输入的密码转化为大写后再进行比较，如图7.40所示。

在单元格C2中输入公式“=IF(EXACT(UPPER(B2),A2),"是","否")”，由于在数据库密码列中字母都是大写，而密码又不区别大小写，因此需要首先将所有输入格式的字母通过LOWER函数转化为大写。然后通过自动填充功能来完成其他单元格的结果。

7.1.19　VALUE函数：将文本转换为数字

VALUE函数的功能是将代表数字的字符串转换成数字。

VALUE函数的表达式是VALUE(text)。

【实际应用】

通常不需要在公式中使用函数VALUE，Excel可以在需要时将文本转换为数字。提供此函数是为了与其他电子表格程序兼容。下面的例子简单的说明了函数的用法，如图7.41所示。

图7.40　验证过程

图7.41　使用VALUE函数

7.2　综合实例

文本函数可以用来处理各种综合问题，在实际生活中有很多的应用，在本节中将举出2个例子来演示如何使用文本函数。

7.2.1　处理数据格式

已知原始数据，主要是包括日期、收入等，但是由于是从外部获取的原始数据，所以出生日期的格式不统一，现在用户需要完成的工作是将日期格式统一成一个标准的格式，还有收入的数据格式等。原始的数据如图7.42所示。

步骤1：根据前面讲解过的方法，转换收入的数据格式。只需要在各个数据前面添加美元符号就可以完成，如图7.43所示。


[image: ]



图7.42　原始数据

图7.43　转换工资收入的格式

步骤2：转换出生日期的格式。根据需要，转换后的出生日期应该是“1958.05”，日期的常规类型。在上面的单元格F2中输入公式“=VALUE(IF(LEN(B2)>5,LEFT(B2,4)&"."&TEXT(RIGHT(B2,LEN(B2)-5),"00"),"19"&LEFT(B2,2)&"."&TEXT(RIGHT(B2,LEN(B2)-3),"00")))”，如图7.44所示。

图7.44　转换出生日期的格式

上面的公式比较复杂，下面需要详细的解释这个公式的各个部分的含义。首先用户需要判断原来的数据中的数据格式，可以通过长度来判断。如果日期格式不是19××格式，那么数据格式的长度最多为5，否则长度一定大于5。因此，可以通过LEN函数来判断数据长度是否大于5大致判断原来的数据格式。

处理其中的一种情况。如果长度大于5，那么就是19××的格式。公式可以选择其中的公式LEFT(B2,4)&"."&TEXT(RIGHT(B2,LEN(B2)-5),"00")得到想要的结果。其中公式TEXT(RIGHT(B2,LEN(B2)-5),"00")可以得到其中月份的两位格式，而前面的LEFT(B2,4)&"."将得到年份。

处理另外一种情况。如果长度小于5，那么年份一定是××格式，因此公式 "19"&LEFT(B2,2)&"."&TEXT(RIGHT(B2,LEN(B2)-3),"00")得到的是最后想要的结果。首先在年份前面加上19，然后选取TEXT(RIGHT(B2,LEN(B2)-3),"00")得到月份的两位格式。

步骤3：利用自动填充功能来完成其他数据表格。上面的公式可以利用自动填充功能来输入其他数据表格的内容，如图7.45所示。

图7.45　最后的结果

7.2.2　处理收款格式

某个公司的会计部门需要处理收款，将收款的数目大小转化为标准的会计收款单中的数据格式，需要得到的结果如图7.46所示。

图7.46　需要得到的效果

上面的结果可以根据收款的数目来决定金额符号"￥"的位置，在金额所在的位置之前的单元格则应该是空格，其他数字则填到相应的位置上。要达到这个效果用户需要为每一个金额位数填写相应的函数公式，下面分步详细介绍。

步骤1：确定“千万”位的数目。在单元格C2中输入公式“=IF(A2>=10000000,MID(RIGHTB(A2×100,10),1,1),IF(A2>=1000000,"￥",""))”。这里使用的函数RIGHTB选择的是数字的字节，因为数字是单字节的。首先判断A2中输入的金额数是否大于千万，然后将A2中输入的金额乘以100得到以“分”为单位的数目，使用MID选取其中的第一位数，就是千万的数字。

步骤2：确定“百万”位的数目。在单元格D2中输入公式“=IF(A2>=1000000,MID(RIGHTB(A2×100,9),1,1),IF(A2>=100000,"￥",""))”。这里使用的函数RIGHTB选择输入数额中百万以后的数据。IF(A2>=100000,"￥","")判断了数字是否大于十万，如果是大于十万而小于百万，那么在百万的数据格中显示的是金额的符号"￥"。

步骤3：确定“十万”位的数目。在单元格E2中输入公式“=IF(A2>=100000,MID(RIGHTB(A2×100,8),1,1),IF(A2>=10000,"￥",""))”。这个公式的原理和上面所讲的相同。

步骤4：确定“万”位的数目。在单元格F2中输入公式“=IF(A2>=10000,MID(RIGHTB(A2×100,7),1,1),IF(A2>=1000,"￥",""))”。

步骤5：确定“千”位的数目。在单元格G2中输入公式“=IF(A2>=1000,MID(RIGHTB(A2×100,6),1,1),IF(A2>=100,"￥",""))”。

步骤6：确定“百”位的数目。在单元格H2中输入公式“=IF(A2>=100,MID(RIGHTB(A2×100,5),1,1),IF(A2>=10,"￥",""))”。

步骤7：确定“十”位的数目。在单元格I2中输入公式“=IF(A2>=10,MID(RIGHTB(A2×100,4),1,1),IF(A2>=1,"￥",""))”。

步骤8：确定“元”位的数目。在单元格J2中输入公式“=IF(A2>=1,MID(RIGHTB(A2×100,3),1,1),IF(A2>=0.1,"￥",""))”。

步骤9：确定“角”位的数目。在单元格K2中输入公“=IF(A2>=0.1,MID(RIGHTB(A2×100,2),1,1),IF(A2>=0.01,"￥",""))”。

步骤10：确定“分”位的数目。在单元格L2中输入公式“=IF(A2>=0.1,MID(RIGHTB(A2×100,2),1,1),IF(A2>=0.01,"￥",""))”。

步骤11：利用自动填充功能完成所有的单元格的公式。用户可以在所需要的单元格的区域中填充相应的公式，如图7.47所示。

图7.47　填充公式


第8章 日期和时间函数

Excel主要提供了14个相关日期和时间的函数。在本章中，将会详细讲解它们各自的功能、表达式和参数，并且结合案例进行说明。

8.1　日期系统介绍

在使用日期和时间函数之前，用户有必要了解Excel中的日期系统。Excel实际支持两种日期系统，在不同的日期系统中，相同日期的处理结果是不同的。这两个日期系统分别是1900年日期系统和1904年日期系统，工作表中的日期和时间函数（主要是函数的参数）都将分别以不同的日期作为参照基础。1900系统中1900年的1月1号为第一天，在日期系列编号中为1；类似的，在1904系统中1904年的1月1号为第一天，在日期系列中编号为1。两种日期系统的最后一天，也就是日期序列的最大值都是9999年12月31号。

用户可以根据需要在两种日期系统中进行切换，具体的操作如下：

步骤1：单击【工具】|【选项】命令，打开【选项】对话框。

步骤2：选择【重新计算】选项卡，在【工作薄选项】选项框中选中【1904年日期系统】复选框，将系统时间设置为“1904年日期系统”，如图8.1所示。

图8.1　设置日期系统

8.2　函数应用实战

8.2.1　DATE函数：返回日期编号

DATE函数的功能是返回特定日期的系列编号。这个编号是指计算的日期相对于起始日期的天数值，例如在1900年日期系统中，DATE函数计算的就是该日期与1900年1月1日之间的天数，然后加上1。

DATE函数的表达式是DATE(year,month,day)。

该函数一共有三个参数，依次分别代表年、月、日。

【实际应用】

某超市在2004年的上半年整理了一份货物上架和下架的日期时间表，现在超市的经理需要知道各个货物在架的时间长度，可以使用DATE函数，如图8.2所示。

图8.2　使用DATE函数

步骤1：在单元格I3中输入公式“=DATE(E3,F3,G3)-DATE(B3,C3,D3)”。这个公式首先使用DATE(E3,F3,G3)计算出下架时间的日期编号，然后使用DATE(B3,C3,D3)计算出上架时间的日期编号，两者的差值就是该货物的在架天数。

步骤2：计算其他货物的在架天数。用户可以将上面的公式分别复制到单元格I4到单元格I9中，就可以得到其他各个货物的在架天数。

上面的计算结果与用户选择的日期系统无关，因为这个例子计算的是相对的差值，与基准的日期没有关系。

【注意事项】

用户在使用DATE函数的时候，需要注意几个问题。

单元格的数字格式会影响返回值。如果函数所在的单元格的数字格式是“日期”，那么DATE函数返回的是对应的日期，而不是日期对应的序列编号。

日期系统的上限是9999年，如果用户使用函数过程中出现了函数参数的溢出，则会返回错误值。

如果用户使用的1900系统，那么如果年份的参数值在0~1899之间时，系统会自动在参数值上加上1900，如图8.3所示。

图8.3　使用DATE函数的注意事项

用户需要特别注意，在月份和日期中年份溢出问题。如果用户在使用函数时，月份大于12那么Excel会从year参数所在的下一年的1月份往上加，推算出相应的确切月份；如果月份等于或小于0，会从year参数所在的上一年的12月开始往上减，推算出确切月份。同理，用户在使用日期的参数时，也会有类似的情况。如果年月日都出现溢出的情况，那么日期的推算会先从年开始，其次是月份，最后推算日。

8.2.2　DATEVALUE函数：返回日期编号

DATEVALUE函数的功能和上面的DATE函数类似，也是计算某个日期的系列编号。它和DATE函数的主要区别在于参数的类型和个数上。

DATEVALUE函数的表达式是DATEVALUE(date_text)。

其中唯一的一个参数date_text是一个字符串，其中的内容是日期，只要是Excel中可以识别的日期格式就可以当作其中的参数。

【实际应用】

某基金管理公司检测不同的基金管理人员的业绩，使用的标准是投资回报率，由于各个基金管理人员的投资日期不同，因此，需要使用DATEVALUE函数计算各自的投资天数，如图8.4所示。

图8.4　使用DATEVALUE函数

在上面的例子中，在单元格H2中输入公式：=DATEVALUE("2000-3-20")-DATEVALUE("1998-2-15")。这个公式首先计算回收金额的日期编号，然后计算出投资的日期编号，两者的差值就是投资的周期。从这个公式中，可以看到DATEVALUE函数的使用方法，用户必须直接引用这些日期，而不能使用单元格的引用。

【注意事项】

使用DATEVALUE函数似乎比DATE函数复杂，但是DATEVALUE函数有一个十分重要的优点，就是他的参数的个数比较少，而且范围广泛。只要是Excel可以识别的日期格式都可以作为DATEVALUE函数的参数，如图8.5所示。

图8.5　DATEVALUE函数的适用范围

上面的例子只是简单列举了一些常见的日期格式，用户在使用Excel的过程中也许会遇到其他不同类型的日期格式，只要这些格式可以被Excel识别，就可以在DATEVALUE函数中正确使用，由于这些优点，DATEVALUE才被广泛使用。

8.2.3　DAY函数：返回日期天数

DAY函数返回一个序列数所代表的日期在当月的天数，只有一个参数，格式是数值或者日期格式。

DAY函数的表达式是DAY(serial_number)。

【实际应用】

在某个写字楼中，有个检查房租的系统，可以检查某个日期是否有人交了房屋的租金，在这个程序中可以用到DAY函数。

步骤1：判定每个客户的交租的日期。在单元格C2中输入公式“=DAY(B2)”，可以提取交租时间中的日期，然后可以自动填充到C6单元格中，分别得到各个客户的交租日期，如图8.6所示。

步骤2：判定某天是否有人交租。在单元格E6中输入下面的公式：“=IF(ISERROR(MATCH(E4,C2:C6,0)),"否","是")”（ISERROR函数用于判断参数是否为错误值）。这个公式中使用到的函数将在后面介绍到，功能是判断在E4中输入的日期是否有人交租，上面的查询结果如图8.7所示。

图8.6　使用DAY函数

图8.7　得到查询的结果

上面的例子中输入的查询日期是15，有客户在15号这天交租，因此返回的结果“是”。

【注意事项】

DAY函数的参数除了是日期格式，还可以是数值格式，这个时候数值格式的参数代表的是某个日期在时间系统中的序列号（类似于DATE函数的返回值），DAY函数的参数说明如图8.8所示。

图8.8　DAY函数的参数使用

上面的例子中38464是2005-4-22这个日期在1900年时间系统中的序列号，也就是DATE(2005,4,22)的返回值。可见，图中的三个参数数值可以得到相同的结果。

8.2.4　DAYS360函数：返回相差天数

DAYS360函数的功能是按照一年360天的计算方法计算，得到两日期相差的天数。由于在一些投资或者合约中一年360天是一个广泛适用的约定，利用这个约定可以使很多步骤得到简化。

DAYS360函数的表达式为DAYS360(start_date,end_date,method)。

其中的start_date表示的是起始日期，end_date表示的是终止日期，method表示的是采用的计算方法。

【实际应用】

本例中，继续使用8.2.2中的基金管理公司的案例，但在计算各个基金管理人投资周期时用的是DAYS360函数，如图8.9所示。

图8.9　使用DAYS360函数

在上面的表格中的单元格G2中输入公式“=DAYS360（D2，E2）”。

按照一年360天来计算的数值和按照当年真实的天数相比投资周期会偏少，具体在图中的表格中F列和G列中的数据对比可以看出。

【注意事项】

DAYS360函数的参数中method表示的是函数的计算方法，主要可以选择美国方法和欧洲方法。

method参数如果是FALSE或者省略，那么选择的计算方法就是美国方法。在这个方法中，如果起始日期是一个月的31号，则起始日期等于同月的30号。如果终止日期是一个月的31号，并且起始日期早于30号，则终止日期等于下一月的1号，否则，终止日期等于本月的30号。

method参数如果是TRUE，那么选择的计算方法就是欧洲方法。在这个方法中，如果起始日期和终止日期是一个月的31号，则起始日期和终止日期都等于本月的30号。

8.2.5　HOUR函数：返回小时数

HOUR函数的功能是计算某个时间值或者代表时间的序列编号对应的小时数。

HOUR函数的表达式为HOUR(serial_number)。

这个函数只有一个参数，表示需要计算小时数的时间，这个参数的数据格式可以是所有Excel可以识别的时间格式。

【实际应用】

某个小型的停车场需要计算每辆车的停车时间，根据每辆车进入停车场和离开停车场的时间差来计算小时数，在这个案例中首先计算每辆车停车的小时数，后面讲到其他的函数还可以计算停车的分钟数，如图8.10所示。

步骤1：在单元格D3中输入公式“=HOUR（C3-B3）”，在这个函数中首先计算出相差的时间，然后用HOUR函数提取时间的小时数，得到的结果如图8.10所示。

步骤2：利用Excel的自动填充功能，得出其他车的停车时间，得到的结果如图8.10所示。

【注意事项】

HOUR函数的参数可以是各个形式的时间格式，但是格式的不同将会得到不同的结果,图8.11列举了一些常见的时间格式及结果。

需要注意的是在输入12小时制的时候，在AM或者PM之前应该加上空格，否则Excel将不能识别为时间格式。最后一个函数中的参数为十进制的数据，代表的是当时时间在24小时内的序列数。

图8.10　使用HOUR函数

图8.11　HOUR函数的应用

8.2.6　MINUTE函数：返回分钟数

MINUTE函数的功能是计算某个时间值对应的分钟数。

MINUTE函数的表达式为MINUTE(serial_number)。

【实际应用】

在停车场的例子中，除了要计算时间数之外，还需要计算分钟数，因此，这个例子接着完成上面停车的分钟数计算，如图8.12所示。

步骤1：在单元格E3中输入公式“=MINUTE（C3-B3）”，计算第一个车子停车的时间，在这个函数中首先计算出相差的时间，然后用MINUTE函数提取时间的分钟数，最后得到的结果如图8.12所示。

步骤2：利用Excel的自动填充功能，计算其他车子的停车时间的分钟数，得到的结果如图8.13所示。

图8.12　输入函数公式

图8.13　计算分钟数的结果

【注意事项】

MINUTE函数的参数和HOUR函数的参数类似，可以是各个形式的时间格式，图8.14列举了一些常见的时间格式及相应的结果。

图8.14　MINUTE函数的应用

HOUR函数中12小时和24小时制的时间会得到不同的结果，但是MINUTE函数却没有这样的不同，分钟数的函数MINUTE只是关注后面的分钟数，因此上面的“2:30 AM”和“2:30 PM”得到相同的结果。

8.2.7　MONTH函数：返回月份

MONTH函数的功能是计算某个日期代表的月份，该函数只有一个参数，表示需要计算月份的日期。

MONTH函数的表达式为MONTH(serial_number)。

【实际应用】

某超市统计了2004年该超市举行过促销活动的产品名称，为了明年的促销计划，该超市需要统计促销活动的旺季和淡季，具体的数据如图8.15所示。下面分步骤详细介绍统计的过程。

步骤1：统计促销月份。在单元格C2中输入公式“=MONTH（B2）”，就可以得到A产品的促销月份，然后利用自动填充功能计算其他产品的促销月份，如图8.15所示。

步骤2：统计促销产品最多的月份。在单元格E6中输入公式“=MODE(C2:C8)”，得到促销产品最多的月份。关于MODE函数已在第4章讲解，最后得到的结果如图8.16所示。

图8.16　使用MONTH函数

图8.16　最后的统计结果

8.2.8　NOW函数：返回当前时间

NOW函数的功能是返回计算机系统内部时钟的当前日期和时间，这个函数没有参数。

NOW函数的表达式为NOW()。

【实际应用】

NOW函数可以动态地统计时间，是随时更新的，返回的是当前的时间。下面的例子中将介绍某个实验如何随时了解实验已经经历的时间数。

步骤1：返回当前的时间。在单元格B2中输入公式“=NOW()”，就可以返回当前的系统时间，如图8.17所示。

步骤2：计算实验经历的时间。在单元格C2中输入公式“=TIME(HOUR(B2-A2),MINUTE(B2-A2),SECOND(B2-A2))”，得到实验经历的时间。TIME函数将在本章的后面内容中讲解，最后得到的结果如图8.18所示。


[image: ]



图8.17　使用NOW函数

图8.18　最后的计算结果

【注意事项】

NOW函数中的返回值与单元格的数据格式有关，常见的有自定义格式和常规格式，返回的都是当前时间或数值，如图8.19所示。

图8.19　NOW函数的注意事项

8.2.9　SECOND函数：返回秒数

SECOND函数的功能是计算某一个时间值的秒数。

SECOND函数的表达式为SECOND(serial_number)。

该函数的参数表示需要计算秒数的时间，参数的数据格式可以是所有Excel可以识别的时间格式。

【实际应用】

继续完成上面讲到的停车场的完整的时间统计系统，如图8.20所示。

图8.20　停车场的时间计算

步骤1：统计A3单元格的汽车的停车时间。在单元格F3中输入公式“=SECOND(C3-B3)”，首先计算出结束停车和开始停车相差的时间，然后用SECOND函数提取时间的秒数。

步骤2：计算其他汽车的停车时间。根据Excel的自动填充功能，得出其他单元格的汽车停车时间的秒数。

8.2.10　TIME函数：返回时间

TIME函数的功能是将指定的时、分、秒合成为时间。

TIME函数的表达式为TIME(hour,minute,second)。

该函数有三个参数，hour表示小时、minute表示分、second表示秒。如果参数minute和参数second的数值大于59，TIME函数将其转换成适当的时或分。

【实际应用】

在NOW函数的应用中，已经使用到TIME函数，也就是在如图8.21所示的表格中的单元格C2中输入公式“=TIME(HOUR(B2-A2),MINUTE(B2-A2),SECOND(B2-A2))”，可以返回经历的时间。

这个函数中首先利用HOUR、MINUTE、SECOND函数计算出相差时间的小时数、分钟数、秒数，然后用TIME函数合成时间。

图8.21　合成实验的时间

8.2.11　TIMEVALUE函数：返回时间

TIMEVALUE函数的功能是将以字符串表示的时间，转化为该时间的序列数字，只有一个参数，代表以Excel的时间格式表示的时间。

TIMEVALUE函数的表达式为TIMEVALUE(time_text)。

【注意事项】

TIMEVALUE函数返回由文本字符串所代表的时间小数值。小数值为0~0.999999999之间的数值，代表从0:00:00 (12:00:00 AM)到23:59:59 (11:59:59 PM)之间的时间。而且参数time_text中的日期信息将被忽略，如图8.22所示。

图8.22　TIMEVALUE函数的应用

8.2.12　TODAY函数：返回当前日期

TODAY函数的功能是返回当前日期的序列号，没有参数。

TODAY函数的表达式为TODAY()。

【实际应用】

某商店的产品管理系统，统计部分产品的生产时间、保质期等，判断当前的时间该产品是否过期，具体的数据如图8.23所示。

图8.23　原始数据

步骤1：计算当前日期。在单元格C2中输入公式“=TODAY()”，得到产品的生产日期，然后利用自动填充来统计其他产品的生产日期，如图8.24所示。

步骤2：计算产品的过期时间。在单元格D2中输入公式“=DATE(YEAR(A2),MONTH(A2),DAY(A2))+366×B2”。这个公式首先计算生产日期的序列号，然后加上保质期，就可以得到过期的时间。然后利用自动填充功能，得出其他单元格中产品的过期时间，如图8.25所示。

图8.24　计算当前的日期


[image: ]



图8.25　计算过期的日期

步骤3：判断是否过期。在单元格E2中输入公式：“=IF(DATE(YEAR(D2),MONTH(D2),DAY(D2))>DATE(YEAR(C2),MONTH(C2),DAY(C2)),"否","是")。这个公式比较了当前时间和过期时间的时间序列数，来判别是否过期，得到产品1是否过期。然后利用自动填充功能，判断其他单元格中产品是否过期，得到的结果如图8.26所示。

图8.26　判断是否过期

8.2.13　WEEKDAY函数：返回星期数

WEEKDAY函数的功能是返回某个日期的星期数。

WEEKDAY函数的表达式为WEEKDAY(serial_number,return_type)。

serial_number代表要查找哪一天的日期，return_type是确定返回值类型的数字。

【实际应用】

WEEKDAY函数可以确定已知日期或者可以推算日期的星期数，最常见的例子就是求某个人的生日的星期数，下面分步骤来详细介绍使用的方法。

步骤1：计算当前的日期。在单元格B2中输入公式“=WEEKDAY(A2,1))”，可以得到出生日期的星期数，如图8.27所示。

步骤2：计算相应的星期数。在单元格B5中输入公式“=WEEKDAY(DATE(B4,11,14))”。这个公式首先使用DATE函数计算出当前的日期，然后利用WEEKDAY函数来确定相应的星期数，如图8.28所示。

图8.27　计算当前的日期

图8.28　计算相应的星期数

【注意事项】

WEEKDAY函数的return_type参数，表示的是返回值的类型，可以取1、2、3。

参数值是1或者省略，那么返回数字1代表星期日，数字7代表星期六。

参数值是2，那么返回的数字1代表星期一，数字7代表星期日。

参数值是3，那么返回的数字0代表星期一，数字6代表星期日。

用户如果不习惯返回的是数字，那么需要在单元格的数字格式中设置为“日期”中的“星期三”格式，下面详细介绍设置过程。

步骤1：先打开【单元格格式】对话框。右击日期数据的单元格，在快捷菜单中选择【设置单元格格式】命令，弹出【单元格格式】对话框。

步骤2：设置数据格式。选择【数字】选项卡，在【类型】选项框中选择【星期三】选项。单击【确定】按钮，完成设置，如图8.29所示。

图8.29　设置单元格的数据格式

8.2.14　YEAR函数：返回年份

YEAR函数的功能是计算日期所代表的相应的年份，只有一个参数serial_number，表示将要计算年份的日期。

YEAR函数的表达式为YEAR(serial_number)。

【实际应用】

某个水文观测站统计了近年发生洪峰的日期，现在需要统计出各个洪峰的年份，这时需要用YEAR函数，具体数据如图8.30所示。

图8.30　使用YEAR函数

在单元格B2中输入公式“=YEAR(A2)”，提取前面的洪峰时间中的年份，然后使用Excel的自动填充功能得到其他的洪峰年份。

8.3　综合实例

本节利用3个实例介绍如何使用日期和时间函数，来处理日常生活中一些常见的问题。

8.3.1　计算节假日

某些西方的节日没有固定的日期，解决这些日期会比较复杂。下面具体计算出如图8.31所示的节假日的日期和星期数。

步骤1：确定计算节假日的年份。图8.31中的节假日主要分成两类，一类是日期固定的节假日，另一类是不固定的节假日。用户需要计算各自的具体日期和星期数。首先需要在单元格B1中输入年份，在本例中输入的年份是2005，如图8.32所示。

图8.31　需要计算的节假日


[image: ]



图8.32　确定节假日的年份

步骤2：计算情人节的具体日期和星期数。在单元格E4中输入公式“=DATE($B$1,C4,D4)”，在单元格F4中输入公式“=WEEKDAY(E4,1)”，这样就计算出2005年的情人节的具体日期和星期数。在这里需要提示用户的是，在E4单元格中输入的公式当中，B1的引用必须是绝对引用，因为在下面的步骤中需要利用Excel中的自动填充的功能计算其他的3个时间的日期，如图8.33所示。

图8.33　计算情人节的日期和星期数

步骤3：计算其他固定日期节假日的日期和星期数。利用自动填充功能来计算其他固定日期的具体日期和星期数，得到的结果如图8.34所示。

图8.34　计算其他固定日期的具体日期和星期数

步骤4：计算西方的劳动节的具体日期。由于西方的劳动节是9月第一个星期一，具体计算其日期的公式比较复杂，在单元格E8中输入公式：“=DATE(B1,9,1)+IF(2<WEEKDAY(DATE(B1,9,1)),7-WEEKDAY(DATE(B1,9,1))+2,2-WEEKDAY(DATE(B1,9,1)))”，得到的结果如图8.35所示。

图8.35　计算劳动节的具体日期

上面公式中的第一部分DATE(B1,9,1)表示的是9月1号，劳动节是在9月份，那么肯定大于或者等于9月1号。后面的逻辑表达式主要是讨论了下面的两种情况：如果2005年9月1号的星期数大于2，也就是9月1号大于星期一，那么9月的第一个星期一离9月1号的天数为7-WEEKDAY(DATE(B1,9,1))+2；如果9月1号小于星期一，那么9月第一个星期一离9月1号的天数为2-WEEKDAY(DATE(B1,9,1))。

从上面的分析可以看出，整个IF函数返回的是2005年劳动节距离2005年9月1号的天数，因此，在原来的9月1号的基础上加上多出的天数就是2005年的劳动节的具体日期。

步骤5：计算母亲节的具体日期。由于母亲节是5月第二个星期日，用户可以在单元格E9中输入公式：“=DATE(B1,5,1)+IF(1<WEEKDAY(DATE(B1,5,1)),7-WEEKDAY(DATE(B1,5,1))+1+(2-1)×7,1-WEEKDAY(DATE(B1,5,1))+(2-1)×7)”，得到的结果如图8.36所示。

图8.36　计算母亲节的具体日期

上面公式中的第一部分DATE(B1,5,1)表示的是5月1号，母亲节是在5月份，那么肯定大于或者等于5月1号。后面的逻辑表达式主要是讨论了下面的两种情况：如果2005年5月1号的星期数大于1，也就是5月1号大于星期日，那么5月的第二个星期日离5月1号的天数为7-WEEKDAY(DATE(B1,5,1))+1+(2-1)×7；如果5月1号小于星期日，那么5月第二个星期日离5月1号的天数为1-WEEKDAY(DATE(B1,5,1))+(2-1)×7。

从上面的分析可以看出，整个IF函数返回的是2005年母亲节距离2005年5月1号的天数，因此，在原来的5月1号的基础上加上多出的天数就是2005年的母亲节的具体日期。

步骤6：计算父亲节的具体日期。由于父亲节是6月第三个星期日，用户可以在单元格E10中输入公式：“=DATE(B1,6,1)+IF(1<WEEKDAY(DATE(B1,6,1)),7-WEEKDAY(DATE(B1,6,1))+1+(3-1)×7,1-WEEKDAY(DATE(B1,6,1))+(3-1)×7)”，得到的结果如图8.37所示。

图8.37　计算父亲节的具体日期

上面公式中的第一部分DATE(B1,6,1)表示的是6月1号，父亲节是在6月份，那么肯定大于或者等于6月1号。后面的逻辑表达式主要是讨论了下面的两种情况：如果2005年6月1号的星期数大于1，也就是6月1号大于星期日，那么6月的第三个星期日离6月1号的天数为7-WEEKDAY(DATE(B1,6,1))+1+(3-1)×7；如果6月1号小于星期日，那么6月第三个星期日离6月1号的天数为1-WEEKDAY(DATE(B1,6,1))+(3-1)×7。

从上面的分析可以看出，整个IF函数返回的是2005年父亲节距离2005年6月1号的天数，因此，在原来的6月1号的基础上加上多出的天数就是2005年的父亲节的具体日期。

步骤7：计算感恩节的具体日期，由于感恩节是11月第四个星期四，用户可以在单元格E11中输入公式：“=DATE(B1,11,1)+IF(5<WEEKDAY(DATE(B1,11,1)),7-WEEKDAY(DATE(B1,11,1))+5+(4-1)×7,5-WEEKDAY(DATE(B1,11,1))+(4-1)×7)”，得到的结果如图8.38所示。

图8.38　计算感恩节的具体日期

上面公式中的第一部分DATE(B1,11,1)表示的是11月1号，父亲节是在11月份，那么肯定大于或者等于11月1号。后面的逻辑表达式主要是讨论了下面的两种情况：如果2005年11月1号的星期数大于5，也就是11月1号大于星期四，那么11月的第四个星期四离11月1号的天数为7-WEEKDAY(DATE(B1,11,1))+5+(4-1)×7；如果11月1号小于星期四，那么11月第四个星期四离11月1号的天数为5-WEEKDAY(DATE(B1,11,1))+(4-1)×7。

从上面的分析可以看出，整个IF函数返回的是2005年感恩节距离2005年11月1号的天数，因此，在原来的11月1号的基础上加上多出的天数就是2005年的感恩节的具体日期。

步骤8：计算日期的星期数。前面的步骤已经将各个节假日的具体日期计算出来，下面对于E列的时间在第F列计算出相应的星期数，这个步骤十分简单，只要使用自动填充功能就可以了，如图8.39所示。

图8.39　通过自动填充功能计算星期数

步骤9：改换年份计算日期和星期数。用户可以将B1单元格中的年份改成2006，看看结果有什么不同，如图8.40所示。


[image: ]



图8.40　2006年节假日的日期和星期数

8.3.2　网吧收费系统

网吧按照小时数进行计费，需要计算每个上网用户的小时数。首先需要计算的是，上网时间小于30分钟的按照0.5小时计算；超过了30分钟按照1小时计算。然后，根据有关规定，不允许用户连续上网超过2天(48小时)，如果时间超过了2天，应该进行超时警告。

在这个例子中，主要的计算数据如图8.41所示，需要计算出相应的上网时间，最后计算出相应的上网费用。

图8.41　各个客户的上网时间数据

步骤1：计算第一位顾客上网的天数。用户可以使用两种方法来计算上网的天数，第一种方法在单元格D3中输入公式“=DAY(C3-B3)”，该公式将C列和B列的数据相减，然后用DAY函数计算天数，如图8.42所示。

图8.42　计算上网的天数

另外的一个方法比较复杂，用户可以在D3单元格中输入下面的公式：

=IF(TIME(HOUR(B3),MINUTE(B3),SECOND(B3))>TIME(HOUR(C3),MINUTE(C3),SECOND(C3)),DATE(YEAR(C3),MONTH(C3),DAY(C3))-DATE(YEAR(B3),MONTH(B3),DAY(B3))-1,DATE(YEAR(C3),MONTH(C3),DAY(C3))-DATE(YEAR(B3),MONTH(B3),DAY(B3)))。

上面的公式首先比较了下网时间是否超过了上网的时间，如果没有超过，那么上网的天数应该是“DATE(YEAR(C3),MONTH(C3),DAY(C3))-DATE(YEAR(B3),MONTH(B3),DAY(B3))-1”；如果超过了那么上网的天数应该是：“DATE(YEAR(C3),MONTH(C3),DAY(C3))-DATE(YEAR(B3),MONTH(B3),DAY(B3))”。这个方法虽然比较复杂，但是道理比较通顺，这个方法可以得到和上面相同的答案，如图8.43所示。

图8.43　计算上网的天数

步骤2：计算第一位顾客上网的小时数。用户同样可以使用两种方法来统计上网的小时数，用户可以在单元格E3中输入公式“=HOUR(C3-B3)”，如图8.44所示。

图8.44　计算上网的小时数

跟前面计算天数类似，也可以在单元格E3中输入如下的公式：=IF(TIME(HOUR(B3),MINUTE(B3),SECOND(B3))>TIME(HOUR(C3),MINUTE(C3),SECOND(C3)),HOUR(1-TIME(HOUR(B3),MINUTE(B3),SECOND(B3))+TIME(HOUR(C3),MINUTE(C3),SECOND(C3))),HOUR(TIME(HOUR(B3),MINUTE(B3),SECOND(B3))-TIME(HOUR(C3),MINUTE(C3),SECOND(C3))))。

上面的公式也是首先比较了下网时间是否超过了上网的时间，如果没有超过，那么第一天上网的时间序列数应该是1-TIME(HOUR(B3),MINUTE(B3),SECOND(B3))；下网的时间序列数是“TIME(HOUR(C3),MINUTE(C3),SECOND(C3))”。如果时间超过了，那么相差的时间序列数为“TIME(HOUR(B3),MINUTE(B3),SECOND(B3))-TIME(HOUR(C3),MINUTE(C3),SECOND(C3))”。这个方法可以得到和上面相同的答案，如图8.45所示。

图8.45　计算上网的小时数

步骤3：计算第一位顾客上网的分钟数。用户同样可以使用两种方法来统计上网的小时数，用户可以在单元格E3中输入公式“=MINUTE(C3-B3)”，如图8.46所示。

图8.46　计算上网的分钟数

用户也可以在单元格F3中输入下面的公式：

=IF(TIME(HOUR(B3),MINUTE(B3),SECOND(B3))>TIME(HOUR(C3),MINUTE(C3),SECOND(C3)),MINUTE(1-TIME(HOUR(B3),MINUTE(B3),SECOND(B3))+TIME(HOUR(C3),MINUTE(C3),SECOND(C3))),MINUTE(TIME(HOUR(B3),MINUTE(B3),SECOND(B3))-TIME(HOUR(C3),MINUTE(C3),SECOND(C3))))。

上面的公式和计算小时数的公式类似，只是在最后使用的是MINUTE函数，如图8.47所示。

图8.47　计算上网的分钟数

步骤4：计算其他顾客上网的天数、小时数和分钟数。利用Excel中的自动填充功能，用户可以用上面的任意一个方法来计算每个卡号的上网时间的天数、小时数和分钟数，得到的结果如图8.48所示。

图8.48　计算各个卡号的上网时间

步骤5：计算第一位顾客上网时间的累积小时数。这个算法比较简单，根据前面对小时数的设置，可以在单元格G3输入公式：“=D3×24+E3+IF(F3=0,0,IF(F3<=30,0.5,1))”，上面的公式中首先判断分钟数是否超过0.5小时，来计算累计小时数，得到结果如图8.49所示。

图8.49　计算第一个卡号的上网累计小时数

步骤6：计算其他顾客的累计小时数。可以使用Excel的自动填充功能来统计其他上网卡号的累计小时数，得到的结果如图8.50所示。

图8.50　计算其他卡号的累计小时数

步骤7：判断是否上网超时。在单元格H3中输入下面的逻辑公式：“=IF(G3>=48,"注意，超时！","否")”，来判断第一个上网卡号上网是否超时，然后利用Excel的自动填充功能判断各个卡号是否超时，如图8.51所示。

图8.51　判断是否超时

步骤8：计算各个卡号的上网费用。计算上网费用只需要根据上网的小时数，然后乘以每小时的费用就可以得到各自的上网费用。用户可以在单元格J3中输入公式：“=IF(G3<48,G3×I3,"超时")”，然后利用自动填充功能来完成所有的计算，如图8.52所示。

图8.52　计算各个客户的上网费用

网吧收费系统可以计算没有超时的卡号的上网总费用，同时对于超时的卡号可以及时地提示其上网的时间超时，可以广泛使用。

8.3.3　计算员工的工龄

人力资源管理中，员工的年龄和工龄是经常需要计算的一个指标。年龄是一个向下取整的数值，没有达到一年的时间都会被省略。工龄除了计算年数之外还要计算月份。两个指标都有不同的计算方法。在这个例子中，主要的计算数据如图8.53所示，需要计算每个员工的年龄和工龄。

图8.53　员工的基本资料

步骤1：计算第一位员工的年龄。用户可以使用两种方法来计算员工的年龄，第一种方法比较简便，在单元格E3中输入公式“=YEAR(TODAY()-C3)-1900”，如图8.54所示。


[image: ]



图8.54　计算员工的年龄

这个公式中用户需要了解和掌握的是为什么要在后面减去1900。这是因为在1900时间系统中如果年份的数值在0~1900之间，那么系统会自动地加上1900，也就是通过公式TODAY()-C3运算得到的年份为32，那么系统会自动加上1900，变成1932。为了得到了员工年龄，用户需要将这个数值减去1900，得到正确的年龄32。如果用户使用的是1904年系统，那么用户需要减去的是1904。

另外一个方法理解起来比较简单，用户可以在单元格E3中输入公式：=IF(TODAY()>=DATE(2005,MONTH(C3),DAY(C3)),2005-YEAR(C3),2005-YEAR(C3)-1)。这个公式中首先判断员工是否已经过了生日，如果员工过了生日，那么返回的年龄是2005-YEAR(C3)；如果员工没有过生日，那么返回的年龄是2005-YEAR(C3)-1，这个方法可以得到和上面相同的结果，如图8.55所示。

图8.55　计算员工的年龄

步骤2：计算第一位员工工龄中的年份。在单元格F3中输入公式：“=YEAR(TODAY()-D3)-1900”，得到的结果如图8.56所示。

图8.56　计算工龄中的年份

步骤3：计算第一位员工工龄中的月份。在单元格G3中输入公式：“=MONTH(TODAY()-D3)-1”，得到的结果如图8.57所示。

图8.57　计算工龄中的月份

步骤4：在这个步骤中介绍另外一个方法来计算工龄的年份。用户可以在单元格F3中输入公式：“=IF(TODAY()>=DATE(2005,MONTH(D3),DAY(D3)),2005-YEAR(D3),2004-YEAR(D3))”。这个公式的含义和前面计算员工年龄的含义基本相似，这里就不再详细解释了，得到的结果如图8.58所示。

图8.58　计算员工的工龄的年份

步骤5：在这个步骤中介绍另外的方法计算员工工龄的月份，用户可以在单元格G3中输入公式：“=IF(TODAY()>=DATE(2005,MONTH(D3),DAY(D3)),INT((TODAY()-DATE(2005,MONTH(D3),DAY(D3)))/30),INT((TODAY()-DATE(2004,MONTH(D3),DAY(D3)))/30))”。这个公式首先判断工龄是否满年，如果员工在今年的工龄满年了，那么超过满年的月份为INT((TODAY()-DATE(2005,MONTH(D3),DAY(D3)))/30)，其中的INT函数是向下取整的函数，在第3章已经详细介绍；如果员工在今年的工龄没有满年，那么超过满年的月份为INT((TODAY()-DATE(2004,MONTH(D3),DAY(D3)))/30)，得到的结果如图8.59所示。

图8.59　计算员工的工龄的月份

步骤6：计算其他员工的年龄和工龄。用户可以利用Excel的自动填充功能来计算其他员工的年龄和工龄，得到的结果如图8.60所示。

图8.60　最后的计算结果

上面两种计算方法都是比较常用的，用户可以在生活中广泛使用，希望用户能够熟练掌握。


第9章 信息函数

信息函数可以确定存储在单元格中的数据类型，同时还可以使单元格在条件满足的情况下返回逻辑值。

信息函数大致可以分为两大类，第一类获取的是单元格的属性信息，包括CELL、INFO、ERROR.TYPE等函数，这些函数的参数和返回值相对都是比较固定的。例如CELL函数返回的主要是单元格的路径及其他的位置信息。用户在使用这类信息函数的时候，需要留意参数的使用方法及返回数值的含义，对于这类函数的参数及返回值在后面都有详细的介绍，用户可以查看对应的内容。

第二类信息函数又可以称为IS类函数，可以检验数值的类型并根据参数取值返回TRUE或FALSE。IS类函数包含了用来检验数值或引用类型的九个工作表函数，分别为ISBLANK、ISTEXT、ISERR、ISERROR、ISNOTEXT、ISNA、ISLOGICAL、ISNUMBER、ISREF。这些函数有着相同的参数Value，表示用于检验的数值。

下面以函数ISNA为例，来介绍如何使用信息函数，如图9.1所示。


[image: ]



图9.1　使用ISNA函数

从上面的例子中用户可以看出ISNA函数的使用方法，这个函数可以用来判断错误值的类型是否是#N/A，如果错误值的类型是#N/A，则返回TRUE，否则返回FALSE。其他IS类的函数的使用方法和这个函数的使用方法相同，只是每个函数判断的内容不同而已。例如ISTEXT函数判断的是单元格中的数据是否是文本格式，ISREF函数判断的是单元格中的数据是否是引用类型。

9.1　函数应用实战

9.1.1　CELL函数：返回单元格信息

CELL函数的功能是返回某一引用区域的左上角单元格的格式、位置或内容等信息。

CELL函数的表达式是CELL(info_type,reference)。

参数info_type是一个文本值，指定所需要的单元格信息的类型；参数reference表示要获取其有关信息的单元格。如果忽略，则info_type中所指定的信息将返回给最后更改的单元格。

【实际应用】

某公司需要在工作表的一个单元格中插入当前工作簿名称。需要使用CELL函数，然后使用前面章节讲解的MID函数来选择需要返回的路径，如图9.2所示。


[image: ]



图9.2　返回路径

图中公式“=CELL("filename")”返回的是当前工作薄的完整路径，包括驱动器盘符和文件的全部路径；公式“=MID(CELL("filename"),1,FIND("[",CELL("filename"))-1)”则利用MID函数选取完整路径中“[”之前的目录路径，而“[”是该文件的名称，得到的是文件的保存路径；公式“=MID(CELL("filename"),FIND("[",CELL("filename"))+1,FIND("]",CELL(

"filename"))-FIND("[",CELL("filename"))-1)”则选取完整路径中[]之间的字符也就是工作薄的名称；公式“=MID(CELL("filename"),FIND("]",CELL("filename"))+1,255)”选择的是工作表的名称。

【注意事项】

下面列出info_type的可能值及相应的结果，如表9.1所示。

表9.1　参数info_type的可能取值和相应的结果

【实际应用】

用户使用ERROR.TYPE检测错误值，同时也可以使用其他函数嵌套得到相应的其他错误信息，如图9.3所示。

图9.3　使用ERROR.TYPE函数

【注意事项】

下面列出各种错误类型的返回值，如表9.2所示。

表9.2　各种错误类型返回值

9.1.3　INFO函数：返回当前操作环境的信息

INFO函数的功能是返回有关当前操作环境的信息。

INFO函数的表达式是INFO(type_text)。

唯一的参数type_text是文本类型，指明所要返回的信息类型。

【实际应用】

用户可以使用INFO函数来查看和当前操作环境有关的信息，如图9.4所示。

图9.4　使用INFO函数

【注意事项】

下面列出各种错误类型的返回值，如表9.3所示。

表9.3　参数输入和返回信息的表格

9.1.4　ISBLANK函数：判断单元格是否为空

ISBLANK函数的功能是判断参数的内容是否为空白。

ISBLANK函数的表达式是ISBLANK(value)。

参数value表示需要进行检验的单元格。

【实际应用】

已知某个班上的两次考试的成绩，现在需要根据情况来计算平均值。如果有空格，也就是没有考试成绩，需要根据条件来显示“缺考”，原始数据如图9.5所示。

在上面的单元格D2中输入公式“=IF(OR(ISBLANK(B2),ISBLANK(C2)),"缺考",AVERAGE(B2,C2))”。首先利用ISBLANK函数判断单元格B2或者C2是否为空，如果是空格，则返回“缺考”，否则返回两个数的平均值，如图9.6所示。

图9.5　原始数据

图9.6计算平均值

利用自动填充功能完成其他单元格的数据，得到的最后结果如图9.7所示。

图9.7　最后结果

9.1.5　ISERR和ISERROR函数：判断参数是否为错误值

ISERR函数和ISERROR函数的功能是判断参数的内容是否为任意错误值。

ISERR函数的表达式是ISERR(value)。

ISERROR函数的表达式是ISERROR(value)。

参数value表示需要进行检验的内容。两个函数的区别在于ISERR函数可以判断的错误除去#N/A类型的错误，而ISERROR函数可以判断的内容是任何类型的错误。

【实际应用】

某超市统计了某个订货系统的数据，由于可能存在有原价统计问题，因此最后统计结果需要经过处理，原始数据如图9.8所示。

图中的单元格中有两个单元格的定价不是数值，原因是没有确定单价，因此需要使用ISERR函数或者ISERROR函数来判断错误值，如图9.9所示。

图9.8　原始数据

图9.9　分析结果

利用自动填充功能完成其他单元格的数据，得到的最后结果如图9.10所示。

图9.10　最后结果

9.1.6　ISLOGICAL函数：判断参数是否是逻辑值

ISLOGICAL函数的功能是判断参数的内容是否是逻辑值。

ISLOGICAL函数的表达式是ISLOGICAL(value)。

【实际应用】

如果检验的内容是逻辑值，将会返回TRUE；否则将会返回FALSE，应用如图9.11所示。

图9.11　使用ISLOGICAL函数

9.1.7　ISNA函数：判断错误值是否是#N/A

ISNA函数的功能是判断参数的内容是否是错误值#N/A。

ISNA函数的表达式是ISNA(value)。

【实际应用】

某单位统计了单位员工的记录，现在需要根据原始的记录来查询相应的员工信息，原始数据如图9.12所示。

如果用户输入的员工编号不在原始数据当中，那么应该返回提示信息，而不能返回错误值#N/A，因此在单元格F4中输入公式“=IF(ISNA(LOOKUP(F3,A2:A11,B2:B11)),"没有记录",LOOKUP(F3,A2:A11,B2:B11))”，如图9.13所示。

图9.12　原始数据

图9.13　查询结果分析

然后在单元格F5中输入公式“=IF(ISNA(LOOKUP(F3,A2:A11,C2:C11)),"没有记录",LOOKUP(F3,A2:A11,C2:C11))”，如图9.14所示。

用户可以在单元格中输入相应的编号，检查上面的公式的设置情况，如图9.15所示。

图9.14　查询结果分析

图9.15　检查函数设置情况

9.1.8　ISNOTEXT函数：判断参数是否是非字符串

ISNOTEXT函数的功能是判断参数的内容是否是非字符串。

ISNOTEXT函数的表达式是ISNOTEXT(value)。

【实际应用】

使用ISNOTEXT函数如果检验的内容是非字符串，将会返回TRUE；否则将会返回FALSE，应用如图9.16所示。

图9.16　原始数据

9.1.9　ISNUMBER函数：判断参数是否是数字

ISNUMBER函数的功能是判断参数的内容是否是数字。

ISNUMBER函数的表达式是ISNUMBER(value)。

【实际应用】

某单位的数据行中对于不同的年份有不同的数据格式，对于2000年份以后的数据在对应的数据前有两个字母，而在2000年以前的数据则没有字母，现在用户需要返回对应的年份，原始数据如图9.17所示。

在单元格B2中输入公式“=IF(ISNUMBER(VALUE(A2)),MID(A2,1,2),MID(A2,3,2))”，首先使用ISNMUBER函数来判断单元格中的文本是否是数字，然后截取相应的年份即可，如图9.18所示。

图9.17　原始数据

图9.18　返回年份

9.1.10　ISREF函数：判断参数是否是引用

ISREF函数的功能是判断参数的内容是否是引用。

ISREF函数的表达式是ISREF(value)。

【实际应用】

使用ISREF函数如果检验的内容是引用，将会返回TRUE；否则将会返回FALSE，应用如图9.19所示。

9.1.11　ISTEXT函数：判断参数是否是文本

ISTEXT函数的功能是判断参数的内容是否是文本。

ISTEXT函数的表达式是ISTEXT(value)。

【实际应用】

如果检验的内容是引用，将会返回TRUE；否则将会返回FALSE，如图9.20所示。

图9.19　使用ISREF函数

图9.20　使用ISTEXT函数

9.1.12　TYPE函数：返回数值的类型

TYPE函数的功能是返回数值的类型。

TYPE函数的表达式是TYPE(value)。

【实际应用】

当某一个函数的计算结果取决于特定单元格的数据类型时，可使用函数TYPE，应用如图9.21所示。

图9.21　使用TYPE函数

9.2　综合实例

信息函数在实际工程中有着广泛的应用，一般情况下，信息函数会和其他函数，例如查询函数等联合起来使用，下面举一个综合的实例说明。

某公司每天需要统计销售额的总量，单位是万元，每天更新输入数据。现在需要完成的是根据每天输入的数据，在某一列显示相应的数据，并在最后一行得到更新后的数据的最大、最小值及其他的统计信息，原始的数据如图9.22所示。

在“数据引用”列需要将原始数据进行筛选，在原始数据列中没有输入数据的时候，将显示#N/A，然后在“数据整理”列得出最后的数据整理结果，下面介绍实现方法。

步骤1：在单元格D2中输入公式“=IF(B2="",NA(),B2)”。这个公式将区分输入数据和没有数据的表格，空表格代表的是当天还没有输入相应的数据，使用函数ISNA将空格转化为错误值#N/A，如图9.23所示。


[image: ]



图9.22　原始数据

图9.23　输入公式

步骤2：利用自动填充功能，完成“数据引用”列的数据，如图9.24所示。

步骤3：在单元格F2中输入公式“=IF(ISNA(D2),"未知",IF(OR(D2=MAX(OFFSET($D$2,,,COUNT(D:D))),D2=MIN(OFFSET($D$2,,,COUNT(D:D)))),D2,"一般值"))”，完成对数据的统计，如果是最大或者最小值，返回相应的数值；如果不是，返回“一般值”；如果没有输入数据，那么返回“未知”，如图9.25所示。

图9.24　完成“数据引用”列的数据

图9.25　输入公式

步骤4：利用自动填充功能，完成“数据整理”列的数据，如图9.26所示。

步骤5：图中的单元格中首先使用ISNA判断单元格是否是错误值#N/A，如果是错误值#N/A，那么返回“未知”，否则返回最大、最小值。最后用户可以输入新的数据，判断函数公式是否设置正确，如图9.27所示。

图9.26　完成“数据整理”列的数据

图9.27　检测公式

图中的单元格中首先输入的是新日期的数据，是其中的最大值，可以看到在最后一列中显示的就是新的最大、最小值，表明公式设置正确。


第10章 财务函数

财务函数主要是一般财务数据计算所使用的函数。下面根据情况分为5个大类：投资函数、本金和利息函数、报酬率函数、折旧函数、有价证券函数。

10.1　投资函数应用实战

投资函数是用于计算投资和收益的函数。这类的函数一般有着相似的参数。例如rate表示投资期限内的各期利率。在表示投资金额和时间方面，这类函数有不同的处理方法。在本节中将会讲解FV、PV、NPV、XNP函数。各个函数分别适用于不同的投资类型。下面将分别介绍各个不同的函数。

10.1.1　FV函数：返回未来值

FV函数的功能是基于固定利率及等额分期付款方式，返回某项投资的未来值。

FV函数的表达式是FV(rate,nper,pmt,pv,type)。

其中rate是各期利率；nper表示总投资期，即该项投资的付款期总数；pmt表示各期所应支付的金额，其数值在整个年金期间保持不变；pv表示现值，即从该项投资开始计算时已经入账的款项，或一系列未来付款的当前值的累积和，也称为本金；type表示数字0或1，用以指定各期的付款时间是在期初还是期末。

【实际应用】

假设今年年初甲将1000元存入银行，并将在以后的9年中每年年初存入1000元。如果银行利率为10%，那么现在需要计算第10年年初甲的银行存款账户余额为多少。可以使用两种方法来计算，第一个方法是直接累加得出结果，第二个方法是使用FV函数。首先列表累加计算，如图10.1所示。

图10.1　累加计算终值

步骤1：在单元格D6中输入公式“=$B$3×(C6+B6)”，计算当年的利息。

步骤2：在单元格E6中输入公式“=D6+C6+B6”，计算当年的年末余款。

步骤3：在单元格B7中输入公式“=E6”，因为第一年的年初余款和当年的年末余款相同，然后在对应的单元格D7、E7计算第一年的利息和年末余款。

步骤4：利用自动填充功能，可以计算得到第10年年初甲的银行存款账户余额，得到的结果是17,531.17，如图10.1所示。

下面用户可以使用FV函数进行核算，在单元格C20中输入公式“=FV(B3,A16,-1000,,1)”，其中A16是投资的期数，得到的结果如图10.2所示。

图10.2　使用FV函数

10.1.2　PV函数：返回现值

PV函数的功能是返回投资的现值。现值为一系列未来付款的当前值的累积和。

PV函数的表达式是PV(rate,nper,pmt,fv,type)。

其他参数和函数FV的参数相同，参数fv表示未来值，或在最后一次支付后希望得到的现金余额。

【实际应用】

某部门经理今年55岁并将于60岁退休。为了奖励该职工，公司决定为其开设一个退休金账户，并将从今年开始连续5年每年年初存入一笔奖金。该账户将保证该职工从60岁退休那年开始的6年中，每年年初可以从退休金账户中提取20000元，则每年的存款额应为多少（其中银行的存款年利率为6%）。

这是一个养老金的实际问题，用户可以使用PV函数进行计算，但是由于情况比较复杂，在这个例子中，用户还是可以使用图标累加计算，如图10.3所示。

图10.3　累加图表

步骤1：由于前5年的存款额数目相同，在单元格B3中保存每年的存款额。

步骤2：在单元格C7~C11中输入公式“=$B$3”，代表前5年每年的存款额。

步骤3：在单元格D7中输入公式“=$B$2×(C7+B7)”，计算55岁的年利息，其中B2表示的是存款利率，(C7+B7)表示的是存款总额。

步骤4：在单元格E7中输入公式“=D7+C7+B7”，计算55岁的年末余额。D7表示的是当年利息，B7表示的是年初余额，C7表示每年的存款额，累加的是年末余额。

步骤5：在单元格C12~C17中输入“-20000”，因为从60岁开始到65岁每年都需要从退休金账户中提取20000元。

步骤6：在单元格B8中输入公式“=E7”，因为第一年年初余款和当年的年末余款相同。

步骤7：利用自动填充功能来计算从55岁到65岁的每年利息和年末余额。

步骤8：由于员工65岁的时候年末余额为0，在50岁的年初余额也是0，同时在60岁以后的每年中取金额为20000，可以循环计算得到存款额为17466。

下面详细介绍如何可以使用FV函数来计算每年的存款额，如图10.4所示。

图10.4　计算养老金存款金额

步骤1：在单元格B4中输入公式“=1/(1+B2)^5×PV(B2,6,-20000,,1)”计算公式中的分子数值。

步骤2：在单元格B5中输入公式“=PV(B2,5,-1,,1)”得到分母的数值。

步骤3：在单元格B6中输入公式“=B4/B5”，将上面两个数据相除得到养老金的年存金额，得到的结果是17466，可见结果相同。

关于养老金的计算公式用户可以查看相关的财务书籍。

10.1.3　NPV函数：返回净现值

NPV函数的功能是通过使用贴现率及一系列未来支出（负值）和收入（正值），返回一项投资的净现值。

NPV函数的表达式是NPV(rate,value1,value2,...)。

其中rate表示某一期间的贴现率，是一个固定值；value1,value2,...表示1~29个参数，代表支出及收入。

【实际应用】

某投资项目各年的现金流分别为-220、90、100、110、120，给定标准回收期为3年。在考虑复利情况下，采用回收期法判断项目是否可行，如图10.5所示。


[image: ]



图10.5　计算回收期是否符合标准

步骤1：计算第一年的净现值。在单元格B12中输入公式“=NPV(B8,B3)+B2”，其中B2表示项目的初始投资额。

步骤2：计算后面几年投资项目的净现值。在单元格B13~B15中分别计算在不同投资回收期中的净现值的大小，如图10.5所示。

步骤3：计算回收期。在单元格B17中输入公式“=-B13/(B5/(1+B8)^3)+A13”，得到项目的相应回收期。

步骤4：判断项目是否可行。在单元格B19中输入公式“=IF(B7-B17>0,"可行","不可行")”，IF函数用来判断项目回收期是否超过标准回收期，如果超过标准回收期，项目不可行，如果没有超过标准回收期，则项目可行。

10.1.4　XNPV函数：返回净现值

XNPV函数的功能是返回一组现金流的净现值，这些现金流不一定定期发生。

XNPV函数的表达式是XNPV(rate,values,dates)。

其中参数rate表示应用于现金流的贴现率；values表示与dates中的支付时间相对应的一系列现金流转；dates表示与现金流支付相对应的支付日期表。

【实际应用】

沿用上面的例子，用户需要使用XPNV函数计算各个时期的净现值，如图10.6所示。

图10.6　计算投资回收期

步骤1：计算在投资时期内投资项目的净现值。在单元格B11~B15中分别使用XNPV函数计算在不同投资回收期中的净现值的大小，如图10.6所示。

步骤2：沿用10.1.3中的后续步骤，完成项目可行性的判断，得到的结果和上面的结果完全相同。

需要特别提醒用户的是，XNPV和NPV函数的参数不同，其中XNPV函数的最后一个参数只能是date类型的数据，不能是其他类型的数据。

10.2　本金和利息函数应用实战

本金和利息计算函数是用于计算本金和利息部分的函数。本金是指未来各期年金现值之和。利息是指在某一段连续时间内，一系列的固定金额支出，例如汽车贷款或购房贷款。

本节要讲解的四个函数是相互联系的。前三个函数PMT、IPMT、PPMT有相同的参数，rate,nepr,pv,fv,type。其中rate表示的是年金利率，nper表示的是年金期限，pv表示本金，fv表示在末期回收的现金余额，type表示时间计算的类型，选择0或者1分别指定付款时间在期末和期初。它们的不同之处在于，PMT计算的是本金和利息之和，IPMT计算的是利息，PPMT计算的是本金。因此函数IPMT和PPMT计算的数值之和与函数PMT计算的数值相同。

最后介绍的CUMIPMT和CUMPRINC函数，分别计算各个阶段累积的利息和本金。是各个时间对应的利息和本金累积之和，这些函数在实际的财务分析经常使用。下面详细介绍这些函数。

10.2.1　PMT函数：返回贷款的每期付款额

PMT函数的功能是基于固定利率及等额分期付款方式，返回贷款的每期付款额。

PMT函数的表达式是PMT(rate,nper,pv,fv,type)。

有关函数PMT中参数的详细说明，请参阅函数PV中的说明。

【实际应用】

假设某客户为了购房，申请了100000元的房屋抵押贷款，贷款的年利率为6%，银行要求10年内还清，则每年末应还款多少？下面用户需要使用PMT函数计算每年还款数额，如图10.7所示。

在图中的单元格B5中输入公式“=PMT(B3,B4,-B2)”，得到的结果如图10.7。为了理解上面的计算结果，下面演示一下这个还款的数额表，如图10.8所示。

图10.7　使用函数计算每年还款数额

图10.8　还款数额表

步骤1：在单元格C9~C18中输入上面计算的结果13586.80，作为每年年末的还款数额。

步骤2：在单元格D9中输入公式“=$B$3×B9”，计算第一年所还的利息。

步骤3：在单元格E9中输入公式“=C9-D9”，计算第一年所还的本金。

步骤4：在单元格B10中输入公式“=B9-E9”，计算第二年的年初本金。

步骤5：利用自动填充功能，可以计算得到10年中每年所还的利息和本金。

在上面的还款数额表中，在第10年的年初本金和所还本金相同，证明在第10年，用户已经还清所有的贷款，而每年的还款额就是由PMT公式计算出来的结果。

10.2.2　IPMT函数：返回付款的利息额

IPMT函数的功能是基于固定利率及等额分期付款方式，返回给定期数内对投资的利息偿还额。

IPMT函数的表达式是IPMT(rate,per,nper,pv,fv,type)。

有关函数IPMT中参数的详细说明，请参阅函数PV中的说明。

【实际应用】

假设某客户为了购房，申请了100000元的房屋抵押贷款，贷款的年利率为6%，银行要求10年内还清，则求每年末付款中的利息。每年还款的利息数额是不等的，一般的会逐年减少，计算的结果如图10.9所示。

图10.9　计算每年的还款利息数值

步骤1：计算第一年所还的利息。在单元格B7中输入公式“=IPMT($B$3,A7,$B$4,$B$2)”，其中A7单元格表示的是第一年还款的期数，得到第一年所还的利息。

步骤2：计算其他年份所还的利息。利用自动填充功能，可以计算其他年份所还的利息，如图10.9所示。

用户可以将上面单元格中的还款利息和图10.8中的还款数额表中的“所还利息”数据列进行比较，可以看到两者的数额相同。

10.2.3　PPMT函数：返回付款的本金额

PPMT函数的功能是基于固定利率及等额分期付款方式，返回投资在某一给定期间内的本金偿还额。

PPMT函数的表达式是PPMT(rate,per,nper,pv,fv,type)。

有关函数PPMT中参数的详细说明，请参阅函数PV中的说明。

【实际应用】

假设某客户为了购房，申请了100000元的房屋抵押贷款，贷款的年利率为6%，银行要求10年内还清，则求每年末付款中的本金。每年还款的本金数额是不等的，一般的会逐年增加，计算的结果如图10.10所示。

图10.10　计算每年还款的本金数值

步骤1：计算第一年所还本金。在单元格B7中输入公式“=PPMT($B$3,A7,$B$4,$B$2)”，其中A7单元格表示的是第一年还款的期数，得到第一年所还的本金。

步骤2：计算其他年份所还的本金。利用自动填充功能，可以计算其他年份所还的本金。

用户可以将上面单元格中的还款本金和图10.8中的还款数额表中的“所还本金”数据列进行比较，可以看到两者的数额相同。

10.2.4　CUMIPMT和CUMPRINC函数：返回阶段利息和本金

CUMIPMT函数的功能是返回一笔贷款在给定的start-period到end-period期间累计偿还的利息数额。

CUMPRINC函数的功能是返回一笔贷款在给定的start-period到end-period期间累计偿还的本金数额。

CUMIPMT函数的表达式是CUMIPMT(rate,nper,pv,start_period,end_period,type)。

CUMPRINC函数的表达式是CUMPRINC(rate,nper,pv,start_period,end_period,type)。

其中参数start_period表示计算中的首期，付款期数从1开始计数；end_period表示计算中的末期。

【实际应用】

还是沿用上面的例子，现在用户需要计算的是相对于第一年还款的时间开始统计，到还款的日期的累计还款的本金和利息数额，如图10.11所示。

图10.11　计算累计还款利息和本金

步骤1：计算第一年的累计偿还利息。在单元格B7中输入公式“=CUMIPMT($B$3,$B$4,$B$2,$A$7,A7,0)”，可以看到由于累计的是第一年的偿还利息，其数值上和第一年所还的利息数相同。

步骤2：计算第一年的累计偿还本金。在单元格C7中输入公式“=CUMPRINC($B$3,$B$4,$B$2,$A$7,A7,0)”，可以看到由于累计的是第一年的偿还本金，其数值上和第一年所还的本金数额相同。

步骤3：利用自动填充功能，得到每年相应的累加还款利息和本金，得到的结果如图10.11所示。

10.3　报酬率函数应用实战

报酬率函数是计算内部资金流量的回报率的函数，本节中介绍的是IRR、MIRR、XIRR三个函数。这三个函数相互联系，其中IRR计算的是定期现金流项目的内部收益率，同时假设各期的投资成本保存不变；MIRR是对IRR函数的一个修正，相对于IRR的修改之处是考虑了投资成本的不同，贷款利率和再投资利率在MIRR中可以分别考虑；XIRR对于IRR的改进之处在于现金流可以发生在不定期。

用户可以根据实际项目的情况选择不同的函数计算投资报酬率，下面详细介绍这三个函数。

10.3.1　IRR函数：返回内部收益率

IRR函数的功能是返回由数值代表的一组现金流的内部收益率。

IRR函数的表达式是IRR(values,guess)。

参数values表示数组或单元格的引用，包含用来计算返回的内部收益率的数字；guess为对函数IRR计算结果的估计值。

【实际应用】

假设某人开公司，投资额为1500000元。预计在今后的5年内的收益分别为400000元、450000元、580000元、660000元和710000元。现在用户需要计算投资1年后、3年后和5年后的内部收益率，如图10.12所示。

图10.12　原始计算的数值

步骤1：计算投资1年后的内部收益率。在单元格E4中输入公式“=IRR(B1:B2,-0.5)”，其中-0.5是用户对投资1年后内部收益率的估计值，得到1年后的内部收益率为73%。

步骤2：计算投资2年后的内部收益率。在单元格E5中输入公式“=IRR(B1:B4)”，得到2年后的内部收益率为-2%。

步骤3：计算投资3年后的内部收益率。在单元格E6中输入公式“=IRR(B1:B6)”，得到的结果如图10.13所示。

图10.13　计算内部收益率

10.3.2　MIRR函数：返回修正内部收益率

MIRR函数的功能是返回某一连续期间内现金流的修正内部收益率。该函数同时考虑了投资的成本和现金再投资的收益率。

MIRR函数的表达式是MIRR(values,finance_rate,reinvest_rate)。

参数values表示一个数组或对包含数字的单元格的引用。这些数值代表着各期的一系列支出（负值）及收入（正值）；参数finance_rate表示现金流中使用的资金支付的利率；参数reinvest_rate表示将现金流再投资的收益率。

【实际应用】

假设某公司在5年前以12%的年利息贷款了2000000元，而公司在这5年内的收入为250000元、430000元、550000元、660000元和750000元。在这5年期间用户将收入再投资，每年的收益率为15%。用户需要计算的是3年后和5年后的修正内部收益率，如图10.14所示。

图10.14　计算的原始数据

步骤1：计算3年后的修正内部收益率。在单元格E4中输入公式“=MIRR(B1:B4,B7,B8)”，其中B1:B4代表的是包含投资期和回收期在内的现金流，得到的结果为-12%。

步骤2：计算5年后的修正内部收益率。在单元格E5中输入公式“=MIRR(B1:B6,B7,B8)”，得到投资5年后的修正内部收益率，如图10.15所示。

图10.15　计算两个不同的内部收益率

10.3.3　XIRR函数：返回内部收益率

XIRR函数的功能是返回一组现金流的内部收益率，这些现金流不一定定期发生。

XIRR函数的表达式是XIRR(values,dates,guess)。

参数values表示与dates中的支付时间相对应的一系列现金流；dates表示与现金流支付相对应的支付日期表；guess表示对函数XIRR计算结果的估计值。

【实际应用】

假设某公司在5年前以12%的年利息贷款了2000000元，而公司在这5年内的收入为250000元、430000元、550000元、660000元和750000元，但是收入的时间不是定期的，用户需要计算这些现金流的内部收益率，如图10.16所示。

图10.16　计算内部收益率

在单元格B9中输入公式“=XIRR(B2:B7,A2:A7))”，得到不定期现金流的内部收益率，结果为18.39%。

10.4　折旧函数应用实战

企业固定资产的使用寿命是有限的。在其使用年限内，固定资产会逐年因有形或者无形的损耗而丧失服务能力。因此有必要将其成本在使用年限内转化为费用，这个过程就是折旧。而本节所讲解的折旧函数就是固定资产折旧值的函数。

本节的函数分别使用不同的方法来计算固定资产的折旧值。其中DB函数和DDB函数拥有相似的参数，主要的区别在于余额递减的因数。用户可以使用DDB函数选择余额递减的速率。SLN函数使用平均法计算各期的折旧值。因此，对于计算的年限而言，各年的折旧值相同；SYD函数计算的方法是年限总和折旧法，所有年限的折旧值之和与成本和残值之差相同；AMORDEGRC函数为法国会计系统提供的计算每年折旧值。讲解这个函数是为了向用户介绍一些特殊的折旧方法。Excel还提供了其他的方法，用户可以根据需要查看相应的帮助文件。下面详细讲解这些函数。

10.4.1　DB函数：固定余额递减折旧

DB函数的功能是使用固定余额递减法，计算一笔资产在给定期间内的折旧值。

DB函数的表达式是DB(cost,salvage,life,period,month)。

参数cost表示资产原值；salvage表示资产在折旧期末的价值（也称为资产残值）；life为折旧期限（有时也称作资产的使用寿命）；period表示需要计算折旧值的期间；month表示第一年的月份数。

【实际应用】

假设某公司在第一年的3月份购买了一台新机器，价值为1500000元，使用年限为6年，估计残值为100000元。现在需要计算每年的折旧值，如图10.17所示。

步骤1：计算第一年折旧值。在单元格B4中输入公式“=DB($A$2,$B$2,$C$2,C4,$D$2)”，得到的第一年的折旧值，如图10.18所示。

图10.17　原始计算的数值

图10.18　计算第一年的折旧值

步骤2：计算其他年份的折旧值。利用自动填充功能，直接拖动公式计算其他年份的折旧值，得到的结果如图10.19所示。

图10.19　计算各年的折旧值

10.4.2　DDB函数：双倍余额递减折旧

DDB函数的功能是使用双倍余额递减法或其他指定方法，计算一笔资产在给定期间内的折旧值。

DDB函数的表达式是DDB(cost,salvage,life,period,factor)。

其中参数的含义和DB函数的参数含义类似，参数factor表示余额递减速率。如果factor被省略，则假设为2（双倍余额递减法）。

【实际应用】

假设某公司在第一年的3月份购买了一台新机器，价值为1500000元，使用年限为6年，估计残值为100000元。现在需要在计算每年的折旧值，原始数据和DB函数相同，但现在使用双倍余额递减法进行计算，如图10.20所示。

在元格C4中输入公式“=DDB($A$2,$B$2,$C$2,C4,2)”，按照双倍约递减法计算得到的第一年折旧值，然后利用自动填充功能对其他的折旧值进行计算，如图10.21所示。

图10.20　计算第一年的折旧值

图10.21　计算不同年限的折旧值

10.4.3　SLN函数：平均折旧

SLN函数的功能是返回某项资产在一个期间中的平均折旧值。

SLN函数的表达式是SLN(cost,salvage,life)。

参数含义和DB函数相同。

【实际应用】

假设某公司在第一年的3月份购买了一台新机器，价值为1500000元，使用年限为6年，估计残值为100000元。现在需要在计算每年的折旧值，原始数据和DB函数相同，但现在使用平均折旧法进行计算，如图10.22所示。

图10.22　计算每年的平均折旧值

在单元格D2中输入公式“=SLN(A2,B2,C2)”，算得每年的平均折旧值。

10.4.4　SYD函数：年限总和折旧

SYD函数的功能是返回某项资产按年限总和折旧法计算的指定期间的折旧值。

SYD函数的表达式是SYD(cost,salvage,life,per)。

【实际应用】

假设某公司在第一年的3月份购买了一台新机器，价值为1500000元，使用年限为6年，估计残值为100000元。现在需要在计算每年的折旧值，原始数据和DB函数相同，但现在使用年限总和折旧法进行计算，如图10.23所示。

在图中的单元格C4中输入公式“=SYD($A$2,$B$2,$C$2,C4)”，将使用年限总和折旧法进行计算得到第一年的折旧值，然后利用自动填充功能对其他的折旧值进行计算，如图10.24所示。

图10.23　计算第一年的折旧值

图10.24　计算不同年限的折旧值

从上面的计算情况可以知道，累计折旧值就是成本和资产残值的差，这正是年限折旧的计算法则。

10.4.5　AMORDEGRC函数：返回折旧

AMORDEGRC函数的功能是返回每个结算期间的折旧值。该函数主要为法国会计系统提供。如果某项资产是在该结算期的中期购入的，则按平均折旧法计算。

AMORDEGRC函数的表达式是AMORDEGRC(cost,date_purchased,first_period,salvage,period,rate,basis)。

其中参数date_purchased表示购入资产的日期；first_period表示第一个期间结束时的日期；rate表示折旧率；Basis表示所使用的年基准。

【实际应用】

某人购买了价值为2400元的资产，购入的日期为2008-8-19，第一个时间结束的日期为2008-12-31，资产残值为300，折旧率为15%，计算第一时期的折旧值，如图10.25所示。

图10.25　计算第一时期的折旧值

在单元格B4中输入公式“=AMORDEGRC(A2,B2,C2,D2,E2,F2,G2)”，得到第一时期的折旧值。

10.5　有价证券函数应用实战

有价证券函数是一类用于计算证券收益的函数。这类函数的主要参数有利息率、收益率、价格、成交日、发行日、证券价值等。这些函数对应的参数与财务函数中的相同。有价证券函数有共同的日计数标准，表示如何计算证券的收益期限，在函数中使用参数basis来表示这个计数标准，这在后面会详细介绍。

10.5.1　ACCRINT函数：返回应付利息

ACCRINT函数的功能是返回定期付息有价证券的应计利息。

ACCRINT函数的表达式是ACCRINT(issue,first_interest,settlement,rate,par,frequency,basis)。

参数issue表示有价证券的发行日；first_interest表示证券的起息日；settlement表示证券的成交日，即在发行日之后，证券卖给购买者的日期；rate表示有价证券的年息票利率；par表示有价证券的票面价值；basis表示日计数基准类型。

【实际应用】

假设某人在2005年8月10日购买了价值为10000元的国库券，发行日期为2005年6月5日，起息日为2005年10月5日，利息为10%。按照半年期付息，以US（NASD）30/360为日计数基准，如图10.26所示。

在单元格B9中输入公式“=ACCRINT(B1,B2,B3,B4,B5,B6,B7)”，得到这个国库券的到期利息，如图10.27所示。


[image: ]



图10.26　原始计算的数值

图10.27　计算到期利息

10.5.2　ACCRINTM函数：返回应计利息

ACCRINTM函数的功能是返回到期一次性付息有价证券的应计利息。

ACCRINTM函数的表达式是ACCRINTM(issue,maturity,rate,par,basis)。

其参数和ACCRINT函数的参数含义类似，参数maturity表示有价证券的到期日。

【实际应用】

假设某人购买了价值为10000元的国库券，发行日期为2005年6月5日，到期日为2005年10月5日，利息为10%，按照半年期付息，以US（NASD）30/360为日计数基准，计算到期的利息，如图10.28所示。

图10.28　计算应付利息

在单元格B7中输入公式“=ACCRINTM(B1,B2,B3,B4,B5)”，得到这个国库券的应付利息。

10.5.3　INTRATE函数：返回利率

INTRATE函数的功能是返回一次性付息证券的利率。

INTRATE函数的表达式是INTRATE(settlement,maturity,investment,redemption,basis)。

参数investment表示有价证券的投资额；redemption表示有价证券到期时的清偿价值。

【实际应用】

假设某人购买了价值为200000元的债券，成交日为2005年1月15日，在2005年8月15日到期。到期时将会返回的金额为207350元，以US（NASD）30/360为日计数基准，计算债券的利率，如图10.29所示。


[image: ]



图10.29　计算一次性付息证券的利率

在单元格B6中输入公式“=INTRATE(B1,B2,B3,B4,B5)”，得到证券的利率。

10.5.4　PRICE函数：返回价格

PRICE函数的功能是返回定期付息的面值$100的有价证券的价格。

PRICE函数的表达式是PRICE(settlement,maturity,rate,yld,redemption,frequency,basis)。

其中参数和DB函数的参数含义类似，参数yld表示有价证券的年收益率。

【实际应用】

假设某人在2004年2月15日购买了价值为$100的债券。该债券的到期日为2013年11月15日，票息半年利率为5.75%，按照半年支付，收益率为6.50%，，以US(NASD)30/360为日计数基准，要求计算该债券的发行价格，如图10.30所示。

在单元格B9中输入公式“=PRICE(B1,B2,B4,B5,B3,B6,B7)”，得到债券的发行价格，如图10.31所示。

图10.30　原始的计算数据

图10.31　计算债券的发行价格

10.5.5　YIELD函数：返回收益率

YIELD函数的功能是返回定期付息有价证券的收益率。

YIELD函数的表达式是YIELD(settlement,maturity,rate,pr,redemption,frequency,basis)。

【实际应用】

假设某人在2004年2月15日以$94购买了价值为$100的债券。该债券的到期日为2013年11月15日，票息半年利率为5.75%，按照半年支付，以US（NASD）30/360为日计数基准，要求计算该债券的收益率，如图10.32所示。

在单元格B9中输入公式“=YIELD(B1,B2,B5,B3,B4,B6,B7)”，得到债券的收益率，如图10.33所示。

图10.32　原始的计算数据

图10.33　计算债券的收益率

10.6　综合实例

假设某公司在2005年1月贷款金额5000000元，贷款的期限为10年，年利率为9.5%。同时该公司利用该贷款购买了机器设置，该机器设置的使用寿命为10年。同时公司该投资项目在10年内带来的利润额如图10.34所示，现在用户需要计算图中的贷款经营表。

图10.34　原始数据表

下面依次按照次序计算图中的数据表格。

步骤1：计算设备的折旧情况。根据前面讲解的各种折旧函数，这里根据需要用户可以使用的函数是DB函数，在单元格B4中输入公式“=DB($B$1,$H$1,$D$1,YEAR(A4)-2005)”，其中嵌套的时间函数的功能是返回当年的period参数，如图10.35所示。

图10.35　计算第一年的设备折旧值

步骤2：计算其他年限的折旧值。根据自动填充功能，计算其他年限的折旧值，如图10.36所示。

步骤3：计算每年归还的利息。现在需要计算的是C列的数据，用户可以使用IPMT函数计算这列的数据，然后利用自动填充功能完成数据。在单元格C4中输入公式“=-IPMT($F$1,YEAR(A4)-2005,$D$1,$B$1)”，如图10.37所示。

图10.36　计算其他年限的折旧值


[image: ]



图10.37　计算每年归还的利息

步骤4：计算每年归还的本金。现在需要计算的是D列的数据，用户可以使用PPMT函数计算这列的数据，然后利用自动填充功能完成数据，在单元格D4中输入公式“=-PPMT($F$1,YEAR(A4)-2005,$D$1,$B$1)”，如图10.38所示。

图10.38　计算每年归还的本金

步骤5：计算每年归还的本利总额。现在需要计算的是E列的数据，用户可以使用PMT函数计算这列的数据，然后利用自动填充功能完成数据，在单元E4中输入公式“=-PMT($F$1,$D$1,$B$1)”，如图10.39所示。

图10.39　计算每年的本利总额

步骤6：计算累计归还的利息。现在需要计算的是F列的数据，用户可以使用CUMIPMT函数计算这列的数据，然后利用自动填充功能完成数据，在单元F4中输入公式“=-CUMIPMT($F$1,$D$1,$B$1,1,YEAR(A4)-2005,0)”，如图10.40所示。

图10.40　计算累计利息

步骤7：计算累计归还的本金。现在需要计算的是G列的数据，用户可以使用CUMPRINC函数计算这列的数据，然后利用自动填充功能完成数据。在单元G4中输入公式“=-CUMPRINC($F$1,$D$1,$B$1,1,YEAR(A4)-2005,0)”，如图10.41所示。

图10.41　计算归还的累计本金

步骤8：计算未还贷款。现在需要计算的是H列的数据，用户可以直接将借款额和所还的本金相减，在单元格H4中输入公式“=$B$1-G4”，如图10.42所示。

图10.42　计算未还贷款额

步骤9：计算投资现值。现在需要计算的是I列的数据，用户可以使用函数NPV，计算这个投资的现值，在单元格I15中输入公式“=NPV(F1,I4:I13)”，如图10.43所示。

步骤10：计算报酬率。用户可以使用函数IRR，计算报酬率，在单元格I16中输入公式“=IRR(I3:I13)”，如图10.44所示。

图10.43　计算投资现值

图10.44　计算报酬率

步骤11：查看最后的结果。上面已经完成这个贷款表的情况，如图10.45所示。

图10.45　查看最后的结果


附录 Excel函数功能速查

逻辑函数

数学、三角和数组函数

统计函数

查询和引用函数

数据库函数

文本和数据函数

日期和时间函数

信息函数

财务函数

